

BASIC SU APPLE

SYBEX

Programmi
in pochi minuti

Stanley R. Trost

EDIZIONE ITALIANA



GRUPPO
EDITORIALE
JACKSON

BASIC SU APPLE

**Programmi
in pochi minuti**

Stanley R. Trost



**GRUPPO
EDITORIALE
JACKSON
Via Rosellini, 12
20124 Milano**

© Copyright per l'edizione originale Sybex Inc. 1983

© Copyright per l'edizione Italiana Sybex Inc. 1984

Il Gruppo Editoriale Jackson ringrazia per il prezioso lavoro svolto nella stesura dell'edizione italiana la signora Francesca Di Fiore e l'Ing. Roberto Pancaldi.

Traduzione italiana a cura del Dr. Pier Luigi Cecioni.

Tutti i diritti sono riservati. Stampato in Italia. Nessuna parte di questo libro può essere riprodotta, memorizzata in sistemi di archivio, o trasmessa in qualsiasi forma o mezzo, elettronico, meccanico, fotocopia, registrazione o altri senza la preventiva autorizzazione scritta dell'editore.

Fotocomposizione: Lineacomp S.r.l. - Via Rosellini, 12 - 20124 Milano

Stampato in Italia da:

S.p.A. Alberto Matarelli - Milano - Stabilimento Grafico

RINGRAZIAMENTI

Molte persone hanno dato un contributo di idee e notevole impegno verso la realizzazione di questo libro. Rudolph Langer lo ha suggerito e ha indicato come apportare numerose migliorie. David Lichtblau è stato di prezioso aiuto nella stesura e nella verifica dei programmi. Barbara Gordon e Joel Kreisman hanno rivisto il manoscritto, controllato i programmi e li hanno preparati per la produzione. Molti altri membri del personale della Sybex hanno formato il gruppo che ha dato vita a questo libro.

Stan Trost

SOMMARIO

Prefazione	IX
CAPITOLO 1 - INTRODUZIONE	1
Come cominciare	1
Commenti (Remarks)	2
Costituzione di una biblioteca di subroutine	3
Scrittura e lancio dei programmi	4
Caratteristiche interattive	5
Sommaro	5
CAPITOLO 2 - FINANZA DOMESTICA E COMMERCIALE	7
Valore finale di un deposito	7
Valore finale di una serie di depositi	10
Valore attuale di una somma	12
Valore attuale di una serie di pagamenti	14
Pagamenti per disporre in futuro di una somma	16
Prelievi di capitale	18
Determinazione del tasso di interesse	20
Valore attuale netto, flussi monetari non costanti	22
Tempo di raddoppio di un capitale	24
Tasso di interesse equivalente	26
Menù dei programmi finanziari	28
CAPITOLO 3 - PROGRAMMI COMMERCIALI PRATICI	31
Ammortamento a rate costanti	31
Ammortamento a rate variabili	34
Secondo metodo di ammortamento a rate variabili	36
Punto di equilibrio	38
Determinazione delle scorte ottimali	40
Prezzo di vendita scontato	42
Media ponderata	44
Commissione di vendita	46
Stipendio con straordinario	48
Strumento per prendere decisioni esecutive	50
Menù dei programmi commerciali	52

CAPITOLO 4 - PROGRAMMI IMMOBILIARI	55
Subroutine immobiliari	56
Calcolo della rata mensile	58
Piano di estinzione di un mutuo	60
Saldo su un mutuo	62
Effetto di pagamenti accelerati	64
Calcolo del debito residuo	66
Prezzo accessibile di una casa	68
Mutuo con altro mutuo	70
Analisi di proprietà affittate	72
Menù dei programmi immobiliari	74
CAPITOLO 5 - PROGRAMMI DI ANALISI DEI DI DATI	77
Menù per i programmi di analisi dei dati	78
Subroutine di analisi dei dati	80
Input dei dati	82
Grafico dei dati	84
Media e deviazione standard	86
Media mobile su tre punti	88
Media mobile ponderata	90
Media su quattro punti	92
Regressione lineare	94
CAPITOLO 6 - PROGRAMMI DI ARCHIVIAZIONE	97
Menù dell'archivio personale	98
Subroutine dell'archivio personale	100
Creazione di file e inserimento di record	102
Contenuto del file	104
Cancellazione di record	106
Ricerca sul file	108
Menù del diario automobilistico	110
Creazione del file e inserimento di record	112
Contenuto del diario automobilistico	114
Cancellazione di record	116
CAPITOLO 7 - PROGRAMMI DI ESERCIZI MATEMATICI	119
Esercizi matematici 1 - Menù	120
Subroutine degli esercizi matematici 1	122
Addizione	124
Sottrazione	126
Moltiplicazione	128
Divisione	130
Esercizi matematici 2 - Menù	132

Addizione in colonna	134
Moltiplicazione lunga	136
Esercizi matematici 3 - Menù	138
Subroutine relative alle frazioni	140
Addizione di frazioni	142
Sottrazione di frazioni	144
Moltiplicazione di frazioni	146
Divisione di frazioni	148
APPENDICE A - SUBROUTINE CENTRALI	151
Subroutine per l'inserimento di nuovi record	151
Subroutine di richiesta di nuova esecuzione	152
Subroutine per creare un file di copia (backup)	153
Subroutine per cancellare record	154
Subroutine per videate di dialoghi	155
Subroutine per input di caratteri	155
Subroutine per input di dati	156
Subroutine per l'intestazione e le videate	156
Subroutine per l'input dei parametri del programma	157
Subroutine per mostrare il menù	158
Subroutine di pausa	159
Subroutine sì o no	159
APPENDICE B - COME UTILIZZARE LE SUBROUTINE CENTRALI ...	161
Esempio di menù	162
Scelta di un altro programma	164
La subroutine di dialogo	166
Subroutine per l'immissione dei dati	168
APPENDICE C - UTILI SUBROUTINE PER LA STAMPANTE	171

PREFAZIONE

BASIC su Apple II - Programmi in pochi minuti vi aiuterà a scoprire tutta la potenza e la flessibilità di un Apple. Questo libro contiene più di 65 pratici programmi e subroutine totalmente verificati e pronti per essere lanciati, per un'ampia gamma di applicazioni commerciali, personali e didattiche. Per utilizzarli non è necessario conoscere il linguaggio di programmazione BASIC: basta inserirli nel computer e usarli a seconda delle proprie esigenze.

Questi programmi vi torneranno utili al momento di prendere importanti decisioni finanziarie; per esempio, potete trovare il punto di equilibrio di una nuova attività commerciale, calcolare i pagamenti mensili necessari per poter disporre in futuro di una determinata somma e trovare il valore futuro di un piano volontario di assicurazione. I programmi di archiviazione vi permetteranno di creare un indirizzario, una rubrica telefonica e un registro delle spese per l'automobile. Tanto gli studenti che coloro che non lo sono più troveranno particolarmente utili i programmi di esercitazione matematica, che consentono di migliorare nei calcoli e possono anche essere di aiuto agli studenti per i compiti a casa.

BASIC su Apple - Programmi in pochi minuti contiene numerosi programmi del tutto originali. È raro che una raccolta di questo genere comprenda esercitazioni sulle frazioni e su lunghe moltiplicazioni e analisi di dati associati a programmi grafici.

Anche l'uso di subroutine standard è originale. Uno dei principali scopi di questo libro è di fornire programmi *brevi* e facilmente inseribili nel computer, per questo sono stati creati tramite un insieme comune di subroutine, che, inoltre, possono semplificare notevolmente la stesura di altri programmi.

Questo libro non vuol insegnare il linguaggio di programmazione BASIC, però l'uso di questi programmi consentirà di impadronirvene in modo indolore e quasi automatico. Con la progressiva maggiore familiarità con il BASIC, i programmi forniti potranno essere modificati e se ne potrà creare di nuovi.

Passiamo adesso in rassegna il contenuto del libro. Il Capitolo 1 spiega come utilizzare i programmi ed esamina le caratteristiche interattive in essi incorporate. Questo capitolo va letto prima di usare i programmi.

Il Capitolo 2 contiene un gruppo di programmi finanziari per la famiglia e il lavoro. Comprende programmi che calcolano il valore futuro di un deposito bancario, quello di una serie di pagamenti (per esempio, per un'assicurazione volontaria), il tasso di interesse di un investimento e il tempo necessario per raddoppiare un capitale impiegato a un determinato tasso. Questi programmi possono essere utili per prendere molte importanti decisioni di investimento.

Il Capitolo 3 contiene una serie di programmi commerciali che permettono di stendere piani di ammortamento, determinare il punto di equilibrio di una nuova attività commerciale, calcolare stipendi con straordinario ed effettuare analisi decisionali. I programmi commerciali possono tornare utili a imprenditori, impiegati e consumatori.

Nel Capitolo 4 presentiamo programmi applicativi per il mercato immobiliare, compresi il calcolo del pagamento di un mutuo, del debito residuo sempre di un mutuo e un'analisi su pagamenti di affitti. Per esempio, si potrà trovare quanto si risparmia in interessi aumentando la rata che si paga su un mutuo. Investitori, e proprietari attuali e futuri di case troveranno utili questi programmi.

Il Capitolo 5 comprende un gruppo di programmi di analisi applicabili a dati sia scientifici che finanziari. Questi programmi permettono di calcolare la media e la deviazione standard, diverse medie mobili e di effettuare regressioni lineari, e inoltre consentono di tracciare grafici dei risultati. Tali programmi possono essere utilizzati anche per effettuare una proiezione delle tendenze della borsa.

Il Capitolo 6 contiene un gruppo di programmi che consentono di tenere archivi di vario genere: si possono compilare indirizzi e rubriche telefoniche e registrare i consumi dell'automobile. È inoltre facile modificare i programmi per altre applicazioni.

Il Capitolo 7 presenta numerosi utili esercizi aritmetici, per studenti di ogni livello e capacità, riferiti all'addizione, alla sottrazione, alla moltiplicazione e alla divisione, sia di numeri interi che di frazioni. Tali programmi permettono migliorare le capacità di calcolo e aiutano addirittura nei compiti a casa.

Il libro ha anche tre appendici. L'Appendice A elenca le subroutine comuni grazie alle quali i programmi dei vari capitoli sono così brevi. Le loro funzioni sono illustrate tramite istruzioni REM. Queste subroutine sono in ordine numerico, per facilitarne il richiamo. L'Appendice B mostra, con esempi e descrizioni, come utilizzarle per creare altri programmi.

L'Appendice C elenca svariate routine che permettono di sfruttare tutte le possibilità di una stampante Epson. Questi programmi possono essere utilizzati per accedere a tutte le modalità di stampa disponibili (condensata, allargata, ribattuta ecc.).

INTRODUZIONE

Questo capitolo mostra come *BASIC su Apple - Programmi in pochi minuti* possa rendere un Apple notevolmente più efficace. Imparerete a inserire i programmi nel computer e a lanciali, creare una biblioteca di subroutine e una di programmi omogenei controllata da menù. Vedrete anche cosa significhi programmazione interattiva e come si estrinsechi in questi programmi.

Per utilizzarli non è necessario conoscere il BASIC; li si può usare così come sono, per risolvere una quantità di problemi commerciali, domestici e didattici. Se però volete ampliare la vostra conoscenza della programmazione, questi programmi mostrano come sfruttare al massimo le grandi possibilità di un computer.

COME COMINCIARE

È necessario un computer Apple II o Apple IIe con una memoria di almeno 64 K e un drive. I programmi di questo libro sono scritti in Applesoft BASIC; le istruzioni sul caricamento e l'utilizzazione del BASIC sono spiegate nel manuale del linguaggio.

Inserite nel drive un dischetto inizializzato con il DOS e accendete il computer. Dopo una breve pausa, sullo schermo compare il cursore:

]

Togliete il dischetto dal drive e rimettetelo a posto.

Per memorizzare i programmi che scriverete ci vorranno due dischi inizializzati (le istruzioni di inizializzazione sono nel manuale Apple DOS), un disco cosiddetto "master" e l'altro per effettuare una copia di riserva. È bene far sempre due copie di ciascun programma.

A questo punto potete inserire nel computer qualunque programma di questo libro. Per farlo, occorre battere le righe dei listati esattamente come compaiono nel libro. Dopo ciascuna riga battere il tasto RETURN. Proviamo con il programma che segue:

```
100 PRINT "HELLO"  
110 END
```

Cominciate scrivendo la prima riga (100 PRINT "HELLO") e inseritela battendo il tasto RETURN. Completate l'inserimento battendo e inserendo anche la seconda riga.

Durante questa operazione, talvolta commetterete errori. Le righe brevi possono essere facilmente corrette ribattendole interamente; il manuale dell'Applesoft spiega altri metodi per la correzione di errori sulle righe più lunghe.

Per lanciare questo programma, battete RUN 100 e premete il tasto RETURN. Si noti cosa compare sullo schermo: il programma ha scritto HELLO sotto l'elenco di istruzioni che avete inserito. Nel prosieguo del capitolo torneremo sull'inserimento, la memorizzazione e il lancio dei programmi.

COMMENTI (REMARKS)

I programmi possono essere resi più comprensibili tramite commenti (*remarks*). Le righe di istruzione che contengono un commento iniziano con il simbolo REM. Nella Figura 1.1, i *remarks* formano un bordo nella parte superiore del listato e spiegano la funzione della riga 130. A parte un'eccezione, quando inserite questi programmi non è necessario che scriviate anche i commenti. L'eccezione è che la prima riga di ciascun programma, dato il modo in cui i programmi sono stati concepiti, deve contenere la parola REM. La Figura 1.2 mostra il listato dello stesso programma dopo che sono state battute le righe necessarie.

```
100 REM -----  
101 N$ = "PROGRAMMA DIMOSTRATIVO"  
102 REM -----  
110 INPUT X  
120 INPUT Y  
130 PRINT X+Y: REM STAMPA LA SOMMA DI X E Y  
140 END
```

Figura 1.1: Listato di programma con *remarks*

```
100 REM  
101 N$ = "PROGRAMMA DIMOSTRATIVO"  
110 INPUT X  
120 INPUT Y  
130 PRINT X+Y  
140 END
```

Figura 1.2: Listato di programma con *remarks* condensati

COSTITUZIONE DI UNA BIBLIOTECA DI SUBROUTINE

L'inserimento dei programmi di questo libro e la formattazione del loro output sono semplificati grazie a una biblioteca centrale di subroutine. Una *subroutine* non è che un programma utilizzato come parte di un altro programma. Per esempio, l'immissione dei dati per molti dei nostri programmi è gestita da una subroutine. Queste subroutine consentono di limitare la lunghezza dei programmi e quindi di facilitarne l'inserimento.

L'Appendice A contiene i listati delle subroutine utilizzate in questo libro. L'Appendice B mostra come servirsi di queste subroutine per creare altri programmi.

L'elenco che segue mostra le subroutine da inserire per i vari capitoli del libro. Le subroutine andrebbero scritte e memorizzate su disco via via che sono necessarie, così ognuna viene scritta una sola volta. I numeri dell'elenco sono quelli della prima riga di ciascuna subroutine.

Capitolo 2-7	4500, 6000, 6500, 7000, 7500, 8000, 8500, 9000, 9500
Capitolo 6	4000, 5000, 5500

Cominceremo costituendo la biblioteca di subroutine per i Capitoli da 2 a 5; i loro listati sono nell'Appendice A. Dato che la subroutine della riga 4000 non è necessaria fino al Capitolo 6, cominciate con la riga 4500. Innanzi tutto scrivete NEW, per cancellare dal computer eventuali programmi precedenti, poi inserite le istruzioni di questa subroutine. (Ricordate che i *remarks*, a parte quello della prima riga, non vanno scritti.) Quello che avete battuto dovrebbe essere uguale a quanto compare nella Figura 1.3.

Continuate a inserire le altre subroutine necessarie per i capitoli 2-5 e 7, dopo di che memorizzatele su disco. Seguite le istruzioni del manuale DOS per poter utilizzare il disco, poi battete

SAVE SUBLIB

La libreria centrale di subroutine è così memorizzata in un file di nome SUBLIB. Questo file verrà unito ai programmi applicativi al momento della loro utilizzazione. Le subroutine che rimangono vanno inserite in SUBLIB quando scriveremo i programmi del Capitolo 6.

```
4500 REM
4660 Q1$ = "VOLETE "
4670 Q3$ = "DI NUOVO? (S O N) "
4690 GOSUB 6000
4710 GOSUB 9500
4720 RETURN
```

Figura 1.3: Listato di programma: subroutine ANCORA

SCRITTURA E LANCIO DEI PROGRAMMI

Siete adesso pronti a scrivere un programma. Battetene le istruzioni come compaiono nei listati. I programmi dei Capitoli 2, 3 e 4 dovrebbero essere memorizzati in un file per capitolo (usate nomi di file che abbiano un significato), dopo di che possono essere lanciati caricando la libreria di subroutine, unendola al file del capitolo desiderato e battendo RUN e il numero della prima riga del programma scelto. Una volta inseriti tutti i programmi di un capitolo, potete compilare un menù per lanciarli. Un menù è un elenco di programmi utilizzabile per scegliere quello che si desidera lanciare. Ogni capitolo contiene istruzioni per creare il proprio menù. I numeri delle prime righe dei programmi di ciascun capitolo sono tutti diversi gli uni dagli altri, il che consente di memorizzare in un unico file i programmi e il menù dei vari capitoli.

Per esempio, il Capitolo 2 contiene alcuni programmi utili nei calcoli finanziari. Scrivete il programma del valore futuro di un investimento e memorizzatelo su disco battendo

SAVE FIN

Quando avrete bisogno di un altro programma finanziario, caricate il primo che avete memorizzato battendo

LOAD FIN

Adesso potete scrivere l'altro programma per poi memorizzarli tutti e due insieme battendo di nuovo

SAVE FIN

È chiaro come questo metodo consenta di costituire una libreria di programmi finanziari.

Le subroutine e i programmi applicativi vengono uniti tramite il programma in Applesoft RENUMBER. Inserite il dischetto DOS nel drive e battete RUN RENUMBER. Adesso inserite il vostro dischetto con i programmi nel drive ed effettuate le seguenti operazioni (accertatevi di aver battuto il tasto RETURN alla fine di ciascuna riga):

```
LOAD SUBLIB
&H
LOAD FIN
&M
RUN300
```

Nota: il programma RENUMBER impiega circa due minuti per riunire le subroutine e i programmi. Prima di battere RUN, aspettate che compaia il cursore.

Dopo aver inserito tutti i programmi del Capitolo 2, scrivete e memorizzate il Menù corrispondente, che vi permetterà di lanciare qualunque programma di quel capitolo scegliendolo dal menù, senza dover più ricordare numeri di riga. Perché il programma del menù possa funzionare, il comando END nell'ultima riga di ciascun programma va sostituito con il comando RETURN.

I programmi nei Capitoli 5-7 devono essere inseriti in un modo leggermente diverso: il programma del menù è all'inizio invece che alla fine, quindi prima va scritto il menù, poi il programma desiderato. I programmi vengono lanciati scegliendoli dal menù. Poiché il menù dei programmi di questi capitoli è all'inizio, non occorre effettuare la sostituzione delle istruzioni END.

CARATTERISTICHE INTERATTIVE

Il termine *interattivo* si riferisce all'interazione fra computer, programmi e utente. Scopo di un programma interattivo è di semplificare la comunicazione fra utente e computer e viceversa.

Questo fine viene conseguito in diversi modi. Quando è possibile, si risponde alle domande premendo un solo tasto, senza dover battere RETURN o ENTER. Per esempio, i programmi vengono selezionati dai menù con una sola cifra e per rispondere sì/no a una domanda basta battere una S o una N. Un'altra caratteristica interattiva consiste nel far scrivere tutte le domande in fondo allo schermo, perché siano sempre nello stesso posto. Le risposte compaiono nella parte superiore dello schermo, perché si possa sempre esaminare i dati inseriti. Viene utilizzato anche il sonoro, per avvertire di possibili errori di input, in particolare, nel caso di risposte sbagliate negli esercizi aritmetici.

SOMMARIO

Questo capitolo contiene le istruzioni dettagliate per creare e memorizzare una libreria di subroutine e per poi utilizzarla con gli altri programmi del libro. Il capitolo spiega anche come creare una libreria di programmi collegati fra loro e selezionabili tramite menù. Per inserire un nuovo programma, lo si aggiunge a quelli già memorizzati.

L'uso di una libreria centrale di subroutine permette a questi programmi di essere molto brevi e facili da inserire, mentre le caratteristiche interattive ne semplificano l'utilizzazione.

FINANZA DOMESTICA E COMMERCIALE

Questo capitolo contiene dieci programmi utili per molte applicazioni finanziarie domestiche e commerciali. Per esempio, supponiamo di voler depositare denaro in un conto bancario per un determinato periodo. Una banca paga mensilmente il 12% composto e un'altra il 12,25% composto trimestralmente. Con questi programmi si può usare il computer per trovare chi paga più interessi.

Può anche darsi che si voglia stipulare un'assicurazione volontaria di pensionamento. Uno dei programmi di questo capitolo permette di calcolare la somma a disposizione al momento del pensionamento. Con un altro programma si può vedere se un capitale impiegato a un certo interesse è sufficiente per smettere di lavorare e vivere di rendita.

L'ultimo programma del capitolo mostra come riunire tutti gli altri per avere un package finanziario controllato da menù.

VALORE FINALE DI UN DEPOSITO

Descrizione

Vi siete mai domandati di quanto aumenta il valore di un deposito bancario? Vi piacerebbe mettere a confronto l'effetto di tassi di interesse e di periodi di capitalizzazione diversi?

Questo programma consente di calcolare il valore finale di un deposito. Vanno specificati la somma depositata, il numero dei periodi di capitalizzazione in un anno, il tasso di interesse annuo e il tempo d'impiego espresso in anni. Il programma assume che il tasso di interesse sia costante.

Esempio

Tommaso deposita 5.000.000 in un conto vincolato per quattro anni. La banca paga un interesse del 16% annuo, capitalizzato trimestralmente. Tommaso vuol sapere di quanto aumenteranno i suoi risparmi in quattro anni. La Figura 2.1 mostra i risultati del calcolo come appaiono sullo schermo. La Figura 2.2 è il listato del programma per calcolare il valore finale.

VALORE FINALE DI UN DEPOSITO

DEPOSITO INIZIALE	5000000
PERIODI CAPIT. ANNUI	4
INTERESSE ANNUO (%)	16
NUMERO DI ANNI	4
VALORE FINALE	9364906

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.1: Videata: Valore finale di un deposito

```
300 REM -----  
310 N$ = "VALORE FINALE DI UN DEPOSITO"  
320 REM -----  
330 K = 1  
340 GOSUB 7500 : REM INTESAZIONE  
350 REM RICHIESTA DI PARAMETRI  
360 Q1$ = "INSERIRE DATI:" Q2$ = ""  
370 Q3$ = "DEPOSITO INIZIALE" : GOSUB 8000  
380 Q3$ = "PERIODI CAPIT. ANNUI" : GOSUB 8000  
390 Q3$ = "INTERESSE ANNUO (%)" : GOSUB 8000  
400 Q3$ = "NUMERO DI ANNI" : GOSUB 8000  
410 REM EFFETUA CALCOLO  
420 PRINT : PRINT "VALORE FINALE"; TAB(23)  
430 A = PAR(1)  
440 FOR I = 1 TO PAR(2) * PAR(4)  
450 A = A + A * PAR(3) / 100 / PAR(2)  
460 NEXT I  
470 A = INT (A * 100 + .5) / 100  
480 PRINT A  
490 REM RICHIESTA SE ANCORA  
500 Q2$ = "CALCOLARE":GOSUB 4500  
510 IF YN$ = "N" THEN END  
520 GOTO 300
```

Figura 2.2: Listato di programma: Valore finale di un deposito

VALORE FINALE DI UNA SERIE DI DEPOSITI

Descrizione

Questo programma calcola il valore finale di una serie di depositi regolari. Per esempio, potete calcolare il valore finale di un'assicurazione volontaria a versamenti costanti. Specificate l'importo pagato, il numero di pagamenti annui, il tasso annuo di interesse e per quanti anni effettuerete i pagamenti. Il programma assume che il numero dei periodi di capitalizzazione annui e quello dei pagamenti annui siano uguali, e che il tasso di interesse sia costante.

Esempio

Anna intende depositare in banca 500.000 lire al mese. Le viene corrisposto un interesse annuo del 13%, capitalizzato mensilmente. Quanto avrà nel conto dopo quattro anni? La Figura 2.3 mostra cosa appare sullo schermo per questo esempio. Il listato del programma è nella Figura 2.4.

VALORE FINALE DI UNA SERIE

PAGAMENTO	500000
PAGAMENTI ANNUI	12
INTERESSE ANNUO (%)	13
NUMERO DI ANNI	4

VALORE FINALE ***** da calcolare *****

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.3 Videata: Valore finale di una serie di depositi

```
530 REM -----  
540 N$ = "VALORE FINALE DI UNA SERIE"  
550 REM -----  
560 K = 1  
570 GOSUB 7500 : REM INTESTAZIONE  
580 REM RICHIESTA DI PARAMETRI  
590 Q1$ = "INSERIRE DATI:"Q2$ = ""  
600 Q3$ = "PAGAMENTO" : GOSUB 8000  
610 Q3$ = "PAGAMENTI ANNUI" : GOSUB 8000  
620 Q3$ = "INTERESSE ANNUO (%)" : GOSUB 8000  
630 Q3$ = "NUMERO DI ANNI" : GOSUB 8000  
640 REM CALCOLA RISULTATO  
650 PRINT : PRINT "VALORE FINALE"; TAB(23)  
660 A = 0  
670 FOR I = 1 TO PAR(2) * PAR(4)  
680 A = PAR(1) + A + A * PAR(3) / 100 / PAR(2)  
690 NEXT I  
700 A = INT (A * 100 + .5) / 100  
710 PRINT A  
720 REM RICHIESTA SE ANCORA  
730 Q2$ = "CALCOLARE" : GOSUB 4500  
740 IF YN$ = "N" THEN END  
750 GOTO 530
```

Figura 2.4: Listato di programma: Valore finale di una serie di depositi

Pagina mancante

Pagina mancante

VALORE ATTUALE DI UNA SERIE DI PAGAMENTI

Descrizione

L'esempio precedente mostra come trovare il valore attuale di una somma futura; con questo programma potete trovare il valore attuale di una serie di pagamenti. Per esempio, supponiamo che Sandra compri la vostra auto e vi paghi 50.000 al mese per tre anni. Dopo un anno vi offre 900.000 lire tutte insieme: volete sapere se vi conviene accettare la sua offerta. Questo programma consente di trovare la risposta; dovrete accettare se il valore attuale del credito è inferiore a 900.000 lire.

Per usare questo programma vanno specificati l'entità del pagamento, il numero dei pagamenti annui, per quanti anni verranno effettuati i pagamenti e il tasso di interesse annuo che si può ottenere su una somma. Si assume che il numero dei pagamenti annui e quello dei periodi di capitalizzazione annui siano identici.

Esempio

Maria riscuote pagamenti di 400.000 lire al mese. Investendo il proprio denaro può ottenere un interesse del 10% e vuol conoscere il valore attuale dei pagamenti. I pagamenti continueranno per dieci anni. La Figura 2.7 mostra il risultato del problema di questo esempio e la Figura 2.8 il listato completo del programma.

VALORE ATTUALE DI UNA SERIE

PAGAMENTO	400000
PAGAMENTI ANNUI	10
INTERESSE ANNUO (%)	15
NUMERO DI ANNI	10
VALORE ATTUALE	30268460

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.7: Videata: Valore attuale di una serie di pagamenti

```
970  REM -----  
980  N$ = "VALORE ATTUALE DI UNA SERIE"  
990  REM -----  
1000 K = 1  
1010  GOSUB 7500 : REM INTESTAZIONE  
1020  REM RICHIESTA DI PARAMETRI  
1030  Q1$ = "INSERIMENTO DATI:" : Q2$ = ""  
1040  Q3$ = "PAGAMENTO" : GOSUB 8000  
1050  Q3$ = "PAGAMENTI ANNUI" : GOSUB 8000  
1060  Q3$ = "INTERESSE ANNUO (%)" : GOSUB 8000  
1070  Q3$ = "NUMERO DI ANNI" : GOSUB 8000  
1080  REM EFFETTUA I CALCOLI  
1090  PRINT : PRINT "VALORE ATTUALE"; TAB(23);  
1100  A = 0  
1110  FOR I = 1 TO PAR(2) * PAR(4)  
1120  A = A + PAR(1) / ((1 + PAR(3) / 100 / PAR(2)) ^ I)  
1130  NEXT I  
1140  A = INT (A * 100 + .5) / 100  
1150  PRINT A  
1160  REM RICHIESTA SE ANCORA  
1170  Q2$ = "CALCOLARE":GOSUB 4500  
1180  IF YN$ = "N" THEN END  
1190  GOTO 970
```

Figura 2.8: Listato di programma: Valore attuale di una serie di pagamenti

PAGAMENTI PER DISPORRE IN FUTURO DI UNA SOMMA

Descrizione

Questo programma consente di determinare i pagamenti regolari necessari per disporre in futuro di una determinata somma di denaro. Per esempio, si può usare questo programma per calcolare i pagamenti necessari per poi avere i soldi per una vacanza, o per sostituire un pezzo di un apparecchiatura o per le spese universitarie. Va specificata la somma futura, il numero di pagamenti annui, il tasso di interesse annuo ottenibile su un investimento e fra quanti anni si desidera il denaro. Il programma assume che il numero dei pagamenti annui e quello dei periodi di capitalizzazione annui siano identici.

Esempio

Sandro ha intenzione di fare versamenti mensili per costituire un fondo per le spese universitarie della figlia. Effettuerà questi pagamenti per 18 anni, e calcola che avrà bisogno di 40.000.000. Su un conto di questo genere può ottenere un interesse del 12%. La Figura 2.9 mostra il risultato di questo esempio e la Figura 2.10 il listato del programma.

PAGAMENTO PER SOMMA FUTURA

VALORE FUTURO	40000000
PAGAMENTI ANNUI	12
INTERESSE ANNUO (%)	12
NUMERO DI ANNI	18

PAGAMENTO ***** valore da calcolare *****

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.9: Videata: Pagamenti per disporre in futuro di una somma

```
1200 REM -----  
1210 N$ = "PAGAMENTO PER SOMMA FUTURA"  
1220 REM -----  
1230 K = 1  
1240 GOSUB 7500 : REM INTESTAZIONE  
1250 REM RICHIESTA DI PARAMETRI  
1260 Q1$ = "INSERIMENTO DATI" : Q2$ = ""  
1270 Q3$ = "VALORE FUTURO" : GOSUB 8000  
1280 Q3$ = "PAGAMENTI ANNUI" : GOSUB 8000  
1290 Q3$ = "INTERESSE ANNUO (%)" : GOSUB 8000  
1300 Q3$ = "NUMERO DI ANNI" : GOSUB 8000  
1310 REM EFFETTUA I CALCOLI  
1320 PRINT : PRINT "PAGAMENTO"; TAB(23)  
1330 I = PAR(3) / 100 / PAR(2)  
1340 Q = PAR(2) * PAR(4)  
1350 A = PAR(1) * I / ((1 + I) ^ Q - 1)  
1360 A = INT (A * 100 + .5) / 100  
1370 PRINT A  
1380 REM RICHIESTA SE ANCORA  
1390 Q2$ = "CALCOLARE" : GOSUB 4500  
1400 IF YN$ = "N" THEN END  
1410 GOTO 1200
```

Figura 2.10: Listato di programma: pagamento per somma futura

PRELIEVI DI CAPITALE

Descrizione

Questo programma può essere utilizzato per determinare per quanto tempo si possono fare prelievi da un conto in banca prima di rimanere senza denaro. Vanno specificati il capitale iniziale, il numero di prelievi annui, il tasso di interesse ottenuto sul capitale rimanente e l'entità del prelievo. Ancora una volta il programma assume che il numero dei prelievi annui e quello dei periodi di capitalizzazione siano identici.

Se la somma prelevata è inferiore all'interesse ottenuto; il conto in banca aumenterà invece di estinguersi. Il programma mostra allora il messaggio **IL CAPITALE STA AUMENTANDO**. Non preoccupatevi se l'esecuzione del programma richiede molto tempo; le operazioni vengono ripetute finché non viene trovato un saldo zero.

Esempio

Paolo e Maria sono sul punto di andare in pensione. Hanno 200.000.000 in banca e ottengono un interesse del 15%. Paolo e Maria intendono vivere con 1.600.000 lire al mese. Per quanto tempo potranno continuare a prelevare soldi? La Figura 2.11 mostra i risultati di questo esempio; il listato del programma compare nella Figura 2.12.

PRELIEVO DI CAPITALI

DEPOSITO INIZIALE	200000000
PRELIEVI ANNUI	12
INTERESSE ANNUO	15
PAGAMENTO	1600000

PAGAMENTI ***** valore da calcolare *****

ANNI DI PAGAMENTI ***** valore da calcolare *****

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.11: Videata: Prelievi di capitali

```
1420 REM -----
1430 N$ = "PRELIEVI DI CAPITALE"
1440 REM -----
1450 K = 1
1460 GOSUB 7500 : REM INTESTAZIONE
1470 REM RICHIESTA DI PARAMETRI
1480 Q1$ = "INSERIMENTO DATI";Q2$ = ""
1490 Q3$ = "DEPOSITO INIZIALE" : GOSUB 8000
1500 Q3$ = "PRELIEVI ANNUI" : GOSUB 8000
1510 Q3$ = "INTERESSE ANNUO (%)" : GOSUB 8000
1520 Q3$ = "PAGAMENTO" : GOSUB 8000
1530 REM EFFETTUA I CALCOLI
1540 PRINT : PRINT "PAGAMENTI"; TAB(23)
1550 A = PAR(1);N = 1
1560 A = A + A * PAR(3) / 100 / PAR(2) - PAR(4)
1570 IF A < PAR(1) THEN 1610
1580 PRINT : PRINT "IL CAPITALE STA AUMENTANDO"
1590 PRINT CHR$( 7)
1600 GOTO 1660
1610 IF A < PAR(4) THEN 1640
1620 N= N + 1
1630 GOTO 1560
1640 PRINT N
1650 PRINT "ANNI DI PAGAMENTI"; TAB(23); N/12
1660 REM RICHIESTA SE ANCORA
1670 Q2$ = "CALCOLARE" : GOSUB 4500
1680 IF YN$ = "N" THEN END
1690 GOTO 1420
```

Figura 2.12: Listato di programma: Prelievi di capitale

DETERMINAZIONE DEL TASSO DI INTERESSE

Descrizione

Il programma di questo esempio è utile per determinare il tasso di interesse quando si conoscono i valori iniziali e finali di un investimento. Per esempio, si può trovare il tasso di interesse sul guadagno ottenuto dalla vendita di una casa. Per usare questo programma si specificano i valori iniziale e finale del capitale, il numero dei periodi di capitalizzazione annui e il numero di anni durante i quali il capitale è aumentato.

Esempio

Quattro anni fa zia Amelia dette 5.000.000 al proprio agente di borsa perchè li investisse. Ha appena ottenuto 9.000.000. Desidera calcolare il tasso di interesse equivalente che ha ottenuto, con un periodo di capitalizzazione trimestrale. Le Figure 2.13 e 2.14 mostrano rispettivamente i risultati come appaiono sul video e il listato del programma di questo esempio.

DETERMINAZIONE DEL TASSO DI INTERESSE

SOMMA INVESTITA	5000000
SOMMA RICAVATA	9000000
PERIODI ANNUI	4
NUMERO DI ANNI	4
INTERESSE ANNUO (%)	14.97

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.13: Videata: Determinazione del tasso di interesse

```
1700 REM -----  
1710 N$ = "DETERMINAZIONE DEL TASSO DI INTERESSE"  
1720 -----  
1730 K = 1  
1740 GOSUB 7500 : REM INTESTAZIONE  
1750 REM RICHIESTA DI PARAMETRI  
1760 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""  
1770 Q3$ = "SOMMA INVESTITA" : GOSUB 8000  
1780 Q3$ = "SOMMA RICAVATA" : GOSUB 8000  
1790 Q3$ = "PERIODI ANNUI" : GOSUB 8000  
1800 Q3$ = "NUMERO DI ANNI" : GOSUB 8000  
1810 REM EFFETTUA I CALCOLI  
1820 PRINT : PRINT "INTERESSE ANNUO (%)"; TAB (23)  
1830 Q = PAR(3) * PAR(4)  
1840 I = ((PAR(2) / PAR(1)) ^ (1 / Q) - 1) * 100 * PAR(3)  
1850 I = INT (I * 100 + .5) / 100  
1860 PRINT I  
1870 REM RICHIESTA SE ANCORA  
1880 Q2$ = "CALCOLARE": GOSUB 4500  
1890 IF YN$ = "N" THEN END  
1900 GOTO 1700
```

Figura 2.14: Listato di programma: Determinazione del tasso di interesse

VALORE ATTUALE NETTO, FLUSSI MONETARI NON COSTANTI

Descrizione

L'analisi basata sul valore attuale netto viene utilizzata dagli analisti finanziari per stabilire se un investimento è conveniente. Specificato un tasso di interesse, un valore attuale netto positivo indica che l'investimento è desiderabile.

Questo programma è utile quando si desidera stabilire il valore attuale netto di una serie di flussi monetari, che possono non essere uguali tutti gli anni. Per utilizzare questo programma si inseriscono l'investimento iniziale e il tasso di interesse, dopo di che è il programma a richiedere ciascun flusso monetario che deve essere fornito come numero positivo o negativo, a seconda che il denaro sia stato ricevuto o distribuito. Il termine degli inserimenti viene comunicato battendo 0.

Esempio

Tina può acquistare un titolo per 5.000.000. Otterrà pagamenti annui di 500.000, 1.500.000, 1.500.000 e 1.000.000. Tina vorrebbe guadagnare sul suo capitale il 12% annuo. Qual'è il valore attuale del suo investimento? Nella Figura 2.15 compare una cifra negativa, a indicare che non ottiene il 12% annuo e che quindi non dovrebbe comprare il titolo. Il listato completo del programma è nella Figura 2.16

VALORE ATTUALE NETTO, FLUSSI NON COSTANTI	
SOMMA INVESTITA	5000000
INTERESSE ANNUO (%)	12
VALORE ATTUALE NETTO	-769410
FLUSSO 1:	500
FLUSSO 2:	1500
FLUSSO 3:	1500
FLUSSO 4:	1500
FLUSSO 5:	1000
FLUSSO 6	?0

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.15: Videata: Valore attuale netto, flussi monetari non costanti

```

1910 REM -----
1920 N$ = "VALORE ATTUALE NETTO, FLUSSI NON COSTANTI"
1930 REM -----
1940 DIM PV(100)
1950 GOSUB 7500 : REM INTESTAZIONE
1960 K = 1
1970 REM RICHIESTA DI PARAMETRI
1980 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
1990 Q3$ = "SOMMA INVESTITA" : GOSUB 8000
2000 Q3$ = "INTERESSE ANNUO (%)" : GOSUB 8000
2010 Q1$ = "INSERIRE I FLUSSI ANNUALI"
2020 Q2$ = "NELL'ORDINE DI RICEVIMENTO"
2030 Q3$ = "BATTERE 0 PER TERMINARE"
2040 GOSUB 6000
2050 FOR I = 1 TO 100
2060 VTAB 18: HTAB 1: PRINT SPC( 35)
2070 VTAB 18: HTAB 1
2080 PRINT "FLUSSO ";I;"
2090 INPUT PV(I)
2100 IF PV(I) = 0 THEN 2150
2110 VTAB I - IN (I / 8) * 8 + 9: PRINT SPC( 35)
3120 VTAB I - IN (I / 8) * 8 + 9: HTAB 5
2130 PRINT "FLUSSO "; I; " "; TAB(25); PV(I)
2140 NEXT I
2150 NPV = - PAR(1)
2160 FOR I = 1 TO 100
2170 NPV = NPV + PV(I) / (1 + PAR(2) / 100) ^ I
2180 NEXT I
2190 VTAB 7: HTAB 1
2200 PRINT "VALORE ATTUALE NETTO"; TAB(23)
2210 A = INT (NPV * 100 + .5) / 100
2220 PRINT A
2230 FOR I = 1 TO 100
2240 PV(I) = 0
2250 NEXT I
2260 RICHIESTA SE ANCORA
2270 Q2$ = "CALCOLARE" : GOSUB 4500
2280 IF YN$ = "N" THEN END
2290 GOTO 1950

```

Figura 2.16: Listato di programma: Valore attuale netto, flussi non costanti

TEMPO DI RADDOPPIO DI UN CAPITALE

Descrizione

Questo programma calcola il tempo necessario perchè un capitale investito raddoppi. Per utilizzare questo programma vanno forniti il tasso annuo di interesse e il numero dei periodi di capitalizzazione annui.

Esempio

La banca di Franco capitalizza gli interessi quattro volte all'anno e paga il 12% di interesse annuo. Quanti anni ci vorranno a Franco per raddopiare un investimento di 5.000.000? I risultati compaiono nella Figura 2.17, il listato nella Figura 2.18.

TEMPO DI RADDOPPIO

PERIODI ANNUI	4
INTERESSE ANNUO (%)	12
TEMPO DI RADDOPPIO (ANNI)	5.86

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.17: Videata: Tempo di raddoppio di un capitale

```
2300 REM -----
2310 N$ = "TEMPO DI RADDOPPIO"
2320 REM -----
2330 REM INTESTAZIONE
2340 K = 1: GOSUB 7500
2350 REM RICHIESTA DI PARAMETRI
2360 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
2370 Q3$ = "PERIODI ANNUI" : GOSUB 8000
2380 Q3$ = "INTERESSE ANNUO (%)" : GOSUB 8000
2390 REM EFFETTUA I CALCOLI
2400 PRINT : PRINT "TEMPO DI RADDOPPIO (ANNI)"; TAB (25)
2410 I = PAR(2) / 100 / PAR(1)
2420 Y = LOG(2) / (PAR(1) * LOG (1 + I))
2430 Y = INT (Y * 100 + .5) / 100
2440 PRINT Y
2450 REM RICHIESTA SE ANCORA
2460 Q2$ = "CALCOLARE" : GOSUB 4500
2470 IF YN$ = "N" THEN GOTO
2480 GOTO 2300
```

Figura 2.18: Listato di programma: Tempo di raddoppio

TASSO DI INTERESSE EQUIVALENTE

Descrizione

Molti programmi di questo capitolo assumono che il numero dei periodi di capitalizzazione e il numero dei pagamenti siano identici. Questo programma permette di determinare un tasso di interesse equivalente quando due periodi non sono uguali. Per usare questo programma, inserire il tasso di interesse annuo, il numero dei periodi di capitalizzazione e il numero di pagamenti annui che si vuole effettuare (periodi desiderati). Il programma calcola un tasso di interesse equivalente utilizzabile.

Esempio

Effettuate versamenti mensili in un conto e desiderate determinare il valore futuro dei depositi. L'interesse su questo conto è del 12% annuo, capitalizzato trimestralmente. Come mostra la Figura 2.19, il tasso di interesse equivalente è del 11.88%. Per ottenere la risposta finale va usata questa cifra e poi lanciato il programma del valore finale di una serie di versamenti. Il listato di questo programma è nella Figura 2.20

TASSO DI INTERESSE EQUIVALENTE

INTERESSE ANNUO (%)	12
PERIODI DISPONIBILI	4
PERIODI DESIDERATI	12
INTERESSE EQUIVAL.	11.88

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 2.19: Videata: Tasso di interesse equivalente

```
2490 REM -----  
2500 N$ = "TASSO DI INTERESSE EQUIVALENTE"  
2510 REM -----  
2520 K = 1  
2530 GOSUB 7500 : REM INTESTAZIONE  
2540 REM RICHIESTA DI PARAMETRI  
2550 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""  
2560 Q3$ = "INTERESSE ANNUO (%)" : GOSUB 8000  
2570 Q3$ = "PERIODI DISPONIBILI" : GOSUB 8000  
2580 Q3$ = "PERIODI DESIDERATI" : GOSUB 8000  
2590 I = PAR(1):Q2 = PAR(2):Q1 = PAR(3)  
2600 IEQ = Q1 * (1 + I / 100 / Q2) ^ (Q2 / Q1) - Q1  
2610 PRINT : PRINT "INTERESSE EQUIVAL.": TAB(25)  
2620 IEQ = INT (IEQ * 10000 + .5) / 100  
2630 PRINT IEQ  
2640 REM RICHIESTA SE ANCORA  
2650 Q2$ = "CALCOLARE" : GOSUB 4500  
2660 IF YN$ = "N" THEN END  
2670 GOTO 2490
```

Figura 2.20: Listato di programma: Tasso di interesse equivalente

MENU' DEI PROGRAMMI FINANZIARI

Tutti i programmi di questo capitolo sono stati inseriti con numeri di riga distinti gli uni dagli altri. Se memorizzate tutti i programmi insieme in un file su disco, potete usare il programma che segue per costruire un menù da cui lanciare i vari programmi. Il menù è concepito in modo da poter scegliere premendo un solo tasto.

Ricordate che quando questi programmi vengono usati con il menù, l'istruzione END alla fine di ciascun programma va sostituita con un'istruzione RETURN. Per inserire il programma del menù, caricare prima i programmi finanziari, poi scrivere quello del menù. Sia l'uno che gli altri vanno poi memorizzati su dischetto con il comando SAVE. Il menù compare nella Figura 2.21. La Figura 2.22 mostra il menù di scelta.

Per tornare al BASIC dopo aver lanciato i programmi finanziari desiderati, premere il tasto RESET: il cursore comparirà su un rigo vuoto e si potrà inserire un altro gruppo di programmi.

1= VALORE FINALE DI UN DEPOSITO
2= VALORE FINALE DI UNA SERIE
3= VALORE ATTUALE DI UNA SOMMA
4= VALORE ATTUALE DI UNA SERIE
5= PAGAMENTI PER DISPORRE DI SOMMA FUTURA
6= PRELIEVI DI CAPITALE
7= DETERMINAZIONE DEL TASSO DI INTERESSE
8= VALORE ATTUALE NETTO, FLUSSI NON COSTANTI
9= TEMPO DI RADDOPPIO
0= TASSO DI INTERESSE EQUIVALENTE

PROGRAMMA SCELTO:

Figura 2.21: Videata: Menù dei programmi finanziari

```
100 REM -----  
110 N$ = "PROGRAMMI FINANZIARI"  
120 REM -----  
130 GOSUB 7500 : REM INTESAAZIONE  
140 REM COSTITUZIONE MATRICE MENU'  
150 X$(1) = "VALORE FINALE DI UN DEPOSITO"  
160 X$(2) = "VALORE FINALE DI UNA SERIE"  
170 X$(3) = "VALORE ATTUALE DI UNA SOMMA"  
180 X$(4) = "VALORE ATTUALE DI UNA SERIE"  
190 X$(5) = "PAGAMENTI PER DISPORRE DI UNA SOMMA FUTURA"  
200 X$(6) = "PRELIEVI DI CAPITALE"  
210 X$(7) = "DETERMINAZIONE DEL TASSO DI INTERESSE"  
220 X$(8) = "VALORE ATTUALE NETTO, FLUSSI NON COSTANTI"  
230 X$(9) = "TEMPO DI RADDOPPIO"  
240 X$(10) = "TASSO DI INTERESSE EQUIVALENTE"  
250 REM MOSTRA MENU'  
260 N = 10: GOSUB 8500  
270 ON X GOSUB 300, 530, 760, 970, 1200, 1420, 1700, 1910, 2300, 2490  
280 GOTO 100
```

Figura 2.22: Listato di programma: Menù dei programmi finanziari

PROGRAMMI COMMERCIALI PRATICI

Questo capitolo presenta dieci utili programmi commerciali che consentono di stendere piani di ammortamento, determinare le scorte ottimali e risolvere molti altri problemi commerciali.

Per esempio, può darsi che siate per intraprendere una nuova attività o che abbiate organizzato una manifestazione di beneficenza; il programma del punto di equilibrio calcolerà quanti pezzi o quanti biglietti dovete vendere per ottenere un profitto. Sia che siate un venditore o che ne abbiate uno alle vostre dipendenze, il programma per il calcolo delle commissioni di vendita vi permette di determinare la commissione corrispondente a un determinato volume di vendite. Il programma “Stipendio con straordinari” rende semplice il calcolo dello stipendio, straordinari compresi. C'è anche un programma per prendere decisioni esecutive, da utilizzare quando tutto il resto non ha funzionato.

Ricordatevi di memorizzare questi programmi via via che li inserite nel computer e di unirli alle subroutine comuni prima di lanciarli.

AMMORTAMENTO A RATE COSTANTI

Descrizione

Questo programma calcola un piano di ammortamento annuale a rate costanti. Per ammortizzare un bene, ne va specificato il valore attuale, il valore finale previsto e la sua vita in anni. Se il bene ha una vita superiore ai dieci anni, il programma mostra il piano per i primi dieci anni, poi fa una pausa. Quando siete pronti per esaminare gli anni da 11 a 20, premete un tasto qualsiasi e il piano viene mostrato.

Esempio

Stendiamo un piano di ammortamento per una copiatrice. Il suo valore attuale è di 5.000.000 e pensiamo di usarla per 6 anni e poi rivenderla per 1.000.000. La Figura 3.1 mostra i risultati dell'esempio e la Figura 3.2 un listato del programma.

AMMORTAMENTO A RATE COSTANTI

VALORE INIZIALE 5000000
 VALORE FINALE PREVISTO 1000000
 VITA (ANNI) 6

ANNO	AMMORT.	VALORE
1	666.67	4333.33
2	666.67	3666.66
3	666.67	2999.99
4	666.67	2333.32
5	666.67	1666.65
6	666.67	999.98

 VOLETE
 CALCOLARE
 DI NUOVO? (S O N)

Figura 3.1: Videata: Ammortamento a rate costanti

```

310 REM -----
320 N$ = "AMMORTAMENTO A RATE COSTANTI"
330 REM -----
340 K = 1
350 GOSUB 7500: REM INTESTAZIONE
360 REM ACQUISIZIONE PARAMETRI DI INPUT
370 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
380 Q3$ = "VALORE INIZIALE" : GOSUB 8000
390 Q3$ = "VALORE FINALE PREVISTO" : GOSUB 8000
400 Q3$ = "VITA (ANNI)" : GOSUB 8000
410 REM STAMPA PIANO DI AMMORTAMENTO
420 D = (PAR(1) - PAR(2)) / PAR(3)
430 D = INT (D * 100 + .5) / 100
440 PRINT
450 PRINT "ANNO AMMORT. VALORE"
460 FOR A = 1 TO PAR(3)
470 VTAB A - INT ((A - 1) / 10) * 10 + 8
480 HTAB 1: PRINT SPC( 35): HTAB 1
490 VTAB A - INT ((A - 1) / 10) * 10 + 8
500 PRINT " ";A; TAB( 9);D; TAB( 23); PAR(1) - D * A
510 IF A / 10 = INT (A / 10) THEN GOSUB 9000
520 NEXT A
530 REM RICHIESTA SE ANCORA
540 Q2$ = "CALCOLARE" : GOSUB 4500
550 IF YN$ = "N" THEN END
560 GOTO 310
    
```

Figura 3.2: Listato di programma: Ammortamento a rate costanti

AMMORTAMENTO A RATE VARIABILI

Descrizione

Questo programma consente di stendere un particolare piano di ammortamento a rate variabili. I piani di questo genere sono accelerati e spesso offrono vantaggi fiscali rispetto al metodo a rate costanti. Con questo metodo il valore contabile residuo di un bene viene moltiplicato per un fattore di accelerazione e il risultato viene diviso per la vita del bene in anni. Occorre fornire al programma il valore contabile iniziale del bene, la sua vita utile e il fattore di accelerazione. Fattori di accelerazione tipici sono 150%, 175% e 200%; questo metodo di ammortamento non considera il valore finale del bene.

Esempio

Ammortizziamo la copiatrice dell'esempio precedente con questo metodo, usando un fattore di accelerazione del 150%. Il piano di ammortamento compare nella Figura 3.3 e il listato del programma nella Figura 3.4.

AMMORTAMENTO A RATE VARIABILI
 VALORE INIZIALE 5000000
 VITA (ANNI) 6
 % AMMORT. ACCELER. 150

ANNO	AMMORT.	VALORE
1	1250	3750
2	937.5	2812.5
3	703.13	2109.37
4	527.34	1582.03
5	395.51	1186.52
6	296.63	889.89

 VOLETE
 CALCOLARE
 DI NUOVO? (S O N)

Figura 3.3: Videata: Ammortamento a rate variabili

```

570 REM -----
580 N$ = "AMMORTAMENTO A RATE VARIABILI"
590 REM -----
600 K = 1: GOSUB 7500
610 REM ACQUISIZIONE PARAMETRI DI INPUT
620 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
630 Q3$ = "VALORE INIZIALE" : GOSUB 8000
640 Q3$ = "VITA (ANNI) : GOSUB 8000
650 Q3$ = "% AMMORT. ACCELER." : GOSUB 8000
660 REM STAMPA PIANO DI AMMORTAMENTO
670 RV = PAR(1)
680 PRINT
690 PRINT "ANNO AMMORT. VALORE"
700 FOR A = 1 TO PAR(2)
710 D = RV * PAR(3) / 100 / PAR(2)
720 D = INT (D * 100 + .5)/100
730 VTAB A - INT ((A - 1) / 10) * 10 + 8
740 HTAB 1: PRINT SPC( 35): HTAB 1
750 VTAB A - INT ((A - 1) / 10) * 10 + 8
760 PRINT " ";A; TAB( 10);D; TAB( 23); RV - D
770 RV = RV - D
780 RV = INT (RV * 100 + .5) / 100
790 IF A / 10 = INT (A / 10) THEN GOSUB 9000
800 NEXT A
810 REM RICHIESTA SE ANCORA
820 Q2$ = "CALCOLARE" : GOSUB 4500
830 IF YN$ = "N" THEN END
840 GOTO 570

```

Figura 3.4: Listato di programma: Ammortamento a rate variabili

SECONDO METODO DI AMMORTAMENTO A RATE VARIABILI

Descrizione

Mostriamo adesso un secondo metodo di ammortamento accelerato, molto diffuso negli Stati Uniti: il valore contabile residuo viene moltiplicato per il rapporto fra il numero degli anni di vita rimanenti e la somma degli anni di vita. Se la vita del bene è di cinque anni, la somma degli anni di vita è

$$5 + 4 + 3 + 2 + 1 = 15$$

Anche in questo caso va specificato il valore contabile attuale, il valore finale e la vita del bene. Il computer semplifica questo calcolo, altrimenti noioso.

Esempio

Stendiamo un piano di ammortamento di questo tipo per la copiatrice degli esempi precedenti. Il piano compare nella Figura 3.5 e il listato del programma nella Figura 3.6.

SECONDO METODO AMMORT. A RATE VARIABILI	
VALORE CONTABILE ATTUALE	5000000
VALORE FINALE PREVISTO	1000000
VITA (ANNI)	6

ANNO	AMMORT.	VALORE
1	1142.86	3857.14
2	952.38	2904.76
3	761.9	2142.86
4	571.43	1571.43
5	380.95	1190.48
6	190.48	1000

 VOLETE
 CALCOLARE
 DI NUOVO? (S O N)

Figura 3.5: Videata: Secondo metodo ammort. a rate variabili

```

850  REM -----
860  N$ = "SECONDO METODO AMMORT. A RATE VARIABILI"
870  REM -----
880  GOSUB 7500: K = 1
890  REM ACQUISIZIONE PARAMETRI DI INPUT
900  Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
910  Q3$ = "VALORE CONTABILE ATTUALE" : GOSUB 8000
920  Q3$ = "VALORE FINALE PREVISTO" : GOSUB 8000
930  Q3$ = "VITA (ANNI)" : GOSUB 8000
940  REM STAMPA PIANO DI AMMORTAMENTO
950  D = PAR(1) - PAR(2)
960  RV = PAR(1)
970  PRINT
980  PRINT "ANNO AMMORT. VALORE"
990  FOR A = 1 TO PAR(3)
1000 B = (PAR(3) + 1) / 2
1010 D1 = D * (PAR(3) + 1 - A) / (PAR(3) * B)
1020 D1 = INT (D1 * 100 + .5) / 100
1030 VTAB A - INT ((A - 1) / 10) * 10 + 8
1040 HTAB 1: PRINT SPC( 38): HTAB 1
1050 VTAB A - INT ((A - 1) / 10) * 10 + 8
1060 PRINT " ";A; TAB( 10);D1; TAB( 23);RV - D1
1070 RV = RV - D1
1080 IF A / 10 = INT (A / 10) THEN GOSUB 9000
1090 NEXT A
1100 REM RICHIESTA SE ANCORA
1110 Q2$ = "CALCOLARE" : GOSUB 4500
1120 IF YN$ = "N" THEN END
1130 GOTO 850

```

Figura 3.6: Listato di programma: Secondo metodo di ammort. a rate variabili

PUNTO DI EQUILIBRIO

Descrizione

Questo programma consente di trovare i quantitativi da vendere perchè un'attività commerciale o produttiva vada in pareggio. Bisogna fornire i costi fissi dell'attività, il costo di produzione o di acquisto unitario del prodotto e il prezzo di vendita unitario.

Esempio

Marco compra degli articoli da regalo a 3000 lire l'uno e pensa di rivenderli per corrispondenza a 6000 lire. Se Marco ritiene che le spese di pubblicità e di vendita siano di 1.500.000, quanti pezzi deve vendere per far pari? La risposta a questo esempio compare nella Figura 3.7; il listato del programma nella Figura 3.8.

PUNTO DI EQUILIBRIO

COSTI FISSI	1500000
COSTO UNITARIO	3000
PREZZO DI VENDITA	6000

PUNTO DI EQUILIBRIO -> 500 UNITA'

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 3.7: Videata: Punto di equilibrio

```
1140 REM -----  
1150 N$ = "PUNTO DI EQUILIBRIO"  
1160 REM -----  
1170 GOSUB 7500:K = 1  
1180 REM ACQUISIZIONE PARAMETRI DI INPUT  
1190 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""  
1200 Q3$ = "COSTI FISSI" : GOSUB 8000  
1210 Q3$ = "COSTO UNITARIO" : GOSUB 8000  
1220 Q3$ = "PREZZO DI VENDITA" : GOSUB 8000  
1230 PRINT  
1240 US = INT (PAR(1) / (PAR(3) - PAR(2)) * 100 + .5) / 100  
1250 PRINT "PUNTO DI EQUILIBRIO -( ";US;" UNITA"  
1260 REM RICHIESTA SE ANCORA  
1270 Q2$ = "CALCOLARE" : GOSUB 4500  
1280 IF YN$ = "N" THEN END  
1290 GOTO 1140
```

Figura 3.8: Listato di programma: punto di equilibrio

DETERMINAZIONE DELLE SCORTE OTTIMALI

Descrizione

I ricercatori operativi hanno messo a punto un metodo per determinare il quantitativo ottimale delle scorte di un produttore. Questo programma effettua i calcoli necessari. Per utilizzarlo, vanno specificati il costo di acquisto, il numero di unità utilizzate annualmente e il costo unitario di giacenza. Il costo di giacenza viene definito come il tasso di interesse al quale viene remunerato il capitale moltiplicato per il costo di acquisto unitario.

Esempio

Roberto è il direttore di produzione di una piccola società manifatturiera. Vuole calcolare le scorte ottimali delle pompe utilizzate nei prodotti della società. La società utilizza 15.000 pompe all'anno. Il loro costo annuale di giacenza è di 5000 lire l'una e il costo di acquisto è di 75.000 lire. Qual'è la scorta ottimale? La Figura 3.9 mostra il risultato dell'analisi di Roberto e la Figura 3.10 il listato del programma.

SCORTE OTTIMALE

COSTO DI ACQUISTO	75000
UNITA' UTILIZZATE ANNUA	15000
COSTO ANNUALE DI GIACENZA	5000

SC. OTT. = 670 UNITA'

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 3.9: Videata: Scorte ottimali

```
1300 REM -----
1310 N$ = "SCORTE OTTIMALI"
1320 REM -----
1330 GOSUB 7500 : K = 1
1340 REM ACQUISIZIONE PARAMETRI DI INPUT
1350 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
1360 Q3$ = "COSTO DI ACQUISTO" : GOSUB 8000
1370 Q3$ = "UNITA' UTILIZZATE ANNUAL." : GOSUB 8000
1380 Q3$ = "COSTO ANNUALE DI GIACENZA" : GOSUB 8000
1390 REM CALCOLARE E STAMPARE
1400 PRINT
1410 EQ = INT ( SQR ( 2 * PAR(1) * PAR(2) / PAR(3)))
1420 PRINT " SC. OTT. = ";EQ;" UNITA'"
1430 REM RICHIESTA SE ANCORA
1440 Q2$ = "CALCOLARE" : GOSUB 4500
1450 IF YN$ = "N" THEN END
1460 GOTO 1300
```

Figura 3.10: Listato di programma: Scorte ottimali

PREZZO DI VENDITA SCONTATO

Descrizione

Questo programma permette di calcolare il prezzo di vendita lordo, con IVA, di un bene che viene venduto scontato. Per usare questo programma va indicato il prezzo al dettaglio, la percentuale di sconto e la percentuale dell'IVA.

Esempio

Beatrice vuol comprare una televisione che viene venduta con lo sconto del 15%. Il prezzo originario netto è di 350.000 lire e l'IVA è del 18%. Quanto dovrà pagare Beatrice? La Figura 3.11 mostra i risultati di questo esempio e la Figura 3.12 il listato del programma.

PREZZO DI VENDITA SCONTATO

PREZZO NETTO	350000
SCONTO (%)	15
IVA (%)	18

PREZZO	350000
SCONTO	52500
IVA	53550

TOTALE	351000
--------	--------

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 3.11: Videata: Prezzo di vendita scontato

```
1470 REM -----
1480 N$ = "PREZZO DI VENDITA SCONTATO"
1490 REM -----
1500 GOSUB 7500:K = 1
1510 REM ACQUISIZIONE DEI PARAMETRI DI INPUT
1520 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
1530 Q3$ = "PREZZO NETTO" : GOSUB 8000
1540 Q3$ = "SCONTO (%)" : GOSUB 8000
1550 Q3$ = "IVA (%)" : GOSUB 8000
1560 REM CALCOLARE E STAMPARE
1570 PRINT
1580 PRINT " PREZZO"; TAB(25); PAR(1)
1590 D = PAR(1) * PAR(2) / 100
1600 D = INT (D * 100 + .5) / 100
1610 PRINT " SCONTO"; TAB(25); D
1620 T = (PAR(1) - D) * PAR(3) / 100
1630 T = INT (T * 100 + .5) / 100
1640 PRINT " IVA"; TAB(25); T
1650 PRINT
1660 PRINT "TOTALE"; TAB(25); PAR(1)-D+T
1670 REM RICHIESTA SE ANCORA
1680 Q2$ = "CALCOLARE": GOSUB 4500
1690 IF YN$ = "N" THEN END
1700 GOTO 1470
```

Figura 3.12: Listato di programma: Prezzo di vendita scontato

MEDIA PONDERATA

Descrizione

Molte sono le occasioni in cui si vuol calcolare una media ponderata di una serie di numeri. Le medie ponderate possono essere utilizzate per trovare il costo medio di azioni acquistate in più riprese a prezzi diversi. Le medie ponderate rappresentano anche un modo per trovare il valore di un inventario quando un bene è stato acquistato a prezzi diversi.

Questo programma calcola rapidamente la media ponderata di una serie di valori. Bisogna fornire il costo unitario seguito dalle unità acquistate. L'inserimento dei dati si chiude battendo 0 sia per il valore unitario che per il numero di unità.

Esempio

Luigi ha comprato azioni di una società elettronica che si è sviluppata rapidamente. Vorrebbe calcolare il costo medio delle sue azioni; ne ha comprate quattro a 10.000 lire l'una, quattro a 40.000 lire e due a 60.000. Le Figure 3.13 e 3.14 mostrano il risultato dei calcoli e il listato del programma.

```
1710 REM -----
1720 N$ = "MEDIA PONDERATA"
1730 REM -----
1740 DIM D(100,1)
1750 REM ACQUISIZIONE PARAMETRI DI INPUT
1760 GOSUB 7500:K = 1
1770 Q1$ = "INSERIMENTO DATI: 0 PER
1780 Q2$ = "" : Q3$ = "" : GOSUB 6000
1790 ND = 0
1800 VTAB 22: HTAB 1: PRINT SPC( 78)
1810 VTAB 22 : HTAB 1 : INPUT "VALORE UNITARIO? ";D(ND,0)
```

Figura 3.14: Listato di programma: Media ponderata

(continua)

MEDIA PONDERATA

VALORE UNITARIO:	10000	UNITA':	4
VALORE UNITARIO:	40000	UNITA':	4
VALORE UNITARIO:	60000	UNITA':	2

LA MEDIA PONDERATA È: 32000

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 3.13: Videata: Media ponderata

```
1820 INPUT "UNITA'? ";D(ND,1)
1830 IF D(ND,1) = 0 THEN 1920
1840 VTAB ND - INT (ND / 10) * 10 + 6
1850 PRINT SPC(39)
1860 VTAB ND - INT (ND / 10) * 10 + 6: HTAB 1
1870 PRINT SPC(38): HTAB 1
1880 VTAB ND - INT (ND / 10) * 10 + 6
1890 PRINT " VALORE UNITARIO: ";D(ND,0)
1900 PRINT TAB(25); "UNITA': ";D(ND,1)
1910 ND = ND + 1: GOTO 1800
1920 REM ADESSO CALCOLARE E STAMPARE
1930 AVE = 0: U = 0
1940 FOR I = 0 TO ND - 1
1950 AVE = AVE + D(I,0) * D(I,1)
1960 U = U + D(I,1)
1970 NEXT I
1980 AVE = AVE / U
1990 VTAB 17: HTAB 3
2000 PRINT "LA MEDIA PONDERATA È: ";AVE
2010 REM RICHIESTA SE ANCORA
2020 Q2$ = "CALCOLARE" : GOSUB 4500
2030 IF YN$ = "N" THEN END
2040 GOTO 1760
```

Figura 3.14: Listato di programma: Media ponderata

COMMISSIONE DI VENDITA

Descrizione

Il programma che verrà inserito adesso calcola la commissione da corrispondere a un venditore. Vanno indicati il minimo garantito mensile, il volume di vendita mensile e la percentuale della commissione. Un risultato negativo indica che il venditore non ha raggiunto la propria quota.

Esempio

Tommaso ha un minimo garantito mensile di 1.200.000 lire e una commissione del 6% sulle vendite di mobilio che effettua. Quanto deve ricevere in un mese in cui ha venduto per 25 milioni? Le Figura 3.15 e 3.16 contengono il risultato dell'esempio e il listato del programma.

COMMISSIONE DI VENDITA

MINIMO GARANTITO	1200000
VENDITE MENSILI	25000000
COMMISSIONE (%)	6
COMMISSIONE MATURATA:	1500000
DOVUTA (MENO MIN. GAR.):	300000

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 3.15: Videata: Commissione di vendita

```
2050 REM -----
2060 N$ = "COMMISSIONE DI VENDITA"
2070 REM -----
2080 GOSUB 7500:K = 1
2090 REM ACQUISIZIONE PARAMETRI DI INPUT
2100 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
2110 Q3$ = "MINIMO GARANTITO" : GOSUB 8000
2120 Q3$ = "VENDITE MENSILI" : GOSUB 8000
2130 Q3$ = "COMMISSIONE (%)" : GOSUB 8000
2140 REM CALCOLARE E STAMPARE
2150 PRINT
2160 PRINT "COMMISSIONE MATURATA: "; TAB(25)
2170 PRINT INT (PAR(2) * PAR(3) + .5) / 100
2180 PRINT "DUE (LESS DRAW): "; TAB( 25);
2190 D = PAR(2) * PAR(3) / 100 - PAR(1)
2200 PRINT INT (D * 100 + .5) / 100
2210 REM RICHIESTA SE ANCORA
2220 Q2$ = "CALCOLARE" : GOSUB 4500
2230 IF YN$ = "N" THEN END
2240 GOTO 2050
```

Figura 3.16: Listato di programma: Commissione di vendita

STIPENDIO CON STRAORDINARIO

Descrizione

Questo programma facilita il calcolo dello stipendio totale dovuto a un dipendente che effettui anche ore straordinarie a un compenso naturalmente superiore. Va inserita la paga oraria di base, il fattore per cui va moltiplicata per trovare quella dello straordinario e il numero di ore normali e di straordinario lavorate dal dipendente.

Esempio

Antonia guadagna 10.000 lire all'ora come tecnica specializzata. In una settimana particolarmente intensa lavora le 40 ore regolari più 12 di straordinario. Volete calcolare il suo stipendio totale, sapendo che lo straordinario viene pagato una volta e mezzo la paga normale. Le Figure 3.17 e 3.18 mostrano il risultato dell'esempio e il listato del programma.

STIPENDIO CON STRAORDINARIO

PAGA ORARIA BASE	10000
FATTORE DI STRAORDINARIO	1.5
ORE SETTIMANA LAVORATIVA	40
ORE DI STRAORDINARIO	12
STIPENDIO SETTIM. BASE	400000
STRAORDINARIO	180000
STIPENDIO TOTALE	580000

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 3.17: Videata: Stipendio con straordinario

```
2250 REM -----
2260 N$ = "STIPENDIO CON STRAORDINARIO"
2270 REM -----
2280 GOSUB 7500:K = 1
2290 REM ACQUISIZIONE PARAMETRI DI INPUT
2300 Q1$ = "INSERIMENTO DATI:" : Q2$ = ""
2310 Q3$ = "PAGA ORARIA BASE" : GOSUB 8000
2320 Q3$ = "FATTORE DI STRAORDINARIO" : GOSUB 8000
2330 Q3$ = "ORE SETTIMANA LAVORATIVA" : GOSUB 8000
2340 Q3$ = "ORE DI STRAORDINARIO" : GOSUB 8000
2350 REM CALCOLARE E STAMPARE
2360 PRINT
2370 D = INT (PAR(1) * PAR(3) * 100 + . 5) / 100
2380 PRINT " STIPENDIO SETTIM. BASE"; TAB(25); D
2390 D = PAR(1) * PAR(2) * PAR(4)
2400 D = INT (D * 100 + .5) / 100
2410 PRINT " STRAORDINARIO"; TAB(25); D
2420 D = PAR(1) * PAR(3) + PAR(1) * PAR(2) * PAR(4)
2430 D = INT (D * 100 + .5) / 100
2440 PRINT " STIPENDIO TOTALE"; TAB(25); D
2450 REM RIHIESTA SE ANCORA
2460 Q2$ = "CALCOLARE" : GOSUB 4500
2470 IF YN$ = "N" THEN END
2480 GOTO 2250
```

Figura 3.18: Listato di programma: Stipendio con straordinario

STRUMENTO PER PRENDERE DECISIONI ESECUTIVE

Descrizione

Quando tutto il resto non ha funzionato e dovete prendere una decisione esecutiva, potete farvi aiutare da questo programma. Battete semplicemente la domanda nel computer: il programma fornisce la risposta secondo un'avanzata teoria decisionale. (È un programma divertente a una festa o in qualche altra occasione sociale.)

Esempio

Edoardo vuol sapere se un rischioso progetto avrà successo. Poichè i suoi collaboratori non riescono a dargli un consiglio definitivo, pone la domanda al computer, come mostra la Figura 3.19. Il listato del programma è nella Figura 3.20.

STRUMENTO PER DECISIONI ESECUTIVE

PER FAVORE FATE LA DOMANDA

=> RIUSCIRA' QUESTO PROGETTO?
SENZA ALCUN DUBBIO

VOLETE
DOMANDARE
DI NUOVO? (S O N)

Figura 3.19: Videata: Strumento per decisioni esecutive

```
2490 REM -----  
2500 N$ = "STRUMENTO PER DECISIONI ESECUTIVE"  
2510 REM -----  
2520 GOSUB 7500: K = 1  
2530 PRINT "PER FAVORE FATE LA DOMANDA  
2540 PRINT  
2550 INPUT "=>";Q$:Y = LEN (Q$)  
2560 IF Y < 6 THEN 2580  
2570 Y = INT (Y / 2): GOTO 2560  
2580 ON Y GOTO 2590, 2600, 2610, 2620, 2630  
2590 PRINT "ASSOLUTAMENTE NO" : GOTO 2640  
2600 PRINT "FORZA, VAI" : TO 2640  
2610 PRINT "SENZA ALCUN DUBBIO" : GOTO 2640  
2620 PRINT "FAI A TESTA O CROCE" : GOTO 2640  
2630 PRINT "TIRATI INDIETRO - PRESTO"  
2640 REM RICHIESTA SE ANCORA  
2650 Q2$ = "DOMANDARE" : GOSUB 4500  
2660 IF YN$ = "N" THEN END  
2670 GOTO 2490
```

Figura 3.20: Listato di programma: Strumento per decisioni esecutive

MENU' DEI PROGRAMMI COMMERCIALI

Se memorizzate tutti i programmi di questo capitolo in un file su dischetto, potete aggiungere il seguente programma di menù che permette di lanciare gli altri. I programmi vengono scelti tramite i tasti numerici del computer. Ricordate di sostituire le istruzioni **END** di ciascun programma con istruzioni **RETURN**. Prima di costruire questo programma, riguardate nel Capitolo 2 le istruzioni di inserimento dei menù. Le Figure 3.21 e 3.22 mostrano il menù e il relativo programma.

1= AMMORTAMENTO A RATE COSTANTI
 2= AMMORTAMENTO A RATE VARIABILI
 3= SECONDO AMMORTAMENTO A RATE VARIABILI
 4= PUNTO DI EQUILIBRIO
 5= SCORTE OTTIMALI
 6= PREZZO DI VENDITA SCONTATO
 7= MEDIA PONDERATA
 8= COMMISSIONE DI VENDITA
 9= STIPENDIO CON STRAORDINARIO
 0= STRUMENTO PER DECISIONI ESECUTIVE

 PROGRAMMA SCELTO:

Figura 3.21: Videata: Menù dei programmi commerciali

```

100 REM -----
110 N$ = "PROGRAMMI COMMERCIALI"
120 REM -----
130 GOSUB 7500 : REM INTESTAZIONE
140 FOR X = 1 TO 1000: NEXT
150 REM CREAZIONE MATRICE DEL MENU'
160 X$(1) = "AMMORTAMENTO A RATE COSTANTI"
170 X$(2) = "AMMORTAMENTO A RATE VARIABILI"
180 X$(3) = "SECONDO AMMORTAMENTO A RATE VARIABILI"
190 X$(4) = "PUNTO DI EQUILIBRIO"
200 X$(5) = "SCORTE OTTIMALI"
210 X$(6) = "PREZZO DI VENDITA SCONTATO"
220 X$(7) = "MEDIA PONDERATA"
230 X$(8) = "COMMISSIONE DI VENDITA"
240 X$(9) = "STIPENDIO CON STRAORDINARIO"
250 X$(10) = "STRUMENTO PER DECISIONI ESECUTIVE"
260 REM MOSTRA MENU'
270 N = 10: GOSUB 8500
280 ON X GOSUB 310, 570, 850, 1140, 1300, 1470, 1710, 2050, 2250, 2490
290 GOTO 100
  
```

Figura 3.22: Listato di programma: Menù dei programmi commerciali

PROGRAMMI IMMOBILIARI

In questo capitolo vengono presentati otto utili programmi immobiliari, che consentono di stendere piani di pagamento di mutui e di effettuare molti altri pratici calcoli.

Due programmi sono particolarmente interessanti. Quello del “Prezzo accessibile di una casa” permette di trovare il prezzo di una casa che una banca è disposta a finanziare; è utile sia per determinare la gamma di prezzo che vi potete permettere che per le contrattazioni, una volta deciso di effettuare l'acquisto. Particolarmente illuminante è anche il programma dei “Pagamenti accelerati”. Un piccolo aumento della rata mensile può ridurre drasticamente l'interesse totale che dovete pagare su un mutuo. Date un'occhiata alla Figura 4.8 e vedrete cosa vogliamo dire! Come negli altri capitoli, c'è anche un programma di selezione tramite menù che potete utilizzare per creare un gruppo di programmi immobiliari.

SUBROUTINE IMMOBILIARI

I programmi di questo capitolo utilizzano tre subroutine che dovrete scrivere subito e memorizzare su disco. Quando siete pronti per battere un programma, caricate queste subroutine, poi inserite il programma e memorizzate subroutine e programma insieme in un nuovo file. (Riunite SUBLIB e questo file prima di lanciare il programma.) Conviene inserire tutti i programmi del capitolo sullo stesso file su disco insieme a queste subroutine. Le subroutine comuni compaiono nella Figura 4.1. Vengono utilizzate per l'input di un calcolo di pagamenti, per il calcolo di un pagamento e per determinare quanto resta da pagare su un mutuo.

```

260 REM -----
270 REM ROUTINE PER CALCOLO RATA
280 REM -----
290 REM SIGNIFICATO PARAMETRI:
300 REM IN=TASSO INTERESSE ANNUO
310 REM YR=NUMERO DI ANNI
320 REM AM=ENTITA' DEL MUTUO
330 REM
340 N1 = 12 * YR: I1 = IN / 100 / 12: V = 1 / (1 + I1)
350 P = AM * I1 / (1 - V ^ N1)
360 RETURN
370 REM -----
380 REM INPUT PER CALCOLO RATA
390 REM -----
400 Q1$ = "INDICARE PARAMETRI"
410 Q2$ = ""
420 Q3$ = "ENTITA' DEL MUTUO" : GOSUB 8000
430 Q3$ = "TASSO INTERESSE ANNUO (%)" : GOSUB 8000
440 Q3$ = "DURATA DEL MUTUO (ANNI)" : GOSUB 8000
450 AM = PAR(1)
460 IN = PAR(2)
470 YR = PAR(3)
480 RETURN
490 REM -----
500 REM CALCOLO DEL SALDO
510 REM -----
520 REM SIGNIFICATO PARAMETRI:
530 REM P=RATA
540 REM IN=TASSO INTERESSE ANNUO
550 REM N=NUMERO DELLA RATA
560 REM AM=ENTITA' DEL MUTUO
570 FOR I = 1 TO N
580 AM = AM - P + IN / 12 / 100 * AM
590 NEXT I
600 RETURN

```

Figura 4.1: Listato di programma: Subroutine Immobiliari

CALCOLO DELLA RATA MENSILE

Descrizione

Questo programma calcola la rata mensile necessaria per estinguere completamente un mutuo. Dovete fornire l'ammontare del mutuo, il tasso di interesse annuo e la durata del mutuo in anni.

Esempio

Daniele e Diana possono comprare una casa per 100 milioni. Intendono pagare 25 milioni in contanti e ottenere un mutuo per il rimanente al 12% di interesse, della durata di 30 anni. Trovare la rata mensile. Le Figure 4.2 e 4.3 mostrano il risultato dell'esempio e il listato del programma.

RATA MENSILE

ENTITA' DEL MUTUO	75000000
TASSO INTERESSE ANNUO (%)	12
DURATA DEL MUTUO (ANNI)	30
LA RATA MENSILE È	771460

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 4.2: Videata: Calcolo della rata mensile

```
610 REM -----  
620 N$ = "RATA MENSILE"  
630 REM -----  
640 GOSUB 7500 : REM INTESTAZIONE  
650 K = 1  
660 GOSUB 370 : REM PARAMETRI  
670 GOSUB 270 : REM CALCOLO RATA  
680 PRINT  
690 PRINT "LA RATA MENSILE È"; TAB (25)  
700 P = INT (P * 100 + .5) / 100  
710 PRINT P  
720 REM RICHIESTA SE ANCORA  
730 Q2$ = "CALCOLARE"  
740 GOSUB 4500  
750 IF YN$ = "N" THEN END  
760 GOTO 610
```

Figura 4.3: Listato di programma: Calcolo della rata mensile

PIANO DI ESTINZIONE DI UN MUTUO

Descrizione

Questo programma mostra sullo schermo un piano di estinzione di un mutuo per qualsiasi anno prescelto. Si inserisce l'entità del mutuo, il tasso di interesse annuo, il termine del mutuo e l'anno per il quale si vuol vedere il piano di estinzione, e si ottiene la quota di capitale e gli interessi che vengono pagati ogni anno. L'interesse pagato è spesso deducibile dal reddito.

Esempio

Marco e Paola hanno un mutuo sulla casa di 60 milioni. Il tasso di interesse annuo è dell'11,5% e il termine è 25 anni. Marco e Paola vorrebbero conoscere la quota di capitale e gli interessi che pagheranno nel sesto anno. La Figura 4.4 mostra il risultato di questo esempio. Il listato del programma è nella Figura 4.5.

```
770 REM -----  
780 N$ = "PIANO DI ESTINZIONE"  
790 REM -----  
800 K = 1  
810 GOSUB 7500 : REM INTESTAZIONE  
820 GOSUB 370 : REM PARAMETRI  
830 REM INSERIMENTO ANNO  
840 Q3$ = "ANNO"  
850 GOSUB 8000:N = PAR(4)  
860 GOSUB 270 : REM CALCOLO RATA  
870 REM CALCOLO PRIMO MESE
```

Figura 4.5: Listato di programma: Piano di estinzione di un mutuo

(continua)

PIANO DI ESTINZIONE

ENTITA' DEL MUTUO	60000000
TASSO INTERESSE ANNUO (%)	11.5
DURATA DEL MUTUO (ANNI)	25
ANNO	6

MESE	CAPITALE	INTERESSE	SALDO
61	61.82	548.06	57127.27
62	62.41	547.47	57064.86
63	63.01	546.87	57001.85
64	63.61	546.27	56938.24
65	64.22	545.66	56874.02
66	64.84	545.04	56809.18
67	65.46	544.42	56743.72
68	66.09	543.79	56677.63
69	66.72	543.16	56610.91
70	67.36	542.52	56543.55
71	68	541.88	56475.55
72	68.66	541.22	56406.89

 VOLETE
 CALCOLARE
 DI NUOVO? (S O N)

Figura 4.4: Videata: Piano di estinzione di un mutuo

```

880 N = 12 * N - 12
890 REM SALDO INIZIALE
900 GOSUB 490
910 PRINT "MESE  CAPITALE INTERESSE SALDO"
920 P = INT (100 * P + .5) / 100
930 FOR J = 1 TO 12
940 REM COMPUTE P & I
950 I1 = INT (IN / 12 * AM + .5) / 100:P1 = P - I1
960 AM = AM - P1:P1 = INT (P1 * 100 + .5) / 100
970 AM = INT (AM * 100 + .5) / 100
980 PRINT N + 1; TAB( 9); P1; TAB( 20);I1; TAB( 30);AM
990 N = N + 1
1000 NEXT J
1010 Q2$ = "CALCOLARE" : GOSUB 4500
1020 IF YN$ = "N" THEN END
1030 GOTO 770
    
```

Figura 4.5: Listato di programma: Piano di estinzione di un mutuo

SALDO SU UN MUTUO

Descrizione

Questo programma fornisce il saldo su un mutuo e potete lanciarlo per trovare quanto dovete pagare se decidete di estinguere il mutuo tutto insieme. Il programma assume che tutti i pagamenti precedenti siano stati esattamente quelli dovuti. Per utilizzare il programma vanno specificati l'entità, il tasso di interesse e la durata del mutuo e il numero del pagamento per il quale si vuole determinare il saldo. Non preoccupatevi se il programma impiega un pò di tempo prima di fornire la risposta.

Esempio

Il mutuo di Giovanni è di 90 milioni. Si tratta di un mutuo trentennale al 9%. Vorrebbe calcolare quanto gli rimane da pagare dopo cinque anni (il sessantesimo pagamento). Le Figure 4.6 e 4.7 mostrano il risultato dell'esempio e il listato del programma.

SALDO

ENTITA' DEL MUTUO	90000000
TASSO INTERESSE ANNUO (%)	9
DURATA DEL MUTUO (ANNI)	30
NUMERO DELLA RATA	60
RATA MENSILE	724160
SALDO DOPO 60 RATE	86292120 .-

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 4.6: Videata: Saldo su un mutuo

```
1040 REM -----
1050 N$ = "SALDO"
1060 REM -----
1070 K = 1
1080 GOSUB 7500 : REM INTESTAZIONE
1090 GOSUB 370 : REM PARAMETRI
1100 REM INSERIRE NUMERO DELLA RATA
1110 Q3$ = "NUMERO DELLA RATA"
1120 GOSUB 8000:N = PAR(4)
1130 GOSUB 270 : REM CALCOLO RATA
1140 GOSUB 490 : REM CALCOLO SALDO
1150 PRINT
1160 PRINT "RATA MENSILE"; TAB(23);
1170 P = INT (P * 100 + .5) / 100
1180 PRINT P
1190 AM = INT (AM * 100 + .5) / 100
1200 PRINT "SALDO DOPO "; N; " RATE"; TAB(23); AM
1210 REM RICHIESTA SE ANCORA
1220 Q2$ = "CALCOLARE" : GOSUB 4500
1230 IF YN$ = "N" THEN END
1240 GOTO 1040
```

Figura 4.7: Listato di programma: Saldo su un mutuo

EFFETTO DI PAGAMENTI ACCELERATI

Descrizione

Questo programma vi sarà molto utile: consente di trovare l'effetto di un aumento delle rate mensili di un mutuo. Un aumento delle rate mensili farà diminuire sia il numero dei pagamenti che l'interesse totale. Per utilizzare questo programma, inserire l'importo, il tasso di interesse annuo e la durata del mutuo, l'anno in cui le rate vengono aumentate e l'entità dell'aumento. Il programma ripete i calcoli finché il saldo del mutuo è zero, quindi può darsi che ci voglia un pò prima che compaia la risposta.

Esempio

Laura ha deciso di aumentare di 100.000 lire le rate mensili del mutuo che paga sulla propria casa. Comincia a pagare di più al settimo anno del mutuo e vorrebbe sapere quanto risparmierà sugli interessi. Il suo mutuo è di 60 milioni al 9.75% e la durata è di 30 anni. Si può vedere dalla Figura 4.8 come questo maggiore pagamento farà risparmiare a Laura 41 milioni. La Figura 4.9 mostra il listato del programma.

```
1250 REM -----
1260 N$ = "PAGAMENTI ACCELERATI"
1270 REM -----
1280 K = 1
1290 GOSUB 7500 : REM INTESTAZIONE
1300 GOSUB 370 : REM PARAMETRI
1310 REM INSERIMENTO ANNO
1320 Q3$ = "ANNO DELL'AUMENTO"
1330 GOSUB 8000:N = PAR(4)
1340 GOSUB 270 : REM CALCOLO RATA
1350 REM CALCOLO PRIMO MESE
1360 N = 12 * N - 12
1370 REM INSERIMENTO AUMENTO
1380 Q3$ = "ENTITA' DELL'AUMENTO"
1390 GOSUB 8000:EX = PAR(5)
1400 REM CALC. STAMPA INTERESSI E SALDO
1410 TI = 0: BAL = 0: TP = 0
```

Figura 4.9: Listato di programma: Effetto di pagamenti accelerati

(continua)

PAGAMENTI ACCELERATI

ENTITA' DEL MUTUO	60000000
TASSO INTERESSE ANNUO (%)	9.75
DURATA DEL MUTUO (ANNI)	30
ANNO DELL'AUMENTO	7
ENTITA' DELL'AUMENTO	100000
ANNI RIMANENTI	20.58
TOTALE INTERESSI PAGATI	84432160
INTERESSI RISPARMIATI	41145190

 VOLETE
 CALCOLARE
 DI NUOVO? (S O N)

Figura 4.8: Videata: Effetto di pagamenti accelerati

```

1420 FOR J = 1 TO N
1430 I1 = IN / 12 / 100 * AM:P1 = P - I1:AM = AM - P1
1440 T1 = TI + I1:TP = TP + P1
1450 NEXT J
1460 REM CALCOLO INTER. SENZA AUMENTO
1470 T2 = 0:42 = PAR(1)
1480 FOR J = 1 TO 12 * YR
1490 I2 = IN / 12 / 100 * A2:P2 = P - I2
1500 A2 = A2 - P2:T2 = T2 + I2
1510 NEXT J
1520 REM CALCOLO CON AUMENTO
1530 J = 0
1540 I1 = IN / 12 / 100 * AM:TI = TI + I1
1550 P1 = P + EX - I1:TP = TP + P1
1560 AM = AM - P1:J = J + 1
1570 IF AM > 0 THEN 1540
1580 PRINT
1590 PRINT "ANNI RIMANENTI"; TAB(25)
1600 D = PAR(4) + J / 12 - 1
1610 PRINT INT (D * 100 + .5) / 100
1620 T1 = INT (T1 * 100 + .5) / 100
1630 T2 = INT (T2 * 100 + .5) / 100
1640 PRINT "TOTALE INTERESSI PAGATI"; TAB(25);TI
1650 PRINT "INTERESSI RISPARMIATI"; TAB(25); T2-T1
1660 Q2$ = "CALCOLARE" : GOSUB 4500
1670 IF YN$ = "N" THEN END
1680 GOTO 1250
    
```

Figura 4.9: Listato di programma: Effetto di pagamenti accelerati

CALCOLO DEL DEBITO RESIDUO

Descrizione

Recentemente, molti mutui prevedono il pagamento del debito residuo alla fine di un determinato periodo. Questo programma permette di calcolare l'entità di questo pagamento. Vanno forniti l'importo, il tasso di interesse e la durata del mutuo e l'anno in cui è dovuto il pagamento del residuo.

Esempio

Massimo ha un mutuo di 125 milioni. Il tasso di interesse annuo è dell'11%. I pagamenti sono ammortizzati su 30 anni e Massimo deve saldare il mutuo dopo dieci anni. Qual'è l'entità del pagamento del residuo di Massimo? Le figura 4.10 e 4.11 mostrano il risultato dell'esempio e il listato del programma.

DEBITO RESIDUO

ENTITA' DEL MUTUO	125000000
TASSO INTERESSE ANNUO (%)	11
DURATA DEL MUTUO (ANNI)	30
ANNO SCADENZA	10
RATA MENSILE	1190400
DEBITO RESIDUO DOPO 10 ANNI	115328200

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 4.10: Videata: Calcolo del debito residuo

```
1690 REM -----  
1700 N$ = "DEBITO RESIDUO"  
1710 REM -----  
1720 K = 1  
1730 GOSUB 7500 : REM INTESTAZIONE  
1740 GOSUB 370 : REM PARAMETRI  
1750 REM INSERIMENTO ANNO SCADENZA  
1760 Q3$ = "ANNO SCADENZA"  
1770 GOSUB 8000:N = 12 * PAR(4)  
1780 GOSUB 270: REM CALC PMT  
1790 REM CALCOLO SALDO  
1800 GOSUB 490  
1810 PRINT  
1820 P = INT (P * 100 + .5) / 100  
1830 PRINT "RATA MENSILE"; TAB(23); P  
1840 AM = INT (AM * 100 + .5) / 100  
1850 PRINT  
1860 PRINT "DEBITO RESIDUO"  
1870 PRINT "DOPO "; N/12; " ANNI"; TAB(23); AM  
1880 REM RICHIESTA SE ANCORA  
1890 Q2$ = "CALCOLARE"  
1900 GOSUB 4500  
1910 IF YN$ = "N" THEN END  
1920 GOTO 1690
```

Figura 4.11: Listato di programma: Calcolo del debito residuo

PREZZO ACCESSIBILE DI UNA CASA

Descrizione

Potete usare questo programma per trovare il prezzo di una casa sulla quale la banca è disposta a concedervi un mutuo. Questo programma può essere preziosissimo durante le negoziazioni. Per utilizzarlo, inserite il tasso di interesse annuo e la durata dei mutui disponibili, il vostro reddito annuo, le tasse e le assicurazioni previste, la percentuale massima del vostro reddito lordo che la banca accetterà in pagamento e la percentuale del pagamento in contanti che effettuerete.

Esempio

Gianni e Franca vorrebbero comprare una nuova casa. Possono ottenere un mutuo trentennale al 13.5%; il reddito annuo della coppia è di 50 milioni e pensano di pagare in contanti il 20% del prezzo della casa. Gianni e Franca ritengono di dover pagare 2.400.000 annui in tasse e assicurazioni. La banca ha detto che la rata mensile del mutuo non può essere superiore al 35% del loro reddito. Le Figure 4.12 e 4.13 mostrano il risultato dell'esempio e il listato del programma.

REZZO ACCESSIBILE DI UNA CASA

TASSO INTERESSE ANNUO (%)	13.5
DURATA DEL MUTUO (ANNI)	30
REDDITO ANNUO COMPRATORE	50000000
TASSE E ASSIC. ANNUE PREV.	2400000
% REDDITO PER RATE	35
% PAGAMENTO IN CONTANTI	20

ENTITA' DEL FINANZIAMENTO	109858560
PREZZO ACCESSIBILE	137323200

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 4.12: Videata: Prezzo accessibile di una casa

```

1930 REM -----
1940 N$ = "PREZZO ACCESSIBILE DI UNA CASA"
1950 REM -----
1960 K = 1
1970 GOSUB 7500 : REM INTESTAZIONE
1980 Q1$ = "FORNIRE PARAMETRI:"
1990 Q2$ = ""
2000 Q3$ = "TASSO INTERESSE ANNUO (%)": GOSUB 8000
2010 Q3$ = "DURATA DEL MUTUO (ANNI)": GOSUB 8000
2020 Q3$ = "TASSE E ASSIC. ANNUE PREV.": GOSUB 8000
2030 Q3$ = "% REDDITO PER RATE": GOSUB 8000
2040 Q3$ = "% PAGAMENTO IN CONTANTI": GOSUB 8000
2050 Q3$ = "% DOWN PAYMENT": GOSUB 8000
2060 PRINT
2070 REM CALCOLO RATA CONCESSA
2080 P = PAR(5) / 100 * PAR(3) / 12 - PAR(4) / 12
2090 REM CALCOLO ENTITA' FINANZIAMENTO
2100 IN = PAR(1) / 100 / 12: N1 = 12 * PAR(2)
2110 V = 1 / (1 + IN): AM = P * (1 - V ^ N1) / IN
2120 AM = INT (AM * 100 + .5) / 100
2130 PRINT "ENTITA' DEL FINANZIAMENTO"; TAB(25); AM
2140 D = AM / (1 - PAR(6) / 100)
2150 D = INT (D * 100 + .5) / 100
2160 PRINT N$; TAB( 25); D
2170 Q2$ = "CALCOLARE" : GOSUB 4500
2180 IF YN$ = "N" THEN END
2190 GOTO 1930

```

Figura 4.13: Listato di programma: Prezzo accessibile di una casa

MUTUO CON ALTRO MUTUO

Descrizione

Vedremo adesso un programma che potete usare per trovare i pagamenti da effettuare su una casa che ha due mutui. Per utilizzare questo programma, inserire il prezzo di acquisto della casa, il denaro disponibile per un pagamento in contanti e l'entità del primo mutuo. Vanno anche forniti i tassi di interesse di entrambi i mutui e la durata del primo. Si assume che il secondo mutuo venga ammortizzato mensilmente su una base di soli interessi.

Esempio

Giovanna ha risparmiato 20 milioni per effettuare un pagamento in contanti su una casa. La casa costerà 140 milioni e Giovanna intende ottenere un primo mutuo di 90 milioni. Questo mutuo avrà una durata di 30 anni e un tasso di interesse del 12%. Giovanna otterrà anche un secondo mutuo su base di soli interessi al 16%. Quanto dovrà pagare in totale? I risultati di questo esempio sono nella Figura 4.1 e il listato del programma è nella Figura 4.15.

```
2200 REM -----  
2210 N$ = "MUTUO CON ALTRO MUTUO"  
2220 REM -----  
2230 GOSUB 7500: K = 1  
2240 Q1$ = "FORNIRE PARAMETRI"  
2250 Q2$ = ""  
2260 Q3$ = "PREZZO DI ACQUISTO" : GOSUB 8000  
2270 Q3$ = "CONTANTI DISPONIBILI" : GOSUB 8000  
2280 Q3$ = "PRIMO MUTUO" : GOSUB 8000
```

Figura 4.15: Listato di programma: Mutuo con altro mutuo

(continua)

MUTUO CON ALTRO MUTUO

PREZZO DI ACQUISTO	140000000
CONTANTI DISPONIBILI	20000000
PRIMO MUTUO	90000000
PRIMO TASSO (%)	12
DURATA PRIMO (ANNI)	30
SECONDO TASSO (%)	16

RATA (PRIMO)	925750
RATA (SECONDO)	400000
RATA TOTALE	1325750

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 4.14: Videata: Mutuo con altro mutuo

```
2290 Q3$ = "PRIMO TASSO (%)" : GOSUB 8000
2300 Q3$ = "DURATA PRIMO (ANNI)" : GOSUB 8000
2310 Q3$ = "SECONDO TASSO (%)" : GOSUB 8000
2320 PRINT
2330 AM = PAR(3):IN = PAR(4):YR = PAR(5)
2340 REM CALCOLO RATA PRIMO
2350 GOSUB 270
2360 P = INT (P * 100 + .5) / 100
2370 PRINT "RATA (PRIMO)"; TAB(23); P
2380 REM CALCOLO RATA SECONDO
2390 IF PAR(2) < PAR(1) - PAR(3) THEN 2420
2400 REM PRINT "IL SECONDO MUTUO NON È NECESSARIO"
2410 GOTO 2470
2420 D = PAR(1) - PAR(2) - PAR(3)
2430 P2 = D * PAR(6) / 100 / 12
2440 P2 = INT (P2 * 100 + .5) / 100
2450 PRINT "RATA (SECONDO)"; TAB(23); P2
2460 PRINT "RATA TOTALE"; TAB(23); P+P2
2470 Q2$ = "CALCOLARE" : GOSUB 4500
2480 IF YN$ = "N" THEN END
2490 GOTO 2200
```

Figura 4.15: Listato di programma: Mutuo con altro mutuo

ANALISI DI PROPRIETA' AFFITTATE

Descrizione

Esaminiamo ora un programma particolarmente utile per un'analisi di investimenti immobiliari. Si può usare questo programma per trovare i pagamenti mensili e il flusso di denaro su proprietà affittate. (Non vengono considerati gli effetti di deprezzamento dovuti alle tasse.) Bisogna inserire l'entità di un eventuale mutuo sulla proprietà affittata, il tasso di interesse annuo, la durata del mutuo e le spese annue per assicurazioni, tasse e manutenzione. Va anche fornito l'affitto mensile ricevuto.

Esempio

Sandro pensa di comprare una casa che poi affitterà per 475.000 al mese. Il primo mutuo sarà di 80 milioni, ammortizzato in 30 anni al 12% di interesse. Sandro calcola che l'assicurazione annua sarà di 500.000 lire, le tasse di 1.600.000 e le spese di manutenzione di 400.000. Quale sarà il pagamento mensile che dovrà effettuare e il suo flusso di denaro. Le Figura 4.16 e 4.17 mostrano il risultato dell'esempio e il listato del programma.

ANALISI DI PROPRIETA' AFFITTATE

ENTITA' DEL MUTUO	80000000
TASSO INTERESSE ANNUO (%)	12
DURATA MUTUO (ANNI)	30
ASSICURAZ. ANNUA	500000
TASSE ANNUE	1600000
MANUTENZIONE ANNUA	400000
REDDITO MENSILE	475000

PAGAMENTO MENSILE	822890
FLUSSO MONET. MENSILE	-566220

VOLETE
CALCOLARE
DI NUOVO? (S O N)

Figura 4.16: Videata: Analisi di proprietà affittate

```
2500 REM -----
2510 N$ = "ANALISI DI PROPRIETA' AFFITTATE"
2520 REM -----
2530 GOSUB 7500:K = 1
2540 REM FORNIRE PARAMETRI
2550 GOSUB 370
2560 Q3$ = "ASSICURAZ. ANNUA" : GOSUB 8000
2570 Q3$ = "TASSE ANNUE" : GOSUB 8000
2580 Q3$ = "MANUTENZIONE ANNUA" : GOSUB 8000
2590 Q3$ = "REDDITO MENSILE" : GOSUB 8000
2600 GOSUB 270 : REM CALCOLO PAGAMENTO
2610 PRINT
2620 P = INT ( P * 100 + .5 ) / 100
2630 PRINT "PAGAMENTO MENSILE"; TAB(23); P
2640 CF = PAR(7) - P - (PAR(4) + PAR(5) + PAR(6)) / 12
2650 CF = INT ( CF * 100 + .5 ) / 100
2660 PRINT "FLUSSO MONET.MENSILE"; TAB(23); CF
2670 Q2$ = "CALCOLARE"
2680 IF YN$ = "N" THEN END
2690 GOTO 2500
```

Figura 4.17: Listato di programma: Analisi di proprietà affittate

MENU' DEI PROGRAMMI IMMOBILIARI

Possiamo nuovamente riunire i programmi del capitolo così da avere un package di programmi immobiliari controllato da menù. Si ricordi di sostituire le istruzioni END con istruzioni RETURN. Le spiegazioni sui programma di menù sono nel Capitolo 2. Il menù e il listato del programma compaiono nelle Figura 4.18 e 4.19.

- 1= CALCOLO RATA MENSILE
- 2= PIANO DI AMMORTAMENTO
- 3= SALDO SU UN MUTUO
- 4= EFFETTO DI PAGAMENTI ACCELERATI
- 5= CALCOLO DEL DEBITO RESIDUO
- 6= PREZZO ACCESSIBILE DI UNA CASA
- 7= MUTUO CON ALTRO MUTUO
- 8= ANALISI DI PROPRIETA' AFFITTATE

PROGRAMMA SCELTO:

Figura 4.18: Videata: Menù dei programmi immobiliari

```

100 REM -----
110 N$ = "PROGRAMI IMMGBILIARI"
120 REM -----
130 GOSUB 7500 : REM INTESTAZIONE
140 X$(1) = "CALCOLO RATA MENSILE"
150 X$(2) = "PIANO DI AMMORTAMENTO"
160 X$(3) = "SALDO SU UN MUTUO"
170 X$(4) = "EFFETTO DI PAGAMENTI ACCELERATI"
180 X$(5) = "CALCOLO DEL DEBITO RESIDUO"
190 X$(6) = "PREZZO ACCESSIBILE DI UNA CASA"
200 X$(7) = "MUTUO CON ALTRO MUTUO"
210 X$(8) = "ANALISI DI PROPRIETA' AFFITTATE"
220 REM MOSTRA MENU'
230 N = 8: GOSUB 8500
240 ON X GOSUB 610, 770, 1040, 1250, 1690, 1930, 2200, 2500
250 GOTO 110

```

Figura 4.19: Listato di programma: Menù dei programmi immobiliari

PROGRAMMI DI ANALISI DI DATI

Questo capitolo descrive un gruppo di programmi di analisi per ridurre dati e tracciare grafici di vari tipi. Si tratta di programmi che ricevono un input di dati, ne tracciano il grafico sullo schermo, calcolano medie mobili ed effettuano analisi di regressione lineare. Il programma "Grafico" traccia anche il grafico dei dati che vengono ridotti dagli altri programmi di questo capitolo.

Questi programmi saranno particolarmente utili per chi investe in borsa: si può inserire i prezzi quotidiani delle azioni e poi utilizzare i vari programmi di regolazione. Questi programmi regolarizzano le variazioni quotidiane dei prezzi e danno un'indicazione di tendenza per periodi più lunghi. I programmi di analisi dei dati possono essere utilizzati anche per l'analisi dei dati scientifici, per tracciare il grafico e regolarizzare variazioni di peso e per tenere sotto osservazione dati meteorologici.

Per i programmi di questo capitolo e dei due che seguono, vanno prima inseriti il programma per il menù e le subroutine comuni, che dovrebbero poi essere memorizzati in un file, al quale vengono poi aggiunti i vari programmi come descritto nel Capitolo 1. Dato che i programmi di questi capitoli *iniziano* con il programma del menù, non è necessario modificare alcuna riga per utilizzare il menù.

MENU' PER I PROGRAMMI DI ANALISI DEI DATI

I programmi di analisi dei dati hanno in comune matrici e un programma di menù. Si noti come il menù offra un programma di input, un programma per tracciare grafici e diversi programmi di analisi. Innanzi tutto, i dati vengono inseriti tramite il programma di input, poi possono essere analizzati o se ne può tracciare il grafico. I dati immessi vengono memorizzati nella matrice D; quelli ridotti con i programmi di analisi sono memorizzati nella matrice R.

Dopo il lancio di ciascun programma si torna al menù. A questo punto i dati possono essere nuovamente analizzati con un programma diverso; si può tracciare il grafico dei dati di input, di quelli ridotti o di entrambi, oppure si possono inserire altri dati.

La Figura 5.1 mostra il menù dell'analisi dei dati e la Figura 5.2 contiene il listato del programma del menù.

1= INPUT DEI DATI
2= GRAFICO DEI DATI
3= MEDIA E DEVIATIONE STANDARD
4= MEDIA MOBILE SU TRE PUNTI
5= MEDIA PONDERATA SU TRE PUNTI
6= MEDIA SU QUATTRO PUNTI
7= REGRESSIONE LINEARE

PROGRAMMA SCELTO

Figura 5.1: Videata: Menù di analisi di dati

```
100 REM -----  
110 N$ = "PROGRAMMI DI ANALISI DI DATI"  
120 REM -----  
130 REM CREAZIONE DELLA MATRICE  
140 DIM P(40), R(40), D(40)  
150 GOSUB 7500 : REM INTESTAZIONE  
160 REM SET UP MENU  
170 X$(1) = "INPUT DEI DATI"  
180 X$(2) = "GRAFICO DEI DATI"  
190 X$(3) = "MEDIA E DEVIATIONE STANDARD"  
200 X$(4) = "MEDIA MOBILE SU TRE PUNTI"  
210 X$(5) = "MEDIA PONDERATA SU TRE PUNTI"  
220 X$(6) = "MEDIA SU QUATTRO PUNTI"  
230 X$(7) = "REGRESSIONE LINEARE"  
240 N = 7  
250 GOSUB 8500: REM DISPLAY MENU  
260 ON X GOSUB 730, 980, 1250, 1460, 1680, 1920, 2150  
270 GOTO 150
```

Figura 5.2: Listato di programma: Menù di analisi dei dati

SUBROUTINE DI ANALISI DEI DATI

Il programma per tracciare i grafici utilizza due subroutine: una traccia le assi; l'altra riferisce i dati alle scale e li traccia. Questi programmi vanno inseriti prima di poter usare il programma grafico. La Figura 5.3 è il listato della subroutine delle assi.

La Figura 5.4 mostra il listato della subroutine che riferisce i dati alle scale e li traccia. Questo programma può tracciare i dati di input, quelli ridotti o entrambi, a seconda del valore della variabile CH\$, controllato dall'utente.

```

280 REM -----
290 REM TRACCIA LE ASSI
300 REM -----
310 HOME
320 REM ASSE DELLE Y
330 FOR I = 1 TO 21
340 VTAB I: HTAB 6: PRINT " I"
350 NEXT I
360 REM SCALA DELL'ASSE DELLE Y
370 FOR I = 1 TO 16 STEP 5
380 VTAB I: HTAB 6: PRINT "-."
390 VTAB I: HTAB 1
400 PRINT MX - (MX - MN) / 4 * (I - 1) / 5
410 NEXT I
420 VTAB 21: HTAB 8
430 REM ASSE DELLE X
440 FOR I = 1 TO 32
450 PRINT "-.";
460 NEXT I
470 REM SUDDIVISIONI ASSE DELLE X
480 FOR I = 8 TO 38 STEP 3
490 VTAB 21: HTAB I: PRINT "I";
500 VTAB 22: HTAB I: PRINT I - 7
510 NEXT I
520 RETURN

```

Figura 5.3: Listato di programma: Subroutine per le assi

```

530 REM -----
540 REM SCALA DEI DATI E GRAFICO
550 REM -----
560 FOR I = TO ND
570 P1 = 21 - INT (20 * (D(I) - MN) / (MX - MN))
580 P2 = 21 - INT (20 * (R(I) - MN) / (MX - MN))
590 IF CH$ <> "I" THEN 630
600 IF P1 > 21 OR P1 < 1 THEN 720
610 VTAB P1: HTAB I + 7: PRINT "*"
620 GOTO 710
630 IF CH$ <> "R" THEN 670
640 IF P2 > 21 OR P2 < 1 THEN 720
650 VTAB P2: HTAB I + 7: PRINT "0"
660 GOTO 710
670 REM DEVONO ESSERE TRACCIATI ENTRAMBI
680 IF P1 > 21 OR P1 < 1 OR P2 > 21 OR P2 < 1 THEN 720
690 VTAB P1: HTAB I + 7: PRINT "*"
700 VTAB P2: HTAB I + 7: PRINT "0"
710 NEXT I: RETURN
720 ERF = 1: RETURN

```

Figura 5.4: Listato di programma: Subroutine per le scale e per tracciare i grafici

INPUT DEI DATI

Descrizione

Questo programma viene utilizzato per inserire i dati che gli altri programmi analizzano e tracciano. È possibile inserire fino a 32 punti. I dati vengono memorizzati nella matrice $D(I)$ e vanno inseriti nell'ordine in cui si presentano; se i dati da inserire sono meno di 32, l'input viene chiuso battendo una D. Lo schermo è organizzato in modo da mostrare gli ultimi dieci dati immessi, per poter effettuare un controllo via via che si procede. Si ricordi che tutti i dati da regolarizzare e di cui si vuol tracciare il grafico vanno prima inseriti tramite il programma di input.

Esempio

Anna Maria vuole immettere nel computer i dati che ha ottenuto in un esperimento di laboratorio. I dati sono i seguenti:

Punto	1	2	3	4	5	6	7	8	9	10
Dato	1	3	5	4	4	5	7	8	9	11
Punto	11	12	13	14	15	16	17	18	19	20
Dato	12	15	13	14	15	16	15	13	12	12
Punto	21	22	23	24	25	26	27	28	29	30
Dato	11	10	7	9	8	7	5	5	4	3
Punto	31	32								
Dato	3	0								

La Figura 5.5 mostra il risultato dell'esempio; la Figura 5.6 contiene il listato del programma.

INPUT DEI DATI

VALORE 10 = 11
VALORE 11 = 12
VALORE 12 = 15
VALORE 13 = 13
VALORE 14 = 14
VALORE 15 = 15
VALORE 6 = 5
VALORE 7 = 7
VALORE 8 = 8
VALORE 9 = 9

BATTERE I VALORI DEI DATI IN ORDINE
PER TERMINARE BATTERE D
VALORE 16 = ?

Figura 5.5: Videata: Input dei dati

```
730 REM -----  
740 N$ = "INPUT DEI DATI"  
750 REM -----  
760 GOSUB 7500 : REM INTESTAZIONE  
770 REM DIALOGO  
780 Q1$ = "BATTERE I VALORI DEI DATI IN ORDINE"  
790 Q2$ = "PER TERMINARE BATTERE D"  
800 Q3$ = "" : GOSUB 6000  
810 ND = 0 : REM NO. OF DATA PTS  
820 FOR I = 1 TO 32  
830 VTAB 23: HTAB 5  
840 VTAB 23: HTAB 1: PRINT SPC( 38): VTAB 23: HTAB 5  
850 PRINT "VALORE "; I; " =";  
860 INPUT D$  
870 IF D$ = "D" THEN 970  
880 VTAB I - INT (I / 10) * 10 + 6  
890 PRINT SPC( 38)  
900 VTAB I - INT (I / 10) * 10 + 6: HTAB 1  
910 PRINT " VALUE";I;" = ";D$  
920 REM CONTO DEI PUNTI  
930 ND = ND + 1  
940 REM MEMORIZZAZIONE DEI DATI  
950 D(I) = VAL (D$)  
960 NEXT I  
970 RETURN
```

Figura 5.6: Listato di programma: Input dei dati


GRAFICO DEI DATI

Descrizione

Questo programma traccia sullo schermo il grafico dei dati inseriti o di quelli ridotti. (I dati ridotti vengono ottenuti tramite i programmi di analisi che seguono.) Per comunicare al programma se tracciare i dati inseriti, quelli ridotti (i dati che sono stati elaborati da uno dei programmi di analisi), o entrambi, rispondere alla domanda del programma rispettivamente I, R o B. Il simbolo * rappresenta i dati inseriti e il simbolo 0 quelli ridotti. Vanno anche specificati i valori massimo e minimo della scala dell'asse delle Y. Se i dati inseriti non rientrano nel campo indicato, compare il messaggio DATI FUORI SCALA e occorre modificare i valori della scala.

Esempio

Tracciamo adesso i dati inseriti nell'esempio precedente. In questo caso scegliamo un valore massimo di 20 e uno minimo di 0. Decidiamo di tracciare i dati di input perchè ancora non li abbiamo ridotti. La Figura 5.7 mostra il grafico che compare sullo schermo e la Figura 5.8 contiene il listato del programma.



PER CONTINUARE BATTERE UN TASTO

Figura 5.7: Videata: Grafico dei dati

```

980 REM -----
990 N$ = "ROUTINE PER TRACCIARE IL GRAFICO"
1000 REM -----
1020 Q2$ = "INPUT O DATI RIDOTTI"
1030 Q3$ = "O ENTRAMBI (I O R O B)"
1040 GOSUB 6000
1050 GOSUB 6500 : REM INSERIMENTO CARATTERE
1060 IF CH$ = "I" OR CH$ = "R" OR CH$ = "B" THEN 1080
1070 GOTO 1050
1080 Q1$ = "":Q2$ = ""
1090 Q3$ = "SCELTA SCALA MAX: "
1100 GOSUB 6000: INPUT MX
1110 Q1$ = "":Q2$ = ""
1120 Q3$ = "SCELTA SCALA MIN: "
1130 GOSUB 6000: INPUT MN
1140 GOSUB 280 : REM TRACCIA LE ASSI
1150 GOSUB 530 : REM TRACCIA IL GRAFICO
1160 IF ERF <> 1 THEN 1210
1170 ERF = 0
1180 Q1$ = "DATI FUORI SCALA"
1190 Q2$ = "PER FAVORE REINSERIRE"
1200 GOTO 1090
1210 VTAB 24: HTAB 1
1220 PRINT "PER CONTINUARE BATTERE UN TASTO"
1230 GET CH$
1240 RETURN

```

Figura 5.8: Listato di programma: Grafico dei dati

MEDIA E DEVIAZIONE STANDARD

Descrizione

Vedremo adesso un programma che calcola la media e la deviazione standard per i dati inseriti con il programma di input. Per usare questo programma basta sceglierlo sul menù dell'analisi dei dati.

Esempio

Anna Maria vuol calcolare la media e la deviazione standard dei dati di laboratorio che ha inserito nel computer. Lancia il programma e ottiene i risultati che compaiono nella Figura 5.9. Il listato del programma di questo esempio è nella Figura 5.10.

MEDIA E DEVIAZIONE STANDARD

MEDIA DEI DATI INSERITI 8.76666667
DEVIAZIONE STANDARD 4.23228334

PER CONTINUARE BATTERE UN TASTO

Figura 5.9: Videata: Media e deviazione standard

```
1250 REM -----
1260 N$ = "MEDIA E DEVIAZIONE STANDARD"
1270 REM -----
1280 GOSUB 7500
1290 REM CALCOLO DELLA MEDIA
1300 ME = 0
1310 FOR I = 1 TO ND
1320 ME = ME + D(I)
1330 NEXT I
1340 ME = ME / ND
1350 REM CALCOLO DELLA DEVIAZ. STANDARD
1360 SD = 0
1370 FOR I = 1 TO ND
1380 SD = SD + (D(I) - ME) ^ 2
1390 NEXT I
1400 SD = (SD / NS) ^ .5
1410 PRINT
1420 PRINT "MEDIA DEI DATI INSERITI"; TAB(20); ME
1430 PRINT "DEVIAZIONE STANDARD"; TAB(20); SD
1440 GOSUB 9000
1450 RETURN
```

Figura 5.10: Listato di programma: Media e deviazione standard

MEDIA MOBILE SU TRE PUNTI

Descrizione


Mostreremo adesso tre programmi di riduzione. Il primo calcola una media mobile su tre punti dei dati memorizzati nella matrice di input. I dati ridotti vengono memorizzati nella matrice $R(I)$ e possono essere mostrati tanto in forma numerica che grafica. La media mobile su tre punti calcola la media dei tre dati che precedono ciascun dato e la utilizza come dato ridotto. Una volta che il programma ha ridotto i dati, domanda

VUOI STAMPARE I DATI

Se rispondete S il programma mostra i dati in forma numerica. Quando tornate al menù principale, potete tracciare i dati ridotti con il programma per tracciare i grafici. Se rispondete con N, il programma torna immediatamente al menù principale, da dove potete lanciare il programma grafico.

Esempio

Anna Maria vuol ridurre i propri dati secondo diverse tecniche. Lancia innanzi tutto il programma per la media mobile su tre punti e traccia il grafico dei risultati insieme a quello dei dati di input. Si possono vedere i dati ridotti nella Figura 5.11; quando i dati ridotti e quelli di input sono uguali, vengono mostrati solo quelli ridotti. Il listato del programma è nella Figura 5.12.



PER CONTINUARE BATTERE UN TASTO

Figura 5.11: Videata: Media mobile su tre punti

```

1460 REM -----
1470 N$ = "MEDIA MOBILE SU TRE PUNTI"
1480 REM -----
1490 GOSUB 7500 : REM INTSTAZIONE
1500 REM INSERIRE RISULTATO IN R(I)
1510 R(1) = 0:R(2) = 0:R(3) = 0
1520 FOR I = 4 TO ND
1530 R(I) = (D(I - 3) + D(I - 2) + D(I - 1)) / 3
1540 NEXT I
1550 Q1$ = "VOLETE STAMPARE I"
1560 Q2$ = "DATI (S O N)"
1570 Q3$ = "": GOSUB 6000
1580 REM CHIEDE SE SI' O NO
1590 GOSUB 9500
1600 IF YN$ = "N" THEN RETURN
1610 HOME
1620 FOR I = 4 TO ND
1630 PRINT R(I),
1640 NEXT I
1650 REM ASPETTA UN TASTO
1660 GOSUB 9000
1670 RETURN

```

Figura 5.12: Listato di programma: Media mobile su tre punti


MEDIA MOBILE PONDERATA

Descrizione

Una seconda tecnica di riduzione è quella della media mobile ponderata. I dati vengono ridotti attribuendo al punto immediatamente precedente un peso tre, al secondo due e al terzo uno. Al punto più recente viene cioè attribuito il peso maggiore. Questo programma può essere richiamato dal menù; i dati che risultano vengono memorizzati nella matrice R(I). Ancora una volta, i dati possono essere mostrati sotto forma numerica o sotto forma grafica, tramite il programma grafico. Si noti che si può memorizzare solo un gruppo di dati ridotti alla volta.

Esempio

Usiamo il programma della media mobile ponderata sui dati di laboratorio di Anna Maria. Nella Figura 5.13 tracciamo sia i dati di input che quelli ridotti tramite il programma della media mobile ponderata. Il listato del programma è nella Figura 5.14.



PER CONTINUARE BATTERE UN TASTO

Figura 5.13: Videata: Media mobile ponderata

```

1680 REM -----
1690 N$ = "MEDIA MOBILE PONDERATA SU 3 PUNTI"
1700 REM -----
1710 GOSUB 7500 : REM INTESTAZIONE
1720 REM INSERIMENTO RISULTATO IN R(I)
1730 R(1) = 0: R(2) = 0: R(3) = 0
1740 FOR I = 4 TO ND
1750 R(I) = (D(I - 3) + 2 * D(I - 2) + 3 * D(I - 1))
1760 R(I) = R(I) / 6
1770 NEXT I
1780 Q1$ = "VOLETE STAMPARE I"
1790 Q2$ = "DATI ( S O N)"
1800 Q3$ = ""
1810 GOSUB 6000
1820 REM CHIEDE SE SI' O NO
1830 GOSUB 9500
1840 IF YN$ = "N" THEN RETURN
1850 HOME
1860 FOR I = 4 TO ND
1870 PRINT R(I),
1880 NEXT I
1890 REM ASPETTA UN TASTO
1900 GOSUB 9000
1910 RETURN

```

Figura 5.14: Listato di programma: Media mobile ponderata


MEDIA SU QUATTRO PUNTI

Descrizione

Esaminiamo adesso un'ultima tecnica di riduzione, la media su quattro punti. In questo caso il primo e il quinto punto precedenti vengono pesati con un fattore uno e il secondo, il terzo e il quarto punto con un fattore due. Questo programma può essere scelto dal menù. I dati ridotti possono essere elencati oppure se ne può tracciare il grafico con il programma grafico.

Esempio

Utilizziamo questa tecnica sui dati di laboratorio di Anna Maria. La Figura 5.15 mostra i dati ridotti insieme a quelli di input. La Figura 5.16 mostra il listato del programma.



PER CONTINUARE BATTERE UN TASTO

Figura 5.15: Videata: Media su quattro punti

```

1920 REM -----
1930 N$ = "MEDIA SU QUATTRO PUNTI"
1940 REM -----
1950 GOSUB 7500 : REM INTESTAZIONE
1960 REM INSERIMENTO RISULTATO IN R(I)
1970 R(1) = 0:R(2) = 0:R(3) = 0:R(4) = 0:R(5) = 0
1980 FOR I = 6 TO ND
1990 D = D(I - 4) + D(I - 3) + D(I - 2)
2000 R(I) = (D(I - 5) + 2 * D + D(I - 1)) / 8
2010 NEXT I
2020 Q1$ = "VOLETE STAMPARE I"
2030 Q2$ = "DATI (S O N)"
2040 Q3$ = "": GOSUB 6000
2050 REM CHIEDE SE SI' O NO
2060 GOSUB 9500
2070 IF YN$ = "N" THEN RETURN
2080 HOME
2090 FOR I = 4 TO ND
2100 PRINT R(I),
2110 NEXT I
2120 REM ASPETTA UN TASTO
2130 GOSUB 9000
2140 RETURN

```

Figura 5.16: Listato di programma: Media su quattro punti

REGRESSIONE LINEARE

Descrizione

La tecnica di regressione lineare consente di trovare la “migliore” linea retta per un insieme di dati. La linea prescelta è quella che meglio corrisponde ai criteri di regressione. Utilizza i dati memorizzati con il programma di input. Questo programma mostra sullo schermo l’equazione della retta; si può anche tracciare la linea con il programma grafico, però non apparirà retta a causa della risoluzione verticale dello schermo.

Esempio

Franco ha ottenuto dei dati da un esperimento e vorrebbe l’equazione della migliore retta corrispondente a quei dati. I dati sono:

Punto	1	2	3	4	5	6	7	8	9	10	11
Dato	1	2	3	5	5	6	7	9	10	11	11

Punto	12	13	14	15	16	17	18	19	20	21	22
Dato	12	13	14	14	15	16	17	19	20	21	22

Il risultato di questo esempio compare nella Figura 5.17; il listato del programma nella Figura 5.18

REGRESSIONE LINEARE

$$Y = .960474309X + .454545444$$

PER CONTINUARE BATTERE UN TASTO

Figura 5.17: Videata: Regressione lineare

```
2150 REM -----  
2160 N$ = "REGRESSIONE LINEARE"  
2170 REM -----  
2180 GOSUB 7500 : REM INTESTAZIONE  
2190 SX = 0: SY = 0:SZ = 0:S2 = 0  
2200 FOR I = 1 TO ND  
2210 SX = SX + I  
2220 SY = SY + D(I)  
2230 XY = XY + D(I) * I  
2240 S2 = S2 + I ^ 2  
2250 NEXT I  
2260 M = (XY - SX * SY / ND) / (S2 - SX ^ 2 / ND)  
2270 B = SY / ND - M * SX / ND  
2280 PRINT  
2290 PRINT " Y = ";M;"X + ";B  
2300 REM AGGIORNAMENTO MATRICE DEI RISULTATI  
2310 FOR I = 1 TO ND  
2320 R(I) = B + M * I  
2330 NEXT I  
2340 GOSUB 9000  
2350 RETURN
```

Figura 5.18: Listato di programma: Regressione lineare

PROGRAMMI DI ARCHIVIAZIONE

Questo sistema mostra come utilizzare il sistema di memorizzazione su dischetto per memorizzare, recuperare e analizzare dati. Cominceremo creando un archivio personale per nomi e indirizzi che può essere utilizzato per rubriche telefoniche, indirizzari e altre applicazioni. Mostreremo poi come creare un archivio relativo all'automobile, su cui registrare i consumi chilometrici.

Di solito le raccolte di programmi non comprendono quelli di archiviazione, perchè possono essere molto complicati. Noi abbiamo potuto ridurne l'abituale complessità, ottenendo programmi di semplice immissione e utilizzazione, tramite la biblioteca di subroutine.

Prima di usare i programmi di questo capitolo, è necessario aggiungere diverse altre subroutine al file SUBLIB. Queste subroutine iniziano alle righe 4000, 5000 e 5500 e i loro listati sono nell'Appendice A.

A differenza da quelli degli altri capitoli, i programmi di questo capitolo devono essere memorizzati in *due* file, uno con i programmi dell'"Archivio personale" e l'altro con quelli dei "Consumi dell'automobile"; ciascun gruppo inizia con un menù.

Questi programmi creano automaticamente sul disco nel drive 1 file che possono essere aggiunti al disco dei programmi o scritti in uno nuovo inizializzato.

Cominciamo definendo alcuni termini. Un *campo* è l'entità più piccola memorizzata in un file. Per l'archivio personale, i campi sono cognome, nome, numero telefonico, via, città provincia e CAP. I campi del registro automobilistico sono data, valore del contachilometri e litri di benzina consumati. Un *record* viene definito come un insieme di campi. Per l'archivio personale un record è nome, indirizzo e numero telefonico di ciascun individuo. Infine, un *file* è un insieme di record.

MENU' DELL'ARCHIVIO PERSONALE

I programmi dell'archivio personale vengono scelti tramite menù. Come potete vedere, i singoli programmi consentono di creare un file, aggiungere record, cancellare record, elencare tutto il file e ricercare singoli record. Innanzi tutto creerete il vostro file tramite il programma "Creazione di file", dopo di che aggiungerete i record con il programma "Inserimento di record". Il programma "Cancellazione di record" permette di eliminare i record non desiderati; potete anche modificare un record cancellandolo e poi reinserendolo con nuove informazioni. Il programma "Contenuto del file" mostra sullo schermo tutto il file. Il programma "Ricerca di un record" consente di trovare tutti i record con un determinato cognome. La Figura 6.1 mostra il menù che appare sullo schermo. La Figura 6.2 contiene il listato del programma del menù.

1= CREAZIONE DEL FILE
2= INSERIMENTO RECORD
3= CONTENUTO DEL FILE
4= CANCELLAZIONE DI RECORD
5= RICERCA SUL FILE

PROGRAMMA SCELTO

Figura 6.1: Videata: Menù dell'archivio personale

```
100 REM -----  
110 N$ = "ARCHIVIO PERSONALE"  
120 REM -----  
127 PRINT SPC( 255) SPC( 185): VTAB 14  
130 DIM S$(10,8)  
140 GOSUB 7500 : REM INTSTAZIONE  
150 X$(1) = "CREAZIONE DEL FILE"  
160 X$(2) = "INSERIMENTO RECORD"  
170 X$(3) = "CONTENUTO DEL FILE"  
180 X$(4) = "CANCELLAZIONE DI RECORD"  
190 X$(5) = "RICERCA SUL FILE"  
200 REM DEFINIZIONE DEI CAMPI  
210 F$(1) = "COGNOME"  
220 F$(2) = "NOME"  
230 F$(3) = "TELEFONO"  
240 F$(4) = "INDIRIZZO"  
250 F$(5) = "CITTA"  
260 F$(6) = "PROVINCIA"  
270 F$(7) = "CAP"  
280 NF = 7:F$ = "PERSONAL FILE.TEXT"  
290 FB$ = "PERSONAL FILE.BACKUP"  
300 N = 5: GOSUB 8500: REM MENU  
310 ON X GOSUB 430, 730, 790, 980, 1080  
320 GOTO 140
```

Figura 6.2: Listato di programma: Menù dell'archivio personale

SUBROUTINE DELL'ARCHIVIO PERSONALE

Quando inserite il programma del menù dovrete inserire anche la subroutine della Figura 6.3. È una routine utilizzata dal programma "Contenuto del file" per mostrare sullo schermo tutti i record del file. Contiene anche una funzione ritardante, che rallenta lo scorrimento dello schermo per dar tempo di leggere i dati che vi compaiono.

```
330 REM -----  
340 REM ELENCO RECORD SU SCHERMO  
350 REM -----  
360 PRINT L;"", P1$; " "; P2$  
370 PRINT ,P3$  
380 PRINT ,P4$  
390 PRINT ,P5$;" ";P6$;" ";P7$  
400 PRINT  
410 FOR I1 = 1 TO 500: NEXT I1  
420 RETURN
```

Figura 6.3: Listato di programma: Subroutine di elencazione archivio personale

CREAZIONE DI FILE E INSERIMENTO DI RECORD

Descrizione

Il programma "Creazione di file" viene utilizzato al momento in cui viene costituito l'archivio personale. Questo programma inizializza il disco e il suo catalogo e lo prepara per l'immissione dei dati. Se "Creazione di file" viene utilizzato su un disco che già contiene un archivio personale, questo viene cancellato. (Il programma avverte che l'archivio verrà cancellato e chiede l'autorizzazione a procedere.) Per scegliere il programma "Creazione di file", premere il tasto 1 quando sullo schermo compare il menù dell'archivio personale.

Il programma "Inserimento di record" serve per aggiungere informazioni al file. Questo programma viene lanciato richiamando il menù e battendo 2. Come per il programma "Creazione di file", viene richiesto di inserire le informazioni che compariranno nell'archivio: cognome, nome, numero telefonico, via, città, provincia e CAP. I record vengono memorizzati nell'ordine di inserimento.

Esempio

La Figura 6.4 mostra cosa appare sullo schermo per un tipico dialogo di inserimento di record. Il file è stato necessariamente creato con il programma "Creazione di file". La Figura 6.5 contiene il listato di entrambi i programmi "Creazione di file" e "Inserimento di record".

```
430 REM -----
440 N$ = "CREAZIONE ARCHIVIO"
450 REM -----
460 GOSUB 7500 : REM INTESTAZIONE
470 Q1$ = "ARCHIVIO PREESISTENTE SARA' CANCELLATO"
480 Q2$ = "" : Q3$ = "CONTINUARE? (S O N)"
490 GOSUB 6000: GOSUB 9500
500 IF YN$ = "N" THEN RETURN
510 K = 1
520 Q1$ = "INSERIRE DATI DEL NOME"
530 Q2$ = ""
```

Figura 6.5: Listato di programma: reazione archivio e inserimento record (continua)

INSERIMENTO RECORD

COGNOME	ROSSI
NOME	MARIO
TELEFONO	456 7890
INDIRIZZO	VIA DANTE 5
CITTA'	SEGRATE
PROVINCIA	MI
CAP	20090

UN ALTRO RECORD DA INSERIRE?
SI' O NO (S/N)

Figura 6.4: Videata: Inserimento record

```
540 FOR L = 1 TO 7
550 Q3$ = F$(L): GOSUB 7000
560 NEXT L
570 PRINT CHR$(4);"OPEN";F$
580 PRINT CHR$(4);"DELETE";F$
590 PRINT CHR$(4);"OPEN";F$
600 PRINT CHR$(4);"WRITE";F$
610 N = 2: PRINT N: FOR L = 1 TO 8: PRINT : NEXT
620 FOR L = 1 TO 8
630 PRINT P$(L)
640 NEXT L
650 PRINT CHR$(4);"CLOSE"
660 GOSUB 7500
670 VTAB 5: HTAB 1
680 PRINT "È STATO APERTO IL NUOVO ARCHIVIO"
690 PRINT
700 PRINT "PER I RECORD USA INSERIMENTO RECORD"
710 GOSUB 9000 : REM PAUSA
720 RETURN
730 REM -----
740 N$ = "INSERIMENTO RECORD"
750 REM -----
760 REM RICHIAMA SUBROUTINE INSER. RECORD
770 GOSUB 4000
780 RETURN
```

Figura 6.5: Listato di programma: Reazione archivio e inserimento record

CONTENUTO DEL FILE

Descrizione

Questo programma mostra il contenuto dell'archivio personale. Il programma viene lanciato richiamando il menù dell'archivio e premendo il tasto 3. Sullo schermo compaiono tutti i record del file, scorrendo sullo schermo. Lo scorrimento viene rallentato da un'apposita funzione.

Esempio

La Figura 6.6 mostra cosa appare sullo schermo nel caso di un archivio personale. Notare come ciascun record ha un numero di record e che i record sono separati gli uni dagli altri da righe vuote. I numeri dei record verranno utilizzati per cancellare i record dal file. La Figura 6.7 mostra il listato del programma.

1. ROSSI, MARIO
456-7890
VIA DANTE 5
SEGRATE, MI 20090
2. ROSSI, ANNA
456-7890
VIA DANTE 5
SEGRATE, MI 20090
3. ESPOSITO, VINCENZO
123-4567
VIA ROMA 3
CORSICO, MI 20100

PER CONTINUARE BATTERE UN TASTO

Figura 6.6: Videata: Contenuto dell'archivio

```

790 REM -----
800 N$ = "CONTENUTO DELL'ARCHIVIO"
810 REM -----
820 GOSUB 7500 : REM INTESTAZIONE
830 HOME : VTAB 5:L = 0
840 PRINT CHR$ (4);"OPEN";F$
850 PRINT CHR$ (4);"READ";F$
860 INPUT N: FOR L = 1 TO 8: INPUT P$: NEXT
865 IF N = 1 THEN 940
870 L = 0: FOR I = 1 TO N - 1
880 L = L + 1
890 PRINT CHR$ (4);"READ";F$
900 INPUT P1$,P2$,P3$,P4$,P5$,P6$,P7$,P8$
910 PRINT CHR$ (4)
920 GOSUB 330 : REM MOSTRA NOME
930 NEXT I
940 PRINT CHR$ (4);"CLOSE"
950 PRINT : PRINT : PRINT : PRINT
960 GOSUB 9000 : REM PAUSA
970 RETURN

```

Figura 6.7: Listato di programma: Contenuto dell'archivio

CANCELLAZIONE DI RECORD

Descrizione

Questo programma consente di cancellare vecchie informazioni dall'archivio personale. Le informazioni possono essere modificate prima cancellandole con questo programma, poi riscrivendole con il programma "Inserimento di record". Con questo programma il record da cancellare viene indicato dal suo numero. Tutti i record successivi vengono rinumerati in modo che non ci siano numeri mancanti. Il numero di record può essere ottenuto tramite il programma "Contenuto del file" o con il programma "Ricerca sul file". La Figura 6.8 mostra il listato del programma di cancellazione.

```
980 REM -----
990 N$ = "CANCELLAZIONE DI RECORD"
1000 REM -----
1010 GOSUB 5500 : REM CANCELL. RECORD
1020 IF L <> - 1 THEN 1040
1030 VTAB 18 : HTAB 1 : PRINT "RECORD NON TROVATO"
1040 REM RICHIESTA DI UN ALTRO
1050 Q2$ = "CANCELLARE" : GOSUB 4500
1060 IF YN$ = "N" THEN RETURN
1070 GOTO 1010
```

Figura 6.8: Listato di programma: Cancellazione di record

RICERCA SUL FILE

Descrizione

Questo programma consente di trovare un record senza dover mostrare tutto il contenuto del file. Va inserito il cognome dell'individuo che si cerca e sullo schermo compare il primo record con quel cognome. Il programma chiede poi se deve continuare la ricerca e comunica se non sono stati trovati individui con quel nome.

Il cognome che si inserisce per la ricerca dev'essere identico a quello che compare nel file. Bisogna cioè scrivere i cognomi sempre nello stesso modo. Per esempio se cercate il cognome francese DU BOIS, ma avete scritto nell'archivio DUBOIS, il programma risponderà di non averlo trovato.

Esempio

Cerchiamo il record VINCENZO ESPOSITO. La Figura 6.9 mostra cosa compare sullo schermo; la Figura 6.10 contiene il listato del programma di ricerca.

```
1080 REM -----
1090 N$ = "RICERCA SUL FILE"
1100 REM -----
1110 GOSUB 7500:F = 0:L = - 1
1120 Q1$ = "RICERCA DEL COGNOME"
1130 Q2$ = "" : Q3$ = "COGNOME: "
1140 GOSUB 6000
1150 INPUT NA$
1160 HOME
```

Figura 6.10: Listato di programma: Ricerca sul file

(continua)

3. ESPOSITO, VINCENZO
123-4567
VIA ROMA 3
CORSICO MI 20100

VOLETE VEDERE
SE CE NE SONO ALTRI? (S O N)

Figura 6.9: Videata: Ricerca sul file

```
1170 VTAB 5: HTAB 1
1180 PRINT CHR$(4);"OPEN",F$
1190 PRINT CHR$(4);"READ",F$
1200 INPUT N: FOR A = 1 TO N
1220 L = L + 1
1230 INPUT P1$, P2$, P3$, P4$, P5$, P6$, P7$, P8$
1240 IF P1$ <> NA$ THEN 1310
1250 REM MOSTRA IL RECORD
1260 PRINT CHR$(4);F = 1: VTAB 14
1270 PRINT SPC( 255) SPC( 145): VTAB 14
1280 Q1$ = "VOLETE VEDERE":Q3$ = ""
1290 Q2$ = "SE CE NE SONO ALTRI? (S O N)"
1300 GOSUB 6000: GOSUB 9500: IF YN$ = "N" THEN 1320
1310 NEXT A : REM RECORD FINITI
1320 PRINT CHR$(4);"CLOSE"
1340 VTAB 18: HTAB 1
1350 IF F=0 THEN PRINT "NON SONO STATI TROVATI RECORD CORRISP."
1360 REM RICHIESTA SE ANCORA
1370 Q2$ = "CERCARE" : GOSUB 4500
1380 IF YN$ = THEN RETURN
1390 GOTO 1080
```

Figura 6.10: Listato di programma: Ricerca sul file

MENU' DEL DIARIO AUTOMOBILISTICO

Costruiremo adesso un file che conterrà il consumo chilometrico della vostra auto. Potete usare il programma inserendo una data, il valore del contachilometri e la quantità di carburante acquistata. Il programma “Contenuto del diario automobilistico” calcola il consumo chilometrico medio per litro e lo mostra. I programmi del diario automobilistico vengono richiamati da menù. La Figura 6.11 mostra il menù e la Figura 6.12 il listato del programma del menù.

1= CREAZIONE DEL FILE
2= INSERIMENTO RECORD
3= CONTENUTO DEL FILE
4= CANCELLAZIONE RECORD

PROGRAMMA SCELTO

Figura 6.11: Videata: Menù del diario automobilistico

```
100 REM -----  
110 N$ = "DIARIO AUTOMOBILISTICO"  
120 REM -----  
130 GOSUB 7500 : REM INTESTAZIONE  
140 X$(1) = "CREAZIONE DEL FILE"  
150 X$(2) = "INSERIMENTO RECORD"  
160 X$(3) = "CONTENUTO DEL FILE"  
170 X$(4) = "CANCELLAZIONE RECORD"  
180 REM DEFINIZIONE DEI CAMPI  
190 F$(1) = "DATA"  
200 F$(2) = "CONTACHILOMETRI"  
210 F$(3) = "LITRI"  
220 NF = 3:F$ = "CAR.TEXT"  
230 FB$ = "CAR.BACKUP"  
240 N = 4: GOSUB 8500: REM MENU  
250 ON X GOSUB 270,540,600,870  
260 GOTO 130
```

Figura 6.12: Listato di programma: Menù del diario automobilistico

CREAZIONE DEL FILE E INSERIMENTO DI RECORD

Descrizione

Il programma di creazione di file viene utilizzato quando si costituisce il diario automobilistico. Questo programma inizializza il disco e il suo catalogo e lo predispone per l'immissione dei dati. Il programma va lanciato una volta sola. Se si vogliono registrare i dati di più automobili, bisogna utilizzare più dischetti.

Il programma di inserimento permette di aggiungere record al diario automobilistico. Si aderisce al formato stabilito nel programma di creazione del file, inserendo la data, il valore del contachilometri e i litri di carburante. Ancora una volta i record sono memorizzati nell'ordine in cui vengono inseriti. Se durante l'inserimento di un record viene commesso un errore, il calcolo dei chilometri percorsi con un litro risulterà probabilmente errato: conviene controllare attentamente i record subito dopo averli inseriti, perchè è difficile effettuare una correzione dopo che sono stati immessi altri record.

Esempio

Il 17 febbraio 1983 avete comprato 14.9 litri di benzina con il contachilometri che mostrava 13.523 chilometri. La Figura 6.13 mostra cosa appare sullo schermo quando questi dati vengono aggiunti a quelli già nel file. La Figura 6.14 è il listato dei programmi di creazione del file e dell'inserimento di record.

```
270 REM -----
280 N$ = CREAZIONE DEL FILE
290 REM -----
300 GOSUB 7500 : REM INTESTAZIONE
310 K = 1
320 Q1$ = "FILE PREESISTENTE SARA' CANCELLATO"
330 Q2$ = "" : Q3$ = "CONTINUARE? (S O N)"
340 GOSUB 6000: GOSUB 9500: IF YN$ = "N" THEN RETURN
```

Figura 6.14: Listato di programma: Creazione del file e inserimento record (continua)

INSERIMENTO RECORD

DATA	17/2/83
CONTACHILOMETRI	33523
LITRI	34.9

UN ALTRO RECORD DA INSERIRE?
SI' O NO (S/N)

Figura 6.13: Videata: Inserimento record

```
350 Q1$ = "DATI PRIMO INSERIMENTO"  
360 Q2$ = ""  
370 Q3$ = "DATA" : GOSUB 7000  
380 Q3$ = "CONTACHILOMETRI" :GOSUB 7000  
390 Q3$ = "LITRI" : GOSUB 7000  
400 PRINT CHR$(4);"OPEN";F$  
410 PRINT CHR$(4);"DELETE";F$  
420 PRINT CHR$(4);"OPEN";F$  
430 PRINT CHR$(4);"WRITE";F$  
440 N = 2: PRINT N  
450 FOR L = 1 TO 8: PRINT : NEXT : FOR I = 1 TO 8  
460 PRINT P$(I): NEXT I  
470 PRINT CHR$(4);"CLOSE";F$  
480 GOSUB 7500: VTAB 5: HTAB 1  
490 PRINT "È STATO APERTO IL NUOVO ARCHIVIO"  
500 PRINT  
510 PRINT "NOW USE ADD RECORD TO ADD TO IT"  
520 GOSUB 9000 : REM PAUSA  
530 RETURN  
540 REM -----  
550 N$ = "INSERIMENTO RECORD"  
560 REM -----  
570 REM RICHIAMA SUBROUTINE INSER. RECORD  
580 GOSUB 4000  
590 RETURN
```

Figura 6.14: Listato di programma: Creazione del file e inserimento record

CONTENUTO DEL DIARIO AUTOMOBILISTICO

Descrizione

Questo programma mostra sullo schermo il contenuto del diario automobilistico. Viene inoltre calcolata e mostrata sullo schermo la percorrenza chilometrica media per ciascun record. Lo schermo scorrerà lentamente in modo da permettere la comoda visione delle informazioni.

Questo programma consente di tener sotto controllo i consumi dell'automobile. Un'improvviso aumento del consumo può essere indice di un guasto.

Esempio

Volete vedere il contenuto del vostro diario automobilistico. La Figura 6.15 mostra cosa appare sullo schermo con questo esempio. La Figura 6.16 è il listato del programma.

```
600 REM -----
610 N$ = "CONTENUTO FILE"
620 REM -----
630 GOSUB 7500 : REM INTESTAZIONE
640 MI = 0:GAL = 0
650 PRINT "N. DATA"; TAB(15); "CHILOMETRI"
660 PRINT TAB(23); "LITRI"; TAB(30); "MEDIA"
670 PRINT CHR$ (4);"OPEN";F$
680 PRINT CHR$ (4);"READ";F$
```

Figura 6.16: Listato di programma: Contenuto del diario automobilistico (continua)

CONTENUTO FILE

N.	DATA	CHILOMETRI	LITRI	MEDIA
1	31/1/83	12730	15.6	
2	7/2/83	12925	13.1	14.89
3	12/2/83	13210	15.8	18.04
4	17/2/83	13523	14.9	21.01
5	22/2/83	13801	15.2	18.29

 PER CONTINUARE BATTERE UN TASTO

Figura 6.15: Videata: Contenuto del diario automobilistico

```

690 INPUT N: FOR L = 1 TO 8: INPUT P$: NEXT
695 IF N = 1 THEN 830
700 FOR I = 1 TO N - 1
710 PRINT CHR$ (4);"READ"; F$
720 INPUT P1$, P2$, P3$, P4$, P5$, P6$, P7$, P8$
730 PRINT CHR$ (4)
740 GAL = VAL (P3$)
750 AVE = ( VAL (P2$) - MI) / GAL
760 MI = VAL (P2$)
770 PRINT I; TAB( 5);P1$; TAB( 15);P2$; TAB( 23);P3$;
780 IF I = 1 THEN PRINT
785 LET AVE = INT (AVE * 100 + .5) / 100
790 IF I <> 1 THEN PRINT TAB( 30);AVE
800 REM PAUSA PER CONSENTIRE LETTURA
810 FOR I1 = 1 TO 500
820 NEXT I1,I
830 PRINT CHR$ (4);"CLOSE"
840 PRINT : PRINT : PRINT : PRINT
850 GOSUB 9000 : REM PAUSA
860 RETURN
  
```

Figura 6.16: Listato di programma: Contenuto del diario automobilistico

CANCELLAZIONE DI RECORD

Descrizione

Questo programma consente di cancellare un record dal file. Lo si può usare per eliminare informazioni che contengono errori. Il record da cancellare va indicato con il suo numero. Quando si cancella un record in mezzo al file, tutti i successivi vengono rinumerati in modo che non rimangano buchi. La Figura 6.17 mostra il listato del programma. I record sbagliati vanno cancellati subito, per evitare che il programma calcoli un consumo errato.

```
870 REM -----
880 N$ = "CANCELLAZIONE DI RECORD"
890 REM -----
900 REM RICHIAMA CANCELLAZIONE DI RECORD
910 GOSUB 5500
920 IF L <> - 1 THEN 940
930 VTAB 18 : PRINT "IL RECORD NON È STATO TROVATO"
940 REM RICHIESTA SE ANCORA
950 Q2$ = "CANCELLARE" : GOSUB 4500
960 IF YN$ = "N" THEN RETURN
970 GOTO 910
```

Figura 6.17: Listato di programma: Cancellazione di record

PROGRAMMI DI ESERCIZI MATEMATICI

Questo capitolo presenta tre gruppi di programmi per utilizzare il computer per esercizi matematici. Ciascun gruppo va memorizzato in un file autonomo e viene controllato da menù, che dev'essere inserito per primo. Gli esercizi previsti riguardano addizione, sottrazione, moltiplicazione e divisione, sia di numeri interi che di frazioni. I tre menù offrono programmi a più livelli di difficoltà e si rivolgono a studenti dalle elementari alle scuole medie superiori. Ciascun programma tiene il conto delle risposte giuste e di quelle sbagliate. Le risposte sbagliate vengono accolte da un "bip". Vi accorgete come questi programmi vi faranno migliorare notevolmente nei calcoli, e che sono anche utili per i compiti a casa!

ESERCIZI MATEMATICI 1 - MENU'

Il primo gruppo di quattro programmi offre esercizi di addizione, sottrazione, moltiplicazione e divisione di due numeri interi. Questo gruppo di programmi inizia con il menù , e l'esercizio desiderato va scelto premendo il tasto corrispondente. Fatto questo, il programma domanda qual'è il numero più grande con cui si vuol lavorare e presenta due numeri e un operatore, a indicare l'addizione, la sottrazione, la moltiplicazione o la divisione. La risposta viene inserita da sinistra a destra; quando è sbagliata suona un "bip". Ciascun programma tiene il conto delle risposte corrette e di quelle sbagliate. (Vi raccomandiamo di far le cose difficili e di scegliere un numero alto!

Per usare questi programmi vanno prima inseriti il programma del menù e due subroutine comuni. La Figura 7.1 mostra il menù e la 7.2 il listato del suo programma.

1= ADD
2= SOTTR
3= MOLT
4= DIV

PROGRAMMA SCELTO

Figura 7.1: Videata: Esercizi Matematici 1 - Menù

```
100 REM -----  
110 N$ = "ESERCIZI MATEMATICI 1"  
120 REM -----  
130 GOSUB 7500 : REM INTESTAZIONE  
140 FOR X = 1 TO 1000: NEXT  
150 K = 1  
160 REM RUN MENU AND CHOOSE  
170 REM TIPO DI ESERCIZIO  
180 X$(1) = "ADD" : X$(2) = "SOTTR"  
190 X$(3) = "MOLT" : X$(4) = "DIV"  
200 N = 4  
210 GOSUB 8500  
220 R = 0:W = 0  
230 Q1$ = "QUAL'È IL NUMERO PIU' GRANDE"  
240 Q2$ = "CHE VOLETE"  
250 Y = X  
260 Q3$ = X$(Y)  
270 REM FARE DOMANDE  
280 GOSUB 6000  
290 PRINT TAB( 20);: INPUT X  
300 ON Y GOSUB 590, 760, 970, 1150  
310 GOTO 100
```

Figura 7.2: Listato di programma: Esercizi Matematici 1 - Menù

SUBROUTINE DEGLI ESERCIZI MATEMATICI 1

I programmi degli esercizi matematici 1 utilizzano due subroutine, per mostrare le domande e stampare e tabulare le risposte, che vanno inserite prima degli esercizi veri e propri. Queste subroutine permettono di ridurre notevolmente la lunghezza dei programmi che seguiranno. Le subroutine generano problemi tramite un generatore di numeri casuali, tengono il conto del numero totale dei problemi svolti e dei totali delle risposte giuste e di quelle sbagliate e generano il suono. Le subroutine sono elencate nella Figura 7.3.

```

320 REM -----
330 REM MOSTRA DOMANDA E
340 REM OTTIENI RISPOSTA
350 REM -----
360 S1 = 4 - LEN ( STR$ (N1))
370 VTAB 5: HTAB 20: PRINT N1$; SPC( S1);N1
380 S2 = 4 - LEN ( STR$ (N2))
390 VTAB 6: HTAB 19: PRINT SN$; SPC( S2);N2
400 VTAB 7: HTAB 20: PRINT "----"
410 VTAB 10 : PRINT "BATTI RISPOSTA: "
420 INPUT A
430 VTAB B
440 PRINT "LA RISPOSTA GIUSTA È: ";
450 RETURN
460 REM -----
470 REM TERMINA IL PROBLEMA
480 REM -----
490 VTAB 12 : PRINT "AVETE RISPOSTO"
500 REM AGGIORNA IL PUNTEGGIO
510 IF S=1 THEN R=R+1 : PRINT "GIUSTO"
520 IF S < > 1 THEN W=W+1 : PRINT "SBAGLIATO."; CHR$(7)
530 PRINT "ESERCIZI SVOLTI: "; TAB(20); R+W
540 PRINT "CORRETTI: "; TAB (20); R
550 PRINT "ERRORI: "; TAB(20); W
560 REM CHIEDE SE ANCORA
570 Q2$ = "ESERCITARVI" : GOSUB 4500
580 RETURN

```

Figura 7.3: Listato di programma: Subroutine degli Esercizi Matematici

ADDIZIONE

Descrizione

L'esercitazione sull'addizione presenta due numeri casuali alla volta. La risposta va fornita da sinistra a destra, il che talvolta significa che il problema va prima risolto su carta. Il numero più alto che comparirà viene determinato dalla risposta data quando compare il menù.

Esempio

L'esempio della Figura 7.4 mostra un'addizione sbagliata. Si noti che sono stati svolti sei problemi e che quattro risposte su sei sono state giuste. Il listato del programma dell'addizione è nella Figura 7.5.

SOTTRAZIONE

Descrizione

L'esercitazione sulla sottrazione è simile a quella sull'addizione. Il metodo di selezione dei due numeri garantisce che la risposta sia sempre positiva. Ricordate che la pratica conduce alla perfezione: fate molti esercizi.

Esempio

La Figura 7.6 mostra una risposta sbagliata. Notate come le risposte a due problemi su tre siano state corrette. La Figura 7.7 mostra il listato dell'esercitazione sulla sottrazione.

SOTTRAZIONE

		20
	-	12
LA RISPOSTA GIUSTA È:		8
BATTI RISPOSTA:	78	
AVETE RISPOSTO GIUSTO!		
ESERCIZI SVOLTI:	3	
CORRETTI:	2	
ERRORI:	1	

VOLETE
ESERCITARVI
DI NUOVO? (S O N)

Figura 7.6: Videata: Sottrazione

```
760 REM -----  
770 N$ = "SOTTRAZIONE"  
780 REM -----  
790 GOSUB 7500  
800 N1 = INT ( RND (1) * X)  
810 N2 = INT ( RND (1) * X)  
820 N3 = N1  
830 IF N1 > N2 THEN 850  
840 N1 = N2:N2 = N3  
850 SN$ = "-"  
860 REM MOSTRA LA DOMANDA E  
870 REM RICEVE LA RISPOSTA  
880 GOSUB 320  
890 REM INTRODUCE 'FLAG'  
900 S = 0  
910 IF A = N1 - N2 THEN S = 1  
920 PRINT SPC( 4 - LEN ( STR$ (N1 - N2)))N1 - N2  
930 REM TERMINA  
940 GOSUB 460  
950 IF YN$ = "N" THEN RETURN  
960 GOTO 760
```

Figura 7.7: Listato di programma: Sottrazione

MOLTIPLICAZIONE

Descrizione

Questo programma permette di esercitarsi a moltiplicare due numeri. Il programma genera numeri casuali e il numero più grande con cui si vuol lavorare viene scelto quando si scelgono gli esercizi sulla moltiplicazione dal menù. Questo programma mostra la risposta su una sola riga. Per moltiplicazioni di due o più cifre può convenire scegliere il programma “Moltiplicazioni lunghe” che compare in seguito nel capitolo e che mostra i passaggi intermedi necessari per arrivare alla risposta.

Esempio

La Figura 7.8 mostra cosa compare sul video per il programma sulla moltiplicazione. Si vede che quattro risposte su cinque hanno ricevuto una risposta corretta. La Figura 7.9 è il listato del programma sulla moltiplicazione.

DIVISIONE

Descrizione

La quarta esercitazione matematica consente di esercitarsi a dividere due numeri interi. Di nuovo, dovete indicare il numero più grande per cui volete dividere (il divisore) al momento in cui scegliete il programma di divisione dal menù. Il programma genera numeri che garantiscono che il risultato della divisione sia intero. Si noti che il numero più grande con cui si lavora è la risposta più grande del programma.

Esempio

La Figura 7.10 mostra un esercizio del programma di divisione. Si noti come questo sia il quinto problema portato a termine e che tutte le risposte sono state giuste. Il listato del programma di divisione è nella Figura 7.11.

DIVISIONE	
	126
	/ 18
LA RISPOSTA GIUSTA È:	7
 BATTI RISPOSTA:	 ??
 AVETE RISPOSTO GIUSTO!	
ESERCIZI SVOLTI:	5
CORRETTI:	5
ERRORI:	0

VOLETE ESERCITARVI DI NUOVO? (S O N)	

Figura 7.10: Videata: Divisione

```

1150 REM -----
1160 N$ = "DIVISIONE"
1170 REM -----
1180 GOSUB 7500
1190 N1 = INT ( RND (1) * X)
1200 N2 = INT ( RND (1) * X)
1210 IF N2 = 0 THEN 1200
1220 N1 = N1 * N2
1230 SN$ = "/"
1240 REM MOSTRA LA DOMANDA E
1250 REM RICEVE LA RISPOSTA
1260 GOSUB 320
1270 REM INTRODUCE 'FLAG'
1280 S = 0
1290 IF A = N1 / N2 THEN S = 1
1300 PRINT SPC( 4 - LEN ( STR$ (N1 / N2)))N1 / N2
1310 REM TERMINA
1320 GOSUB 460
1330 IF YN$ = "N" RETURN
1340 GOTO 1150

```

Figura 7.11: Listato di programma: Divisione

ESERCIZI MATEMATICI 2 - MENU'

Gli “Esercizi matematici 2” sono costituiti da due esercitazioni matematiche avanzate. Il primo programma consente di esercitarsi a sommare numeri incolonnati; il secondo permette di far pratica nelle moltiplicazioni di grossi numeri facendo lunghe moltiplicazioni. Per entrambi i programmi, la risposta andrebbe trovata con penna e carta e poi inserita nel computer. Si noti come il programma di moltiplicazione mostri tutti i passaggi intermedi. La Figura 7.12 contiene il menù degli “Esercizi matematici 2” e la 7.13 il listato del programma del menù.

1= ADDIZIONE IN COLONNA
2= MOLTIPLICAZIONE LUNGA

PROGRAMMA SCELTO

Figura 7.12: Videata: Esercizi Matematici 2 - Menù

```
100 REM -----  
110 N$ = "ESERCIZI MATEMATICI 2"  
120 REM -----  
130 GOSUB 7500 : REM INTSTAZIONE  
140 FOR X = 1 TO 1000: NEXT  
150 X$(1) = "ADDIZIONE IN COLONNA"  
160 X$(2) = "MOLTIPLICAZIONE LUNGA"  
170 REM CREA IL MENU' E LO MOSTRA  
180 N = 2: GOSUB 8500  
190 ON X GOSUB 210,610  
200 GOTO 100
```

Figura 7.13: Listato di programma: Esercizi Matematici 2 - Menù

ADDIZIONE IN COLONNA

Descrizione

Il programma di addizione in colonna consente di esercitarsi ad addizionare numeri in colonna. Questa esercitazione viene scelta premendo il numero 1 quando compare il menù. Il programma chiede poi di fornire il valore più grande e quanti numeri si vuole addizionare. Il programma tiene il conto delle risposte giuste e sbagliate.

Esempio

La Figura 7.14 mostra l'addizione corretta di cinque numeri. La Figura 7.15 è il listato del programma per questa esercitazione.

```
210 REM -----
220 N$ = "ADDIZIONE IN COLONNA"
230 REM -----
240 GOSUB 7500 : REM INTESTAZIONE
250 Q1$ = ""
260 Q2$ = "QUAL'È IL NUMERO PIU' GRANDE"
270 Q3$ = "CHE VOLETE ADDIZIONARE"
280 GOSUB 6000
290 INPUT LG
300 Q2$ = "QUANTI NUMERI"
310 GOSUB 6000
320 INPUT NU
330 R = 0:W = 0
340 GOSUB 7500 : REM INTESTAZIONE
350 T = 0
360 FOR I = 1 TO NU
370 N1 = INT ( RND (1) * LG)
380 T = T + N1
```

Figura 7.15: Listato di programma: Addizione in colonna

(continua)

ADDIZIONE IN COLONNA

	9
	9
	14
	20
	19
	71
LA RISPOSTA GIUSTA È:	71

BATTI RISPOSTA: ?71

AVETE RISPOSTO GIUSTO!

ESERCIZI SVOLTI: 1

CORRETTI: 1

ERRORI: 0

 VOLETE
 ESERCITARVI
 DI NUOVO? (S O N)

Figura 7.14: Videata: Addizione in colonna

```

390 PRINT TAB( 20);
400 PRINT SPC( 6 - LEN ( STR$ (N1)));N1
410 NEXT I
420 PRINT TAB( 20); "-----"
430 VTAB NU + 6
440 PRINT "LA VOSTRA RISPOSTA: "; TAB(19)
450 INPUT A
460 VTAB NU + 4
470 PRINT "LA RISPOSTA GIUSTA: "; TAB(20)
480 PRINT SPC/ 6 - LEN ( STR$ (T));T
490 VTAB NU + 8:S = 0
500 IF A = T THEN S = 1
510 PRINT "AVETE RISPOSTO ";
520 IF S=1 THEN R=R+1 : PRINT "GIUSTO"
530 IF S < > 1 THEN W=W+1 : PRINT "SBAGLIATO."
540 PRINT "ESERCIZI SVOLTI: "; TAB(20); R+W
550 PRINT "CORRETTI: "; TAB(20); W
560 PRINT "ERRORI: "; TAB(20); W
570 REM CHIEDE SE ANCORA
580 Q2$ = "ESERCITARVI" : GOSUB 4500
590 IF YN$ = "N" THEN RETURN
600 GOTO 340
    
```

Figura 7.15: Listato di programma: Addizione in colonna

MOLTIPLICAZIONE LUNGA

Descrizione

Presentiamo adesso un programma che consente di far pratica nell'effettuare moltiplicazioni lunghe. Per scegliere questo programma, premete il tasto 2 dal menù degli "Esercizi matematici 2", dopo di che fornite il numero di cifre con cui volete lavorare. Vengono mostrati due numeri da moltiplicare; trovate la risposta su foglio e inseritela nel computer. Sullo schermo appare la risposta giusta e i passaggi necessari per trovarla. Un esame di questi passaggi rivelerà dove è stato commesso un eventuale errore.

Esempio

La Figura 7.16 mostra il risultato di una moltiplicazione di due grossi numeri. Si noti che ci sono tutti i passaggi intermedi. La Figura 7.17 contiene il listato del programma.

```
610 REM -----
620 N$ = "MOLTIPLICAZIONE LUNGA"
630 REM -----
640 GOSUB 7500 : REM INTESTAZIONE
650 Q1$ = ""
660 Q2$ = "QUANTE CIFRE?"
670 Q3$ = "VOLETE MOLTIPLICARE?"
680 GOSUB 6000
690 INPUT N
700 GOSUB 7500
710 N1 = 0:N2 = 0
720 FOR I = 0 TO N - 1
730 N1(I) = INT ( RND (1) * 10)
740 N1 = N1 + N1(I) * 10 ^ I
750 N2(I) = INT ( RND (1) * 10)
760 N2 = N2 + N2(I) * 10 ^ I
770 NEXT I
```

Figura 7.17: Listato di programma: Moltiplicazione lunga

(continua)

MOLTIPLICAZIONE LUNGA

```
      8801
      4447
-----
      61607
      352040
      3520400
-----
     39138047
```

BATTETE LA RISPOSTA: ->? 39138047

AVETE RISPOSTO GIUSTO!

VOLETE
MOLTIPLICARE
DI NUOVO? (S O N)

Figura 7.16: Videata: Moltiplicazione lunga

```
780 N1 = INT (N1);N2 = INT (N2)
790 HTAB 20: PRINT SPC( 10 - LEN ( STR$ (N1)));N1
800 HTAB 20: PRINT SPC( 10 - LEN ( STR$ (N2)));N2
810 PRINT TAB( 20); "-----"
820 VTAB 16
830 PRINT "BATTETE LA RISPOSTA: ->";
840 INPUT A
850 VTAB 6
860 FOR I = 0 TO N - 1
870 S1 = 10 - LEN ( STR$ (N1 * N2(I) * 10 ^ I))
880 HTAB 20: PRINT SPC( S1);N1 * N2(I) * 10 ^ I
890 NEXT I
900 PRINT TAB( 20); "-----"
910 S1 = 10 - LEN ( STR$ (N1 * N2))
920 PRINT TAB( 20); SPC( S1);N1 * N2: PRINT
930 VTAB 18 : PRINT "AVETE RISPOSTO ";
940 S = 0
950 IF A = N1 * N2 THEN S = 1
960 IF S=1 THEN PRINT "GIUSTO"
970 IF S <> 1 THEN PRINT "SBAGLIATO."; CHR$(7)
980 Q2$ = "MOLTIPLICARE" : GOSUB 4500
990 IF YN$ = "N" THEN RETURN
1000 GOTO 700
```

Figura 7.17: Listato di programma: Moltiplicazione lunga

ESERCIZI MATEMATICI 3 - MENU'

L'ultimo gruppo di esercizi matematici consente di esercitarsi nell'addizione, la sottrazione, la moltiplicazione e la divisione di frazioni. Le risposte vanno trovate su carta e poi inserite nel computer. Tutte le risposte di questa sezione sono ridotte alla forma del massimo comun denominatore; non vengono effettuate le conversioni in numeri misti. Ciascun programma tiene il conto delle risposte giuste e sbagliate.

I quattro programmi sulle frazioni sono controllati da un menù comune. La Figura 7.18 mostra il menù degli "Esercizi matematici 3"; la Figura 7.19 contiene il listato del programma del menù.

1= ADD
2= SOTTR
3= MOLT
4= DIV

PROGRAMMA SCELTO

Figura 7.18: Videata: Esercizi Matematici 3 - Menù

```
100 REM -----  
110 N$ = "ESERCIZI MATEMATICI 3"  
120 REM -----  
130 GOSUB 7500 : REM INTESTAZIONE  
140 FOR X = 1 TO 1000: NEXT  
150 REM SCEGLIERE DA MENU'  
160 REM TIPO DI ESERCIZIO  
170 X$(1) = "ADD" : X$(2) = "SOTTR"  
180 X$(3) = "MOLT" : X$(4) = "DIV"  
190 N = 4: GOSUB 8500  
200 Y = X  
210 R = 0:W = 0  
220 Q1$ = "QUAL'È IL DENOMINATORE PIU' GRANDE"  
230 Q2$ = "CHE VOLETE"  
240 Q3$ = X$(Y)  
250 GOSUB 6000  
260 PRINT TAB( 20);: INPUT X  
270 ON Y GOSUB 920,1130,1360,1580  
280 GOTO 100
```

Figura 7.19: Listato di programma: Esercizi Matematici 3 - Menù

SUBROUTINE RELATIVE ALLE FRAZIONI

Prima di scrivere i programmi sulle frazioni, dovrete inserire le quattro subroutine della Figura 7.20. Queste subroutine generano le frazioni e il massimo comun denominatore, mostrano la domanda e stampano la risposta.

Le subroutine utilizzano un generatore di numeri casuali per scegliere i numeratori e i denominatori. Inoltre mostrano sullo schermo il problema in forma frazionaria, come appare qui sotto:

$$\frac{4}{5} + \frac{5}{6}$$

Quando si risponde, le subroutine chiedono separatamente il numeratore e il denominatore.

```
290 REM -----
300 REM QUESTA ROUTINE GENERA
310 REM NUMERATORI E
320 REM DENOMINATORI
330 REM -----
340 DEF FN ND(X) = INT ( RND (1) * X)
350 N1 = FN NS(X): IF N1 = 0 THEN 350
360 N2 = FN NS(X): IF N2 = 0 THEN 360
370 D1 = FN NS(X): IF D1 = 0 THEN 370
380 D2 = FN NS(X): IF D2 = 0 THEN 380
390 REM CONTROLLARE CHE LA SOTTR. SIA POSITIVA
400 IF N1 / D1 >= N2 / D2 THEN 440
410 REM ALTRIMENTI INVERTIRE
420 N3 = N2:N2 = N1:N1 = N3
430 D3 = D2:D2 = D1:D1 = D3
440 RETURN
```

Figura 7.20: Listato di programma: Subroutine relative alle frazioni

(continua)

```

450 REM -----
460 REM ROUTINE PER TROVARE
470 REM IL MASSIMO COMUN
480 REM DENOMINATORE
490 REM -----
500 U = N3:V = D3
510 G = U - V * INT ( U / V )
520 IF G = 0 THEN RETURN
530 U = V
540 V = G
550 GOTO 510
560 REM -----
570 REM MOSTRA LA DOMANDA E
580 REM RICEVE LA RISPOSTA
590 REM -----
600 VTAB 4: HTAB 20
610 S1 = 3 - LEN ( STR$ (N1));S2 = 3 - LEN ( STR$ (N2))
620 PRINT SPC( S1);N1; TAB( 26); SPC( S2);N2
630 PRINT TAB( 20);"--- ";SN$;" ---"
640 S1 = 3 - LEN ( STR$ (D1));S2 = 3 - LEN ( STR$ (D2))
650 PRINT TAB( 20); SPC( S1);D1; TAB( 26); SPC( S2); D2
660 VTAB 13
670 PRINT "BATTI IL NUMERATORE: ";
680 INPUT AN
690 PRINT "BATTI IL DENOMINATORE: ";
700 INPUT AD: PRINT
710 VTAB 8 : PRINT "LA RISPOSTA GIUSTA È: ";
720 RETURN
730 REM -----
740 REM TERMINA L'ESERCIZIO
750 REM -----
760 S1 = 3 - LEN ( STR$ (N3));S2 = 3 - LEN ( STR$ (D3))
770 PRINT TAB( 20); SPC( S1);N3
780 PRINT TAB( 20); "---"
790 PRINT TAB( 20); SPC( S2);D3
800 S = 0
810 IF AN = N3 AND AD = D3 THEN S = 1
820 VTAB 16 : PRINT "AVETE RISPOSTO ";
830 REM UPDATE SCORE
840 IF S=1 THEN PRINT "GIUSTO" : =+1
850 IF S <> 1 THEN PRINT "SBAGLIATO."; CHR$(7) : W=W+1
860 PRINT "ESERCIZI SVOLTI."; TAB(20); R+W
870 PRINT "CORRETTI."; TAB(20); R
880 PRINT "ERRORI."; TAB(20); W
890 REM CHIEDE SE ANCORA
900 Q2$ = "ESERCITARVI" : GOSUB 4500
910 RETURN

```

Figura 7.20: Listato di programma: Subroutine relative alle frazioni

ADDIZIONE DI FRAZIONI

Descrizione

Questo programma consente di esercitarsi nell'addizione di due frazioni. Per sceglierlo, battere 1 quando compare il menù di "Esercizi matematici 3". Si deve anche fornire il denominatore più grande con cui si vuol lavorare. (Non abbiate paura dei grandi denominatori!)

Esempio

La Figura 7.21 mostra l'addizione di due frazioni. Notate come questo sia il secondo problema di una serie. La Figura 7.22 contiene il listato del programma.

ADDIZIONE

$$\frac{2}{4} + \frac{3}{6}$$

LA RISPOSTA GIUSTA È: $\frac{1}{1}$

BATTERE NUMERATORE: ?1

BATTERE DENOMINATORE: ?1

AVETE RISPOSTO GIUSTO!

ESERCIZI SVOLTI: 2

CORRETTI: 1

ERRORI: 1

VOLETE

ESERCITARVI

DI NUOVO? (S O N)

Figura 7.21: Videata: Addizione di frazioni

```
920 REM -----
930 N$ = "ADDIZIONE"
940 REM -----
950 GOSUB 7500 : REM INTESTAZIONE
960 REM GENERA NUMER. E DENOM.
970 GOSUB 290
980 SN$ = "+"
990 REM MOSTRA LA DOMANDA
1000 REM E RICEVE LA RISPOSTA
1010 GOSUB 560
1020 REM BEFORE CLEANUP
1030 N3 = D2 * N1 + D1 * N2
1040 D3 = D1 * D2
1050 TROVA IL MASSIMO
1060 REM COMUN DENOMINATORE
1070 GOSUB 730
1080 N3 = N3 / V:D3 = D3 / V
1090 REM TERMINA
1100 GOSUB 730
1110 IF YN$ = "N" THEN RETURN
1120 GOTO 920
```

Figura 7.22: Listato di programma: Addizione di frazioni

SOTTRAZIONE DI FRAZIONI

Descrizione

Con questo programma potete esercitarvi a sottrarre frazioni. Per sceglierlo premete 2 quando compare il menù e fornite anche il denominatore più grande con cui volete lavorare. Come l'esercizio sulle sottrazioni del menù 1, questo programma genera numeri in modo che il risultato sia sempre positivo.

Esempio

La Figura 7.23 mostra un esempio di sottrazione di frazioni, la 7.24 contiene il listato del programma.

SOTTRAZIONE	$\frac{9}{7} - \frac{5}{9}$
LA RISPOSTA GIUSTA È:	$\frac{46}{63}$
BATTERE NUMERATORE:	?46
BATTERE DENOMINATORE:	?63
AVETE RISPOSTO GIUSTO!	
ESERCIZI SVOLTI:	1
CORRETTI:	1
ERRORI:	0

 VOLETE
 ESERCITARVI
 DI NUOVO? (S O N)

Figura 7.23: Videata: Sottrazione di frazioni

```

1130 REM -----
1140 N$ = "SOTTRAZIONE"
1150 REM -----
1160 GOSUB 7500 : REM INTESAZIONE
1170 REM GENERA NUMER. E DENOM.
1180 GOSUB 290
1190 REM SEMPLIFICAZIONE
1200 N3 = D2 * N1 - D1 * N2
1210 REM ESCLUDE ZERO
1220 IF N3 = 0 THEN 1180
1230 SN$ = "-"
1240 REM MOSTRA LA DOMANDA E
1250 REM RICEVE LA RISPOSTA
1260 GOSUB 560
1270 D3 = D1 * D2
1280 REM TROVA IL MASSIMO
1290 REM COMUN DENOMINATORE
1300 GOSUB 450
1310 N3 = N3 / V:D3 = D3 / V
1320 REM TERMINA
1330 GOSUB 730
1340 IF YN$ = "N" THEN RETURN
1350 GOTO 1130

```

Figura 7.24: Listato di programma: Sottrazione di frazioni

MOLTIPLICAZIONE DI FRAZIONI

Descrizione

Il programma di moltiplicazione di frazioni vi consente di esercitarvi nel moltiplicare due frazioni. Questo programma viene scelto dal menù delle frazioni premendo il numero 3. Imparare a moltiplicare frazioni è molto facile, ed è utile, così non abbiate paura di provare questo esercizio.

Esempio

La Figura 7.25 mostra un esempio del programma di moltiplicazione delle frazioni. Notate come sia stata data una risposta giusta a due problemi su quattro. La Figura 7.26 contiene il listato della moltiplicazione di frazioni.

MOLTIPLICAZIONE

$$\frac{5}{6} \times \frac{2}{6}$$

LA RISPOSTA GIUSTA È: $\frac{5}{18}$

BATTERE NUMERATORE: ?5
 BATTERE DENOMINATORE: ?18

AVETE RISPOSTO GIUSTO!

ESERCIZI SVOLTI: 4
 CORRETTI: 2
 ERRORI: 2

 VOLETE
 ESERCITARVI
 DI NUOVO? (S O N)

Figura 7.25: Videata: Moltiplicazione di frazioni

```

1360 REM -----
1370 N$ = "MOLTIPLICAZIONE"
1380 REM -----
1390 GOSUB 7500 : REM INTESTAZIONE
1400 REM GENERA NUMER. E DENOM.
1410 GOSUB 290
1420 SN$ = "X"
1430 REM MOSTRA LA DOMANDA E
1440 REM RICEVE LA RISPOSTA
1450 GOSUB 560
1460 REM SEMPLIFICAZIONE
1470 N3 = N1 * N2
1480 D3 = D1 * D2
1490 REM TROVA IL MASSIMO
1500 REM COMUN DENOMINATORE
1510 GOSUB 450
1520 N3 = N3 / V
1530 D3 = D3 / V
1540 REM TERMINA
1550 GOSUB 730
1560 IF YN$ = "N" THEN RETURN
1570 GOTO 1360
    
```

Figura 7.26: Listato di programma

DIVISIONE DI FRAZIONI

Descrizione

L'ultimo programma sulle frazioni permette di esercitarsi a dividere frazioni. Per sceglierlo, premete 4 quando compare il menù. La divisione di frazioni è come la moltiplicazione, solo che la seconda frazione (il divisore) va invertita. Questo programma consentirà di migliorare in questa operazione e la renderà meno ostica.

Esempio

La Figura 7.27 mostra un esempio del programma e la 7.28 ne contiene il listato.

DIVISIONE

$$\frac{6}{4} / \frac{3}{5}$$

LA RISPOSTA GIUSTA È:

$$\frac{5}{2}$$

BATTERE NUMERATORE: 75

BATTERE DENOMINATORE: 72

AVETE RISPOSTO GIUSTO!

ESERCIZI SVOLTI: 1

CORRETTI: 1

ERRORI: 0

VOLETE

ESERCITARVI

DI NUOVO? (S O N)

Figura 7.27: Videata: Divisione di frazioni

```
1580 REM -----
1590 N$ = "DIVISIONE"
1600 REM -----
1610 GOSUB 7500 : REM INTSTAZIONE
1620 REM GENERA NUMER. E DENOM.
1630 GOSUB 290
1640 REM SEMPLIFICAZIONE
1650 N3 = N1 * D2
1660 D3 = D1 * N2
1670 REM TROVA IL MASSIMO
1680 REM COMUN DENOMINATORE
1690 GOSUB 450
1700 N3 = N3 / V
1710 D3 = D3 / V
1720 SN$ = "/"
1730 REM MOSTRA LA DOMANDA
1740 REM E RICEVE LA RISPOSTA
1750 GOSUB 560
1760 REM TERMINA
1770 GOSUB 730
1780 IF YN$ = "N" THEN RETURN
1790 GOTO 1580
```

Figura 7.28: Listato di programma: Divisione di frazioni

APPENDICE A

SUBROUTINE CENTRALI

Queste subroutine ci hanno permesso di scrivere programmi utili molto brevi e quindi di facile inserimento; l'Appendice B spiega come possano essere usate in altre applicazioni.

Questa appendice contiene i listati delle subroutine utilizzate nel corso di tutto il libro; molte istruzioni REM facilitano la comprensione delle loro funzioni.

Per non doverle battere molte volte, le si può memorizzare in un unico file e aggiungerle ai vari programmi come spiegato nel Capitolo 1. Si ricordi che è sufficiente battere solo il primo REM di ciascuna subroutine.

```
4000 REM "INSERREC"  
4010 REM -----  
4020 REM INSERISCE RECORD NEL FILE  
4030 REM  
4040 REM QUESTA ROUTINE INSERISCE  
4050 REM RECORD IN UN FILE  
4060 REM  
4070 REM PARAMETRI DI RICHIESTA:  
4080 REM NF=NUMERO DEI CAMPI  
4090 REM IN UN RECORD (MAX 8)  
4100 REM F$(L)=NOMI DEI CAMPI  
4110 REM F$=NOME DEL FILE IN CUI INSERIRE  
4120 REM  
4130 REM PARAMETRI RISULTANTI:  
4140 REM NESSUNO  
4150 REM  
4160 REM INSERISCE RECORD FINCHÉ  
4170 REM L'UTENTE INTERROMPE  
4180 REM VENGONO CREATI 8 CAMPI  
4190 REM ANCHE SE VUOTI  
4200 REM  
4210 REM -----
```

Figura A.1: Subroutine per l'inserimento di nuovi record

(continua)

```

4220 N1 = 0:D$ = CHR$ (4)
4230 K = 1: GOSUB 7500
4240 N1=N1+1 : Q1$ = "INSERIRE DATI"
4250 Q2$ = ""
4260 FOR L = 1 TO NF:Q3$ = F$(L): GOSUB 7000: NEXT L
4270 FOR L = 1 TO 8:S$(N1,L) = P$(L): NEXT L
4280 IF N1 = 10 THEN GOSUB 4340:N1 = 0
4290 Q1$ = "INSERIRE UN ALTRO RECORD"
4300 Q2$ = "" : Q3$ = "SI' O NO (Y/N)"
4310 GOSUB 6000 : GOSUB 9500 : IF YN$ = "S" THEN 4230
4320 IF N1 > 0 THEN GOSUB 4340
4330 RETURN
4340 GOSUB 5000
4350 PRINT D$;"OPEN";FB$: PRINT D$;"OPEN",F$
4370 PRINT D$;"READ";FB$: INPUT N2: PRINT D$
4380 PRINT D$;"WRITE";F$: PRINT N1 + N2: PRINT D$
4390 FOR L = 1 TO N2: PRINT D$: "READ";FB$
4400 INPUT P1$, P2$, P3$, P4$, P5$, P6$, P7$, P8$
4410 PRINT D$: PRINT D$;"WRITE";F$: PRINT P1$
4420 PRINT P2$: PRINT P3$: PRINT P4$: PRINT P5$
4430 PRINT P6$: PRINT P7$: PRINT P8$: NEXT L
4440 PRINT D$;"WRITE";F$: FOR L = 1 TO N1
4450 FOR L1 = 1 TO 8: PRINT S$(L,L1): NEXT L1,L
4460 PRINT D$;"CLOSE": RETURN
4488 PRINT D$;"DELETE";F$: PRINT D$;"OPEN"; F$

```

Figura A.1: Subroutine per l'inserimento di nuovi record

```

4500 REM "ANCORA"
4510 REM -----
4520 REM RICHIESTA DI NUOVA ESECUZIONE
4530 REM
4540 REM QUESTA ROUTINE DOMANDA ALL'UTENTE SE CONTINUARE
4560 REM
4570 REM PARAMETRI DI RICHIESTA:
4580 REM Q2$=STRINGA CHE SPECIF. OPERAZ. DA SVOLGERE
4600 REM
4610 REM PARAMETRI RISULTANTI:
4620 REM YN$ = "S" O "N" PER SI' O NO
4640 REM
4650 REM -----
4660 Q1$ = "VOLETE"
4670 Q3$ = "DI NUOVO? (S O N)"
4680 REM DIALOGO IN CORSO
4690 GOSUB 6000
4700 REM DOMANDA SE SI' O NO
4710 GOSUB 9500
4720 RETURN

```

Figura A.2: Subroutine di richiesta di nuova esecuzione

```

5000 REM "BAKFIL"
5010 REM -----
5020 REM COPIA UN FILE
5040 REM QUESTA ROUTINE CREA UNA COPIA DI UN FILE
5070 REM PARAMETRI DI RICHIESTA:
5080 REM F$ = NOME DEL FILE DA COPIARE
5090 REM FB$ = NOME DEL FILE DI COPIA
5110 REM -----
5120 PRINT CHR$(4);"OPEN";F$
5130 PRINT CHR$(4);"OPEN";FB$
5140 PRINT CHR$(4);"DELETE";FB$
5150 PRINT CHR$(4);"OPEN";FB$
5160 PRINT CHR2(4);"READ";F%
5170 INPUT N: PRINT CHR$(4)
5180 PRINT CHR$(4);"WRITE";FB$
5190 PRINT N: PRINT CHR$(4)
5200 FOR R = 1 TO N
5210 PRINT CHR$(4);"READ";F$
5220 INPUT P1$, P2$, P3$, P4$, P5$, P6$, P7$, P8$
5230 PRINT CHR$(4)
5240 PRINT CHR$(4);"WRITE";FB$
5250 PRINT P1$: PRINT P2$: PRINT P3$: PRINT P4$
5260 PRINT P5$: PRINT P6$: PRINT P7$: PRINT P8$
5270 PRINT CHR$(4)
5280 NEXT R
5290 PRINT CHR$(4);"CLOSE"
5300 RETURN

```

Figura A.3: Subroutine per creare un file di copia (backup)

```

5500 REM "CANGREC"
5510 REM -----
5520 REM CANCELLA RECORD
5530 REM
5540 REM QUESTA ROUTINE CANCELLA
5550 REM UN RECORD DA UN FILE
5560 REM
5570 REM PARAMETRI DI RICHIESTA:
5580 REM FB$ = NOME DELLA COPIA
5590 REM DA CREARE
5600 REM F$ = NOME DEL FILE DA CUI
5610 REM CANCELLARE
5620 REM
5630 REM PARAMETRI RISULTANTI:
5640 REM L=-1 SE IL RECORD NON
5650 REM VIENE TROVATO
5660 REM
5670 REM ALL'UTENTE VIENE DOMANDATO
5680 REM IL NUMERO DEL RECORD
5690 REM VIENE CREATO UN FILE
5700 REM DI COPIA
5710 REM
5720 REM -----
5730 N$ = "CANCELLARE RECORD" : GOSUB 7500 : D$=CHR$(4)
5740 Q1$ = "NUMERO DEL RECORD DA CANCELLARE"
5750 Q2$ = "":Q3$ = "": GOSUB 6000
5760 INPUT L : L=L+1 : REM RICEVE IL NUMERO DEL RECORD
5770 GOSUB 5000 : REM FILE DI COPIA
5780 PRINT D$;"OPEN";FB$: PRINT D$;"OPEN";F$
5790 PRINT D$;"READ";FB$
5800 INPUT N: IF N < L THEN L = - 1: GOTO 5990
5810 PRINT D$: PRINT D$;"WRITE";F$: PRINT N - 1
5820 PRINT D$: IF L = 1 THEN 5900
5830 FOR I = 1 TO L - 1
5840 PRINT D$;"READ";FB$
5850 INPUT P1$, P2$, P3$, P4$, P5$, P6$, P7$, P8$
5860 PRINT D$: PRINT D$;"WRITE";F$
5870 PRINT P1$: PRINT P2$: PRINT P3$: PRINT P4$
5880 PRINT P5$: PRINT P6$: PRINT P7$: PRINT P8$
5890 PRINT D$: NEXT I
5900 PRINT D$;"READ";FB$
5910 INPUT P1$, P2$, P3$, P4$, P5$, P6$, P7$, P8$
5920 IF N = L THEN 5990
5930 FOR I = L + 1 TO N: PRINT D$;"READ";FB$
5940 INPUT P1$, P2$, P3$, P4$, P5$, P6$, P7$, P8$
5950 PRINT D$: PRINT D$;"WRITE";F$
5960 PRINT P1$: PRINT P2$: PRINT P3$: PRINT P4$
5970 PRINT P5$: PRINT P6$: PRINT P7$: PRINT P8$
5980 PRINT D$: NEXT I
5990 PRINT D$;"CLOSE": RETURN

```

Figura A.4: Subroutine per cancellare record

```

6000 REM "DIALOGO"
6010 REM -----
6020 REM CONVERSAZIONE
6030 REM
6040 REM PARAMETRI DI RICHIESTA:
6050 REM Q1$, Q2$, Q3$ SONO LE DOMANDE
6060 REM DA MOSTRARE SULLO SCHERMO
6070 REM
6080 REM PARAMETRI RISULTANTI:
6090 REM NESSUNO
6100 REM
6110 REM LE STRINGHE DI DOMANDA NON UTILIZZATE
6120 REM DOVREBBERO ESSERE RESE UGUALI A ZERO (""")
6130 REM
6140 REM -----
6150 REM SGOMBRA LA ZONA DI DIALOGO
6160 REM TRACCIA IL BORDO, ANNULLA IL CONTENUTO DELLE RIGHE
6170 FOR I = 20 TO 23
6180 REM SCELTA DI UNA RIGA
6190 VTAB I: HTAB 1
6200 FOR J = 1 TO 39
6210 IF I = 20 THEN PRINT "-";
6220 IF I < > 20 THEN PRINT " ";
6230 NEXT J
6240 NEXT I
6250 VTAB 21: HTAB 1
6260 PRINT Q1$ : REM PRIMA DOMANDA
6270 PRINT Q2$ : REM SECONDA DOMANDA
6280 PRINT Q3$ : REM TERZA DOMANDA
6290 RETURN

```

Figura A.5: Subroutine per videate di dialoghi

```

6500 REM "INCAR"
6510 REM -----
6520 REM INPUT DI CARATTERI
6530 REM
6540 REM PARAMETRI DI RICHIESTA:
6550 REM NESSUNO
6560 REM
6570 REM PARAMETRI RISULTANTI:
6580 REM CH$ = CARATTERE DI INPUT
6590 REM
6600 REM -----
6610 REM ASPETTA IL CARATTERE DI INPUT
6620 GET CH$
6630 RETURN

```

Figura A.6: Subroutine per input di caratteri

```

7000 REM "INDAT"
7010 REM -----
7020 REM INPUT DI DATI
7030 REM
7040 REM PARAMETRI DI RICHIESTA:
7050 REM Q1$, Q2$ = ISTRUZIONI DELL'UTENTE
7070 REM Q3$ = NOME DEI SINGOLI DATI
7080 REM
7090 REM PARAMETRI RISULTANTI:
7100 REM P$(K) = MATRICE DEI DATI
7110 REM
7120 REM VIENE INSERITA UNA RIGA PER VOLTA
7140 REM PRIMA RICHIESTA CON K=1
7150 REM K WILL BE UPDATED AUTOMATICALLY
7160 REM
7170 REM -----
7180 REM FA DOMANDE
7190 GOSUB 6000
7200 REM OTTIENE I DATI
7210 INPUT P$(K)
7220 REM MOSTRA SULLO SCHERMO LA DOMANDA E LA RISPOSTA
7230 VTAB K + 2: HTAB 3
7240 PRINT Q3$; TAB( 25);P$(K)
7250 REM AGGIORNA L'INDICE
7260 K = K + 1
7270 RETURN

```

Figura A.7: Subroutine per input di dati

```

7500 REM "INTEST"
7510 REM -----
7520 REM VIDEATA DELL'INTESTAZIONE
7530 REM
7540 REM QUESTA ROUTINE CANCELLA
7550 REM LO SCHERMO E MOSTRA
7560 REM IL TITOLO DEL PROGRAMMA
7570 REM
7580 REM PARAMETRI DI RICHIESTA:
7590 REM N$ = TITOLO DEL PROGRAMMA
7600 REM
7610 REM PARAMETRI RISULTANTI:
7620 REM NESSUNO
7630 REM
7640 REM -----
7650 REM CANCELLA LO SCHERMO
7660 HOME
7670 REM TITOLO
7680 PRINT N$: PRINT
7690 RETURN

```

Figura A.8: Subroutine per l'intestazione e le videate

```

8000 REM "INPAR"
8010 REM -----
8020 REM INPUT DEI PARAMETRI
8030 REM
8040 REM PARAMETRI DI RICHIESTA
8050 REM Q1$, Q2$ SONO ISTRUZIONI DELL'UTENTE
8060 REM Q3$ = NOME DI DATO
8070 REM
8080 REM PARAMETRI RISULTANTI:
8090 REM PAR(K) = DATO OTTENUTO
8100 REM
8110 REM CHIEDE CON K=1 PER
8120 REM IL PRIMO DATO
8130 REM K VIENE AGGIORNATO AUTOMATICAMENTE
8140 REM
8150 REM -----
8160 REM FA DOMANDE
8170 GOSUB 6000
8180 REM OTTIENE VALORE
8190 INPUT PAR(K)
8200 REM MOSTRA SU SCHERMO DOMANDA E RISPOSTA
8210 VTAB K + 2: HTAB 3
8220 PRINT Q3$: TAB( 25);PAR(K)
8230 K = K + 1
8240 RETURN

```

Figura A.9: Subroutine per l'input dei parametri del programma

```

8500 REM "MENU"
8510 REM -----
8520 REM PROGRAMMA DEL MENU'
8530 REM
8540 REM QUESTO PROGRAMMA MOSTRA
8550 REM UN MENU' E SCEGLIE UN PROGRAMMA
8560 REM
8570 REM PARAMETRI DI RICHIESTA:
8580 REM X = NUM. ELEMENTI DEL MENU'
8590 REM X$(I) = MATRICE DEI NOMI DEI PROGRAMMI
8600 REM
8610 REM PARAMETRI RISULTANTI:
8620 REM X = NUMERO DEL PROGRAMMA SCELTO
8630 REM
8640 REM -----
8650 HOME
8660 REM MOSTRA IL MENU'
8670 FOR I = 1 TO N
8680 IF I < 10 THEN PRINT I;"=" ";X$(I)
8690 IF I = 10 THEN PRINT "0=" ";X$(I)
8700 NEXT I
8710 REM FA LE DOMANDE
8720 Q1$="" : Q2$="" : Q3$ = "PROGRAMMA SCELTO: "
8730 GOSUB 6000
8740 REM INPUT DI UN CARATTERE
8750 GOSUB 6500
8760 X = VAL (CH$): IF X = 0 THEN X = 10
8770 REM CONTROLLA SE CARATTERE RIENTRA NEL CAMPO
8780 IF X >= 1 AND X <= N THEN RETURN
8790 Q1$ = "SCELTA NON VALIDA, SCEGLIERE DI NUOVO"
8800 GOSUB 6000
8810 GOTO 8750

```

Figura A.10: Subroutine per mostrare il menù

```

9000 REM "PAUSA"
9010 REM -----
9020 REM ASPETTA CHE VENGA BATTUTO UN TASTO
9030 REM
9040 REM QUESTA ROUTINE ASPETTA CHE
9050 REM L'UTENTE BATTI UN TASTO
9060 REM
9070 REM PARAMETRI DI RICHIESTA:
9080 REM NESSUNO
9090 REM
9100 REM PARAMETRI RISULTANTI:
9110 REM NESSUNO
9120 REM
9130 REM -----
9140 Q1$ = "PER CONTINUARE BATTERE UN TASTO"
9150 Q2$ = ""; Q3$ = ""
9160 REM RICHIAMA IL PROGRAMMA DI DIALOGO
8170 GOSUB 6000
9180 REM ASPETTA CHE VENGA BATTUTO UN TASTO
9190 GET X$
9200 RETURN

```

Figura A.11: Subroutine di pausa

```

9500 REM "SI' /NO"
9510 REM -----
9520 REM SI' O NO
9530 REM
9540 REM PARAMETRI DI RICHIESTA:
9550 REM NESSUNO
9560 REM
9570 REM PARAMETRI RISULTANTI:
9580 REM YN$ = "S" O "N"
9590 REM
9600 REM -----
9610 REM ASPETTA CHE VENGA BATTUTA LA RISPOSTA
9620 GET YN$
9630 IF YN$ = "S" OR YN$ = "N" THEN 9660
9640 REM NON SI' O NO
9650 GOTO 9620
9660 PRINT YN$: RETURN

```

Figura A.12: Subroutine si o no

COME UTILIZZARE LE SUBROUTINE CENTRALI

I lettori che vogliono migliorare nella programmazione dovrebbero scrivere i propri programmi; tramite le subroutine centrali di questo libro possono risparmiare molto tempo e sforzi. Questa appendice vi aiuterà a imparare a usarle: ne verranno esaminate diverse e ne verrà analizzato il funzionamento. Includiamo anche alcuni programmi che mostrano come sia possibile utilizzarle.

SCRITTURA DI UN MENU'

La subroutine di scrittura di menù inizia alla riga 8500. Questa subroutine fa apparire un menù sullo schermo e aspetta che l'utente prema un tasto numerico per indicare il programma che vuol lanciare.

Le Figure B.1 e B.2 mostrano il listato del programma e quanto appare sullo schermo per un menù esemplificativo. Costruiamo la matrice $X(I)$ con i nomi dei programmi disponibili; poi rendiamo N uguale al numero delle scelte possibili e richiamiamo la subroutine del menù. Questa subroutine mostra i nomi dei programmi e aspetta che l'utente scelga, dopo di che dà il valore del programma prescelto. Il valore inserito viene utilizzato da un'istruzione GOSUB per la scelta di un programma.

```

100 REM -----
110 N$ = "DIMOSTRAZIONE DI MENU"
120 REM -----
130 GOSUB 7500 : REM INTESTAZIONE
140 REM CREA LA MATRICE DEL MENU'
150 X$(1) = "PROGRAMMA 1"
160 X$(2) = "PROGRAMMA 2"
170 X$(3) = "PROGRAMMA 3"
180 X$(4) = "PROGRAMMA 4"
190 REM MOSTRA MENU'
200 N = 4: GOSUB 8500
210 VTAB 6: HTAB 1
220 REM SCELTA PROGRAMMA
230 ON X GOTO 240, 250, 260, 270
240 PRINT X$(1); " SCELTO" : END
250 PRINT X$(2); " SCELTO" : END
260 PRINT X$(3); " SCELTO" : END
270 PRINT X$(4); " SCELTO" : END

```

Figura B.1: Listato di programma: Esempio di menù

```

1= PROGRAMMA 1
2= PROGRAMMA 2
3= PROGRAMMA 3
4= PROGRAMMA 4

```

```

-----
PROGRAMMA SCELTO:

```

Figura B.2: Videata: Esempio di menù

SCELTA DI UN ALTRO PROGRAMMA

Tutti i programmi di questo libro utilizzano la subroutine ANCORA. Questa subroutine domanda all'utente, mostrando tre righe di dati, se vuol lanciare di nuovo il programma. La subroutine genera il primo e il terzo rigo, mentre il secondo va specificato quando la si richiama.

La subroutine ANCORA memorizza S o N nella variabile YN\$, che il programma valuta per decidere se rilanciare o meno il programma.

Le Figure B.3 e B.4 mostrano il listato del programma e quanto appare sullo schermo in un'utilizzazione esemplificativa di ANCORA. Si noti come in questo esempio la domanda che appare è DIMOSTRARE QUESTA ROUTINE.

```
280 REM -----
290 N$ = "DIMOSTRAZIONE DELLA SUBROUTINE 'ANCORA'"
300 REM -----
310 GOSUB 7500 : REM INTESTAZIONE
320 Q2$ = "DIMOSTRARE QUESTA ROUTINE"
330 GOSUB 4500 : REM DOMANDA SE ANCORA
340 IF YN$ = "S" THEN 280
350 END
```

Figura B.3: Listato di programma: Dimostrazione della subroutine ANCORA

DIMOSTRAZIONE DELLA SUBROUTINE 'ANCORA'

```
-----
VOLETE
DIMOSTRARE QUESTA ROUTINE
DI NUOVO? (S O N)
```

Figura B.4: Videata: Dimostrazione della subroutine ANCORA

LA SUBROUTINE DI DIALOGO

Questa subroutine viene utilizzata per far apparire tre righe di testo in fondo allo schermo. È possibile generare tre domande diverse; tre domande o istruzioni sono sufficienti per la maggior parte dei programmi. Spesso viene scritta un'istruzione sul primo rigo e domande sul secondo e sul terzo. Questa routine traccia inoltre un bordo e cancella dallo schermo eventuali residui di domande precedenti. La routine termina portando il cursore dopo la terza domanda. Le Figure B.5 e B.6 mostrano il listato e quanto appare sullo schermo per un'esemplificazione di un DIALOGO.

```
360 REM -----
370 N$ = "DIMOSTRAZIONE DELLA SUBROUTINE 'DIALOGO'"
380 REM -----
390 GOSUB 7500 : REM INTESTAZIONE
400 REM INSERIMENTO DELLE DOMANDE
410 Q1$ = "QUESTA È LA PRIMA DOMANDA"
420 Q2$ = "" : REM DOMANDA NULLA
430 Q3$ = "QUESTA È L'ULTIMA DOMANDA"
440 REM INVIA LE DOMANDE
450 GOSUB 6000
460 END
```

Figura B.5: Listato di programma: Dimostrazione della subroutine DIALOGO

DIMOSTRAZIONE DELLA SUBROUTINE 'DIALOGO'

QUESTA È LA PRIMA DOMANDA

QUESTA È L'ULTIMA DOMANDA

Figura B.6: Videata: Dimostrazione della subroutine DIALOGO

SUBROUTINE PER L'IMMISSIONE DEI DATI

Il programma IMMATR gestisce l'immissione di dati per molti dei programmi del libro e può essere facilmente utilizzabile per altri programmi. Si dà inizio al processo di immissione cancellando lo schermo con la subroutine INIZ e rendendo l'indice K uguale a 1. Siete adesso pronti a inserire parametri dei programmi. Per ogni parametro, rendete Q3\$ uguale al nome del parametro e richiamate la subroutine a 8000. La subroutine IMMATR domanda il nome del parametro in fondo allo schermo e aspetta che venga fornito un dato. Il parametro compare poi sulla parte superiore dello schermo. IMMATR memorizza i parametri nella matrice PAR(K) nell'ordine di inserimento. Le Figure B.7 e B.8 mostrano il listato di un programma e quanto appare sullo schermo in un esempio di utilizzazione di questa subroutine.

```

470 REM -----
480 N$ = "DIMOSTRAZIONE DI INPUT DI DATI"
490 REM -----
500 GOSUB 7500 : REM INTESTAZIONE
520 REM PREDISPONE ZONA DI DIALOGO
530 Q1$ = "INSERIMENTO DATI" : Q2$ = ""
540 REM RICEVE IL PRIMO DATO
550 Q3$ = "PREZZO AL DETTAGLIO" : GOSUB 8000
560 REM RICEVE SECONDO DATO
570 Q3$ = "SCONTO (%)" : GOSUB 8000
580 PRINT : PRINT "PREZZO ="; TAB(25)
590 P = PAR(1) - PAR(2) / 100 * PAR(1)
600 REM ARROTONDAMENTO A DUE CIFRE
610 P = INT (P * 100 + .5) / 100
620 PRINT P
630 END

```

Figura B.7: Listato di programma: Dimostrazione di input di dati

```

DIMOSTRAZIONE DI INPUT DI DATI

      PREZZO AL DETTAGLIO    300000
      SCONTO (%)            10

PREZZO =                      270000

```

```

-----
INSERIMENTO DATI
SCONTO (%)?10

```

Figura B.8: Videata: Dimostrazione di input di dati

UTILI SUBROUTINE PER LA STAMPANTE

Queste subroutine vi torneranno utili se possedete una stampante Epson, che è in grado di scrivere in molti modi interessanti. Potete stampare in modalità normale, allargata, condensata, in risalto, ribattuta e in risalto e ribattuta contemporaneamente. La Figura C.1 mostra le varie modalità.

Le subroutine di questa appendice consentono di accedere ai diversi stili di scrittura della stampante. La modalità in risalto è molto utile per scrivere lettere; quella condensata scrive 132 colonne su fogli da 80 colonne.

La Figura C.2 mostra il programma del menù di scelta per controllare la stampante. La Figura C.3 elenca le subroutine utilizzabili per scegliere i vari stili. La Figura C.4 dà il listato del programma che utilizza le subroutine della stampante per generare gli stili della Figura C.1.

MODALITA' DI STAMPA DISPONIBILI SULLA STAMPANTE EPSON

QUESTA E' LA STAMPA NORMALE

QUESTA E' LA STAMPA CONDENSATA

QUESTA E' LA STAMPA ALLARGATA

QUESTA E' LA STAMPA IN RISALTO

QUESTA E' LA STAMPA RIBATTUTA

QUESTA E' LA STAMPA IN RISALTO RIBATTUTA

Figura C.1: Modalità di stampa disponibili sulla stampante Epson

```
100 REM -----
110 N$ = "FUNZIONI PER CONTROLLO STAMPA"
120 REM -----
130 GOSUB 7500
140 REM CREA LA MATRICE DEL MENU'
150 X$(1) = "MODALITA' CONDENSATA SI'"
160 X$(2) = "MODALITA' CONDENSATA NO"
170 X$(3) = "MODALITA' ALLARGATA SI'"
180 X$(4) = "MODALITA' ALLARGATA NO"
190 X$(5) = "MODALITA' IN RISALTO SI'"
200 X$(6) = "MODALITA' IN RISALTO NO"
210 X$(7) = "MODALITA' RIBATTUTA SI'"
220 X$(8) = "MODALITA' RIBATTUTA NO"
230 X$(9) = "STAMPA TESTO ESEMPLIFICATIVO"
240 REM MOSTRA MENU'
250 N = 9: GOSUB 8500
260 ON X GOSUB 280, 330, 380, 430, 480, 530, 580, 630, 680
270 GOTO 100
```

Figura C.2: Listato di programma: Menù di scelta dei controlli di stampa

```

280 REM -----
290 N$ = "STAMPA CONDENSATA SI"
300 REM -----
310 PRINT CHR$ (15);
320 RETURN
330 REM -----
340 N$ = "STAMPA CONDENSATA NO"
350 REM -----
360 PRINT CHR$ (18);
370 RETURN
380 REM -----
390 N$ = "STAMPA ALLARGATA SI"
400 REM -----
410 PRINT CHR$ (14);
420 RETURN
430 REM -----
440 N$ = "STAMPA ALLARGATA NO"
450 REM -----
460 PRINT CHR$ (20);
470 RETURN
480 REM -----
490 N$ = "STAMPA IN RISALTO SI"
500 REM -----
510 PRINT CHR$ (27) + CHR$ (69);
520 RETURN
530 REM -----
540 N$ = "STAMPA IN RISALTO NO"
550 REM -----
560 PRINT CHR$ (27) + CHR$ (70);
570 RETURN
580 REM -----
590 N$ = "STAMPA RIBATTUTA SI"
600 REM -----
610 PRINT CHR$ (27) + CHR$ (71);
620 RETURN
630 REM -----
640 N$ = "STAMPA RIBATTUTA NO"
650 REM -----
660 PRINT CHR$ (27) + CHR$ (72);
670 RETURN

```

Figura C.3: Listato di programma: Subroutine per scegliere gli stili di stampa

```

680 REM -----
690 N$ = "TESTO ESEMPLIFICATIVO"
700 REM -----
710 HOME
720 PRINT "ACCENDERE LA STAMPANTE"
730 PRINT "QUANDO PRONTI BATTERE UN TASTO"
740 GOSUB 9000 : REM ASPETTA CARATTERE
750 REM PER IL TITOLO USA CARATTERI ENFATIZZATI
760 PR 1
770 GOSUB 480
780 PRINT
790 PRINT "MODALITA' DI STAMPA DISPONIBILI SULLA STAMPANTE EPSON"
800 GOSUB 530
810 PRINT : PRINT : PRINT
820 T1$ = "QUESTA È LA STAMPA "
830 PRINT T1$; "NORMALE" : PRINT
840 GOSUB 280
850 PRINT T1$; "CONDENSATA" : PRINT
860 GOSUB 330: GOSUB 380
870 PRINT T1$; "ALLARGATA" : PRINT
880 GOSUB 430: GOSUB 480
890 PRINT T1$; "IN RISALTO" : PRINT
900 GOSUB 530: GOSUB 580
910 PRINT T1$; "RIBATTUTA" : PRINT
920 GOSUB 630: GOSUB 480: GOSUB 580
930 PRINT T1$; "IN RISALTO E RIBATTUTA"
940 GOSUB 530: GOSUB 630: PR 0
950 END

```

Figura C.4: Listato di programma: Subroutine utilizzate per scegliere gli stili di stampa della Figura C.1

730863

Non c'è bisogno di essere un programmatore per leggere e usare questo libro; le poche informazioni necessarie sono contenute nell'introduzione. Leggetela, sedetevi davanti al calcolatore e cominciate dal programma che vi ispira di più.

Ce ne sono tanti, più di 65, e risolvono problemi che vanno dalla "economia domestica", alle applicazioni commerciali, ai calcoli statistici, alla creazione degli archivi.

E proprio se non siete già dei programmatori, dopo un po' vi verrà la tentazione di provare a realizzare qualche programma da soli. Non è difficile: con qualche rudimento di BASIC potete capire come sono fatti i programmi di questo libro, cominciate a modificarli e poi realizzare applicazioni originali, anche di una certa complessità.

I sottoprogrammi di uso generale che sono presentati nelle appendici semplificheranno notevolmente il vostro lavoro.

Basic Skills Program in Mathematics

Stanley R. Trost

GRUPPO
EDITORIALE
JACKSON

